Ingo Kl¨ockl

OCHEM

Benutzerhandbuch OCHEM Version 3-0e 2001-04-10

18. April 2001

Ingo Kl¨ockl Institut f¨ur Kernchemie Universit¨at Mainz Fritz Strassmann-Weg 2 D-55099 Mainz kloeckl@vkcmzd.chemie.uni-mainz.de ingo.kloeckl@2k-software.de Gesetzt mit L^ATEX und OCHEM Version 3-0e 2001-04-10 .

OCHEM Version 3-0e 2001-04-10 © 1999– 2001 Ingo Kl¨ockl This program can be redistributed and/or modifi ed under the terms of the LaTeX Project Public License Distributed from CTAN archives in directory

macros/latex/base/lppl.txt; either version 1 of the License, or any later versi-

on.

Inhaltsverzeichnis

1 Einführung 1

1.1 ¨Ubersicht 1

1.2 Installation 2

1.3 Einbindung in L^ATEX 3

1.4 Ausgabeformate 11

1.5 Makrodefi nitionen 12

1.6 Ausblick 14

2 Tutorial 15

2.1 Grundlegende Elemente 15

2.1.1 Einfache Formeln 15 2.1.2 Formeln anordnen 18

2.2 Komplexere Formeln 20

2.2.1 Symbolische Winkel 20 2.2.2 Lineare Systeme 24 2.2.3 Linear anellierte Ringe 24 2.2.4 Manuelle Erstellung anellierter Systeme 27 2.2.5 Angular anellierte Ringe 29 2.2.6 (Kohlenstoff-) Br¨ucken 30

2.3 Reaktionsketten 33

2.3.1 Horizontale lineare Anordnung 33

2.3.2 Mehrzeilige Schemata 36 2.3.3 Vertikale Anordnung 37 2.3.4 Kontext und Verzweigung 40 2.3.5 Reaktionsabschnitte als Einheit betrachten 41 2.3.6 Kontexte zusammenfassen 43 2.3.7 Zusammenf¨uhrung mehrerer Zweige 47 2.3.8 Mischung der Grundelemente 54

2.4 Zyklohexan und Saccharide 55

2.5 ¨Anderung von Parametern 61

2.6 Elektronendarstellungen 62

2.6.1 Elektronenkonfi guration 62 2.6.2 Elektronenverschiebungen 63

2.7 Beispiele 65

2.7.1 Nylonsynthese 66 2.7.2 Terpen-Biosynthese 68

2.8 Weitere Verbindungsklassen 72

2.9 Entwicklung eigener Ringstrukturen 72

3 Alphabetische Befehlsreferenz 77

3.1 Benutzerbefehle 77

3.1.1 arrow 78 3.1.2 atom 81 3.1.3 bond 83 3.1.4 bracket 90 3.1.5 branch 91 3.1.6 joinh 92 3.1.7 joinv 93 3.1.8 emove 94 3.1.9 fbox 96 3.1.10 formula 97 3.1.11 gotoXY 98 3.1.12 multiline 99

3.1.13 orbital 100 3.1.14 restore 101 3.1.15 restoreXY 102 3.1.16 ring 103 3.1.17 save 110 3.1.18 savecontext 110 3.1.19 saveXY 111 3.1.20 scale 114 3.1.21 shiftXY 115 3.1.22 set 116 3.1.23 setcontext 117 3.1.24 space 119

3.2 Sonderbefehle 120

3.2.1 font 121 3.2.2 package 122 3.2.3 require 124 3.2.4 schema 125

3.3 Die Bibliothek mncyclib.pm 126

3.4 Die Bibliothek bicyclib.pm 128

3.5 Die Bibliothek polycyclib.pm 130

3.6 Das Includefi le natur.inc 133

3.7 Das Includefi le utils.inc 139

3.8 Includefi les f¨ur Strukturen nach KEKULE: condensed.inc und cyclohexanes.inc 141

A Das Programm 143

A.1 History 143

A.2 Danke sch¨on! 143

B Mehr Stoff für Chemiesüchtige 145

Index 147

Die Reaktionsschemata

1– 1 Verwendung von Ringstrukturen, hier des Bizyklohexans, aus einer Bibliothek. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8

2– 1 Die Positionierungsparameter vermitteln den Zusammenhang zwischen den einzelnen Formeln. In der linken H¨alfte von links nach rechts angeordnete Formeln, in der rechten H¨alfte ¨ubereinandergestellte Formeln (von oben nach unten). . . . . . . . . . . . . . . . . 19

2– 2 Horizontales Reaktionsschema: Darstellung eines Benzothiophens. 34

2– 3 Horizontales Reaktionsschema mit Formelnamen, die auf eine gemeinsame Linie unterhalb des Reaktionspfeils ausgerichtet sind (fester Abstand des Textes zur Mittellinie). . . . . . . . . . . . . . . . 35

2– 4 Horizontales Reaktionsschema mit Formelnamen, die automatisch in konstantem Abstand auf einer gemeinsamen Linie unterhalb des Reaktionspfeils ausgerichtet sind. . . . . . . . . . . . . . . . . . . . 36

2– 5 Synthese von Phenanthren, mehrzeilig gesetzt mit multiline. . . 38

2– 6 Vertikales Reaktionsschema: Darstellung der 7-Oxononans¨aure. Rechts dasselbe Schema mit Bennenung der Komponenten. Die Texte weisen einen festen Abstand zur Formelunterkante auf. . . . . . . . . 39

2– 7 Was passiert mit dem Boran? Mehrfachverzweigungen, von derselben Formel ausgehend, k¨onnen es zeigen. Der Kontext der Verzweigungsstelle, hier des Borans, muß daf¨ur gespeichert werden. . . . . 42

2– 8 Ein unabh¨angiges vertikales Schema wird als einzeiliges multiline- Objekt gesetzt und kann so als Einheit in den horizontalen Formelverlauf eingef¨ugt werden. . . . . . . . . . . . . . . . . . . . . . . . 43

2– 9 Ein Superkontext mit multiline erlaubt die Klammerung mehrerer Formeln. Er wird durch ein multiline-Objekt gebildet. . . . . . . 44

2– 10 Unabh¨angig vom Superkontext k¨onnen Verzweigungen auf die einzelnen Teilformeln ausgerichtet werden, wenn deren Kontexte einzeln gespeichert werden. . . . . . . . . . . . . . . . . . . . . . . . 45

2– 11 Mit multiline angeordnete Formelbeschriftungen und Verzweigungen, erm¨oglicht durch individuelle Kontexte. . . . . . . . . . . . . 46

2– 12 Die Synthese eines Kronenethers als Beispiel f¨ur die Zusammenf¨uhrung zweier horizontaler Ketten mit joinh. . . . . . . . . . . . . . . . . 49

2– 13 Homogene geschachtelte Zusammenf¨uhrungen mehrerer Reaktionszweige. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 50

2– 14 Synthese der Chrysanthemums¨aure, gesetzt mit joinv. . . . . . . . 52 viii Die Reaktionsschemata

2– 15 Synthese einer Diazoverbindung. Horizontale und vertikale Zusammenf¨ugung gemischt. . . . . . . . . . . . . . . . . . . . . . . . . . 53 2– 16 Gemischte joinv/joinh-Befehle. . . . . . . . . . . . . . . . . . . 54 2– 17 Zusammenf¨uhrung horizontaler Reaktionsketten und multiline gemischt. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 55 2– 18 Das Gleichgewicht zwischen den beiden Hauptkonformeren des Zyklohexans nutzt die verschiedenen Konformeren des Zyklohexans. . 61 2– 19 Hauptreaktionen k¨onnen durch h¨ohere Linienbreite von Nebenwegen abgehoben werden. Die Linienbreite wird mit set eingestellt. . 62 2– 20 Der kettenverl¨angernde Schritt der Biosynthese der Isoprenoide wird durch Elektronenverschiebungen verdeutlicht. . . . . . . . . . . . . 64 2– 21 Die Elektronenverschiebung zeigt deutlich den Verlauf der Friedel- Crafts-Acylierung eines Aromaten. . . . . . . . . . . . . . . . . . . 65 2– 22 Die Nylonsynthese als Beispiel f¨ur ein komplexes Schema. . . . . . 69 2– 23 Die Biosynthese der bizyklischen Monoterpene aus der geradkettigen Vorstufe Farnesol. . . . . . . . . . . . . . . . . . . . . . . . . . 73 3– 1 Anordnung der oberen und unteren Beschriftung mit verschiedener Ausrichtung. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 79 3– 2 Die Anordnung der oberen und unteren Beschriftung bei verschiedenen Winkeln des Pfeils. . . . . . . . . . . . . . . . . . . . . . . . 79 3– 3 Komplette Reaktionsketten k¨onnen als ” Beschriftung“ auf Pfeile aufgesetzt werden. . . . . . . . . . . . . . . . . . . . . . . . . . . . . 80 3– 4 Mehrere Reaktionsketten zweigen von derselben Formel ab. Der Kontext dieser Formel muß daf¨ur gespeichert werden. . . . . . . . . . . 119

1. Einführung

1.1 Übersicht

Das vorliegende Paket soll den Chemiker in die Lage versetzen, organische Strukturformeln sowie ganze Reaktionsschemata in L^ATEX-Texte zu integrieren. Ziel ist, anhand einer einfachen Beschreibung der Formeln und Reaktionen die genaue Anordnung der einzelnen Formeln innerhalb des Schemas einem Chemie-Compiler zu ¨uberlassen, ¨ahnlich wie TEX anhand der Befehle Textst¨ucke selbst¨andig anordnet. Analog der Erzeugung von Graphiken mit MetaPost werden die chemischen Beschreibungen mit der L^ATEX-Bearbeitung in eine Textdatei geschrieben, die nach dem Compilieren TEX-Fragmente mit den Formelbildern liefert. Die Fragmente werden automatisch an der Stelle der Chemiebeschreibung in das L^ATEX-Dokument eingef¨ugt. Es ist nicht erforderlich, diese Fragmente neben den eigentlichen Dokumentdateien aufzuheben, da sie jederzeit durch den Compiler neu erzeugt werden k¨onnn en, so wie auch Metapost-Graphiken wieder aus den Sourcen generierbar sind. Haben sich die Formeln jedoch nicht ge¨andert, ist eine Neucompilation nicht erforderlich und der Schritt der Compilation kann ausgespart werden. In diesem Falle m¨ussen die Fragmente nat¨urlich noch existieren.

Die Aufmerksamkeit der Formelzeichnung richtet sich vor allem auf eine planare Wiedergabe, worunter auch g¨angige schematische Stereodarstellungen verstanden werden, wie Sesselkonformationen des Zyklohexans oder Verbr¨uckungen bei Polyzyklen, zum Beispiel des Morphins oder Crassanins. Da das Paket von einer geometrischen Beschreibung der Strukturen ausgeht, k¨onnen komplizierte Ger¨uste, wie sie beispielsweise Alkaloiden zu eigen sind, gewisse Probleme aufwerfen, wenn Sie eine r¨aumliche Darstellung anfertigen wollen. Sie m¨ussen in einem solchen Falle ¨uber die gew¨ohnliche Menge an Bindungswinkeln und -l¨angen hinaus individuelle Berechnungen dieser Werte anstellen und fl eiß igen Gebrauch von saveXY und

restoreXY machen^1. OCHEM ist in dieser Hinsicht kein molecular modelling-Tool

(seufz). Ich glaube, alle zur Beschreibung auch solcher Strukturen erforderlichen (graphischen) Primitive bereitgestellt zu haben; wenn Sie keine M¨oglichkeit sehen, bestimmte Strukturen zu erzeugen, bin ich ¨uber einen Hinweis dankbar.

Mechanistiker k¨onnen mit Hilfe kleiner gebogener Pfeile Elektronenverschiebungen andeuten. Sie werden sehen, daß die unz¨ahligen Mausbewegungen bei Benutzung eines WYSIWYG-Graphikeditors eine getreue Repr¨asentierung in Textform besitzen und die Formelbeschreibung rasch anwachsen lassen k¨onnen. Auch hier gilt: falls jemand eine gute Idee zur Beschreibung der Pfeilchen hat, m¨oge er sie vorbringen (am besten mit Implementierung :-) )

Die Formeln werden mit maximaler Qualit¨at in Form eines Fragments mit PostScript- Code in die TEX-Dateien geschrieben, sodaß sie nur mit einem PostScript-Treiber und -drucker sichtbar sind! Verf¨ugen Sie ¨uber keinen Zugang zu PostScript-Druckern,

1Nichts f¨ur Warmduscher und Schattenparker also.

k¨onnen Sie mit dem Programm GV oder seinen Derivaten allerdings die PS-Ausgabedateien in andere Druckerformate konvertieren. Die Texte, die in den Formeln auftreten, werden mit TEX gesetzt und erfordern somit kein PostScript. Die Entscheidung zugunsten von PostScript h¨angt (neben der Menge an freier Zeit) vor allem mit den bescheidenen graphischen F¨ahigkeiten des originalen L^ATEX zusammen, das keine Linien unter beliebigem Winkel oder Kreise mit freiem Radius zeichnen kann. Der Compiler erzeugt jedoch eine generische Ausgabe, die vom Modul be.pm in verschiedene fi nale Formate umgesetzt wird. Hier ist auch ein L^ATEX-Ausgabeformat vorgesehen, das nur L^ATEX-Mittel nutzt, aber aus den erw¨ahnten Gr¨unden im allgemeinen keine so sch¨onen Formeln liefert (vergleichen Sie selbst, in welchen F¨allen die Ausgabe hochwertig ist und in welchen M¨angel auftreten). Dieses native L^ATEX-Ausgabeformat soll k¨unftig jedoch verbessert werden, wobei vermutlich das Paket eepic zum Einsatz kommen wird.

1.2 Installation

Der Chemie-Compiler besteht aus einem Perl-Script, das die eigentliche Compilierung vornimmt, sowie einigen Modulen, ebenfalls in Perl geschrieben. Sie ben¨otigen die folgenden Dateien in einem Perl-gerechten Verzeichnisbaum:

chemie.pl
streambuf.pm
loc.pm
bbox.pm
be.pm
mncyclib.pm
bicyclib.pm

Das File chemie.pl stellt den eigentlichen Compiler dar und muß in ein Verzeichnis kopiert werden, das im Suchpfad liegt (Variable PATH f¨ur Windows), die anderen Dateien enthalten Hilfsklassen (jaja, alles sehr objektorientiert, wow!). Die *lib.pm- Module sind Bibliotheken f¨ur Ringstrukturen. Alle .pm-Dateien m¨ussen in ein Verzeichnis im Perl-Baum kopiert werden, in dem Bibliotheksdateien liegen, typischerweise kann dies /perl/5.005/lib sein.

Zur Einbindung in L^ATEX-Dokumente ist die Paketdatei

ochem.sty

in einem von L^ATEX durchsuchten Verzeichnis erforderlich. Um die Gr¨oß en der TEX- Texte zu ermitteln, ruft der Compiler intern L^ATEX auf. Je nachdem, wie auf Ihrem Rechnern das L^ATEX-System gestartet wird, m¨ussen Sie den Wert von $latexcmd im Script chemie.pl ¨andern. Der Default ist zu folgendem Aufruf von L^ATEX ¨aquivalent:

latex <datei>

Die Zeichenfolge latex muß zum Beispiel durch tex386 ersetzt werden, falls Sie L^ATEX wie folgt starten:

tex386 <datei>

M¨ochten Sie die Formeln mit Mitteln von L^ATEX setzen, ben¨otigen Sie noch die Pakete

epic und eepic. Sie werden im jeweiligen L^ATEX-Dokument hinzugeladen.

Wollen Sie einen Pr¨aprozessor f¨ur Makroersetzung einsetzen, ben¨otigen Sie auch diesen. F¨ur den GNU-Pr¨aprozessor M4 existieren Makros zum Satz von Naturstoffen sowie allgemein-n¨utzlicher Natur in den folgenden Includedateien:

natur.inc
utils.inc

Diese Include-Dateien m¨ussen ebenfalls in ein Verzeichnis im Suchpfad kopiert werden.

1.3 Einbindung in L^ATEX

Die Beschreibung der Formeln und Reaktionsschemata erfolgt in einer speziellen Beschreibungssprache in ASCII-Dateien. Diese Dateien k¨onnen entweder manuell erstellt oder online aus einem L^ATEX-Dokument mit eingebetteten Formeln erzeugt werden. Hier soll haupts¨achlich die letzte Methode dargestellt werden, da sie die gr¨oß ere Bequemlichkeit bietet und die Formelbeschreibung und der korrespondierende Text nicht auf mehrere Dateien verteilt sind.

Sie m¨ussen im L^ATEX-Dokument das Paket ochem.sty laden und erhalten damit die neue Umgebung chemistry, innerhalb derer die Formelprogramme stehen:

% file beispiel.tex
\documentclass{...}
\usepackage{ochem}

\begin{document}

\begin{chemistry}[phenol]
  formula(C,C)
  { ring(){ 3: bond(90) atom("OH",L); }
  }
\end{chemistry}

\begin{chemistry}
  formula(C,C,"Benzen",HR,24)
  { ring(){ } }
\end{chemistry}

\end{document}

Die vollst¨andige Bearbeitung eines solchen Chemiedokumentes erfordert zwei L^ATEX- Bearbeitungen. Im ersten Lauf mit

latex beispiel

werden die Inhalte aller chemistry-Umgebungen in eine Datei \jobname.chm geschrieben, und zwar in Form einzelner schema-Befehle (siehe unten), die die Formeln und Schemata der einzelnen Umgebungen enthalten. Die Namen der Bl¨ocke werden durch den Parameter der chemistry-Umgebungen festgelegt und k¨onnen bei der Fehlersuche behilfl ich sein. Lassen Sie diesen Parameter fort, wird eine laufende Nummer eingesetzt.

Die nach dem ersten Lauf resultierende Datei \jobname.chm muß nun mit dem Formelcompiler ¨ubersetzt werden

chemie.pl beispiel.chm

und liefert f¨ur jede chemistry-Umgebung eine Datei, die L^ATEX-Code enth¨alt. Der Name dieser Dateien wird ebenfalls aus dem ersten Parameter und der Extension

.ctx gebildet, f¨ur das Beispiel ¨ubersetzen Sie also die Datei beispiel.chm und er-

halten phenol.ctx. (Die Extension .ctx kann als ” chemistry TEX“ gedeutet werden; aufgrund der tempor¨aren Natur dieser Dateien, die jederzeit durch Neucompilierung erzeug- und ¨uberschreibbar sind, m¨ussen sie nicht wie normale TEX-Files verwaltet werden, Befehle wie ls *.tex lieferten sonst eine un¨ubersehbare F¨ulle von Kleinstdateien, auch das Packen aller .tex-Dateien in ein Archiv w¨urde diese tempor¨aren Dateien miterfassen.) Der Compiler schreibt f¨ur jede bearbeitete chemistry- Umgebung deren Namen (phenol im Beispiel) in eckigen Klammern auf das Terminal:

This is OCHEM chemistry compiler version 1.0b 2001-04-10
requires OCHEM.STY 3.0e
[phenol.ctx] [1.ctx]

Diese Angabe ist n¨utzlich, da Sie, falls ein Abbruch durch fehlerhaften Chemiecode erfolgt, leicht die Umgebung fi nden k¨onnen, die den Fehler enth¨alt.

Die nach der Compilierung erhaltenen .ctx-Dateien werden bei der zweiten Bearbeitung mit L^ATEX

latex beispiel

automatisch eingeladen und bilden den vollst¨andigen Formelinhalt. Da in jedem Lauf versucht wird, sie zu laden, erhalten Sie f¨ur jede Datei die Warnung, daß sie noch nicht existiere.

L^ATEX-Umgebung für Direktbefehle

Bei einigen Gelegenheiten m¨ussen Sie direkt Chemiebefehle des Abschnitts 3.2 in die .chm-Zwischendatei schreiben, zum Beispiel um Bibliotheken zu laden oder Makrodefi nitionen f¨ur den Pr¨aprozessor einzuf¨ugen. Sie benutzen zu diesem Zwecke die chemspecial-Umgebung, die Ihre Befehle ohne Umsetzung in die Programmdatei schreibt:

\begin{chemspecial}
  <direkte Befehle>
\end{chemspecial}

Umgang mit dem Compiler

Zur Erzeugung der fertigen L^ATEX-Fragmente mit den Formelbildern muß das beim ersten L^ATEX-Lauf gebildete Programmfi le, hier beispiel.chm, compiliert werden. Sie benutzen dazu das Perlscript chemie.pl:

chemie.pl beispiel.chm

Eventuell m¨ussen Sie auf Ihrem Rechner, um das Perlscript zu starten, einen Aufruf der folgenden Form vornehmen:

perl chemie.pl beispiel.chm

Es existieren folgende Compilerswitches:

-type s : Legt das Ausgabeformat fest, s kann einen der Werte PS, PSLATEX oder LATEX annehmen. Default: PSLATEX.

-trace lev : Druckt eine Aufrufl iste der internen Funktionen bis zu einer maximalen Schachtelungsebene von lev aus, mit deren Hilfe eventuell Fehler besser gefunden werden k¨onnen (vor allem Programmierfehler :-) ). Default: 0.

-pp : Weist den Compiler an, einen Pr¨aprozessorlauf mit der Eingabedatei zu starten, um Makros et cetera aufzul¨osen. Sie m¨ussen chemie.pl f¨ur Ihren Pr¨aprozessor angepaß t haben. (Default: kein Pr¨aprozessorlauf.)

Ausgabedateien im reinen L^ATEX-Format werden mit folgendem Aufruf erzeugt:

chemie.pl -type LATEX beispiel.chm

Eingabekonventionen

Bei der Eingabe der Formeln sind einige Konventionen zu beachten:

• Texte wie Formelnamen oder Bezeichner von Variablen sind mit doppelten Anf¨uhrungszeichen einzuleiten und zu beenden:

set("rXS",24)
formula(C,C,"Formelname",HA,24)
{ ... }

Stehen innerhalb der Zeichenkette Anf¨uhrungszeichen, so sind diese zu verdoppeln:

formula(C,C,"Kaurans""aure",HA,24)
{ ... }

Die Verdoppelung ist ¨ahnlich wie in FORTRAN ein Hinweis f¨ur den Parser, ein literales Anf¨uhrungszeichen als Textbestandteil zu lesen und es nicht als Ende der Zeichenkette zu interpretieren.

Treten Anf¨uhrungszeichen zu Beginn des Namens auf, stehen also drei Zeichen " unmittelbar hintereinander, m¨ussen Sie, um Fehlinterpretationen zu vermeiden, folgende Konstruktion eingeben:

formula(C,C,"{}""Ostrogen",HA,24)
{ ... }

Die leere Klammer {} wird sp¨ater beim Textsatz mit L^ATEX ignoriert, trennt aber das einfache vom doppelten Anf¨uhrungszeichen.

• Maß angaben wie L¨angen sind entweder in einer symbolischen Form anzugeben (zum Beispiel N f¨ur eine normallange Bindung) oder als Maß zahl ohne Einheit, es wird stets Punkt (pt) angenommen.

Kommentare beginnen mit einem Prozentzeichen und gelten bis zum Ende der Zeile, in der dieses steht:

% ganze Zeile ist Kommentar

formula(L,R) % Formel von Atropin
{ ... }

Prozentzeichen innerhalb von Anf¨uhrungszeichen sind Bestandteil der Zeichenkette und werden nicht als Kommentareinleitung betrachtet (es k¨onnen also Ausbeuteangaben in die Schemata integriert werden).

Da der M4-Pr¨aprozessor das Hashzeichen # als Kommentarmarke benutzt, treten zwei Probleme auf:

• Prozentzeichen werden von M4 nicht als Kommentarbeginn erkannt und auch in den Kommentaren unerw¨unschterweise Ersetzungen durchgef¨uhrt.

• Tritt in einer Formel das Zeichen # auf, wie beispielsweise in bond(#1), w¨are durch den Pr¨aprozessorlauf der dem Hashzeichen folgende Zeilenrest verschwunden, was zu Fehlermeldungen des Chemiecompilers f¨uhrt.

Zu Beginn eines jeden Includefi les wird daher mit der Zeile

changecom(‘//’)

die Folge // als Kommentarmarke f¨ur M4 gew¨ahlt. Um Zeilen auszukommentieren, schreiben Sie dann

formula(L,R) % normaler Kommentar
{ %// TERPEN(O) M4-Makro auskommentiert
   ...
   Me_ bond(#1) %// Kommentar
}

Der Chemiefont

Der Befehl \chemfont spezifi ziert Kommandos zur Auswahl des Zeichensatzes, der f¨ur den Satz von Textzeichen innerhalb der Formeln zum Einsatz kommen soll. Als Standard ist \sffamily vorgesehen, das heiß t alle Zeichen sind ebenso groß wie die Buchstaben des Grundtextes, aber serifenlos. Sie k¨onnen im Vorspann des Dokumentes einen anderen Font einstellen, zum Beispiel die Serifengrundschrift in kleinerem Schriftgrad:

\chemfont{\rmfamily\small}

Oder die aktuelle Grundschrift:

\chemfont{}

Ringstrukturmodule

Wie in Abschnitt 2.9 des Tutorial geschildert werden wird, k¨onnen Sie Bibliotheken mit eigenen Ringstrukturen aufbauen oder bereits vorhandene nutzen. Sie werden bei Bedarf dem L^ATEX-Dokument, das sie ben¨otigt, hinzugef¨ugt. Jede verwendete Bibliothek muß vor ihrer Verwendung durch einen Befehl in der chemspecial-Um-gebung geladen werden:

\begin{chemspecial}
  require("<Lib>")
\end{chemspecial}

Um die mitgelieferte Bibliothek bicyclib.pm mit bizyklischen Naturstoffringen einzusetzen, muß Ihr Dokument sinngem¨aß dem in Schema 1– 1 gezeigten entsprechen.

Neue Gleitumgebung

Das Paket enth¨alt eine weitere Umgebung schema, die ¨ahnlich der figure-Um-gebung gleitende Abbildungen mit eigenem Nummernkreis f¨ur Reaktionsschemata bereitstellt. Eine einfache Anwendung ist folgende:

\begin{schema}
  \begin{chemistry}[phenolsyn]
     ...
  \end{chemistry}
  \caption{Eine einfache Phenolsynthese.\label{phen}}
\end{schema}

Im zweispaltigen Layout k¨onnen Sie die Sternform (schema*-Umgebung) verwenden, um die Schemata ¨uber beide Spalten zu setzen.

8 Kapitel 1 — Einführung

\documentclass[...]{...}
\usepackage{ochem}

\begin{document}

\begin{chemspecial}
  require("bicyclib")
\end{chemspecial}

\begin{chemistry}[campher]
multiline(1,L)
{ formula(L,R,"Zyklopentadien")
  { ring(,,H1=3=,,5,0){} }

  formula(L,R){ atom("+") }

  formula(L,R,"Methylacrylat")
  { bond(-90,=U) bond(-30) atom("COOCH$_3$",L) }

  arrow(){}

  formula(L,R)
  { ring("bc221h",,5=){ 2: bond(-30) atom("COOCH$_3$",L); }
  };
}
\end{chemistry}

\end{document}

Zyklopentadien

+ COOCH_3

Methylacrylat

COOCH_3

Schema 1–1 Verwendung von Ringstrukturen, hier des Bizyklohexans, aus einer Bibliothek.

Formeln sind L^ATEX-Boxen!

Wie Sie aus der ¨Uberschrift entnehmen k¨onnen, bilden die chemistry-Umgebun-gen normale L^ATEX-Boxen, genauer picture-Umgebungen oder -boxen, sodaß Sie zur Positionierung der Formeln im Text alles anwenden k¨onnen, was auch mit L^ATEX- Abbildungen gestattet ist, zum Beispiel die Anordnung mehrerer vertikaler schmaler Reaktionsketten nebeneinander oder das Plazieren von Formeln mitten im Textfl uß oder in Abbildungen oder eigenen Zeilen.

Formeln als EPS-, PNG- ... Graphiken

Die bislang dargestellte Methode, eine Formel mithilfe der chemistry-Umgebung in ein L^ATEX-Dokument einzubinden, setzt voraus, daSSie stets die vom Formelcompiler erzeugten kleinen L^ATEX-Dateien bereit haben. An sich nicht schlimm, wird hierdurch jedoch eine Benutzung dieser Formeln nur in L^ATEX-Dokumenten m¨oglich. Sie k¨onnen jedoch auch durch eine Paketoption separate ein Dokument erstellen, das ausschließ lich chemistry-Umgebungen enth¨alt, die durch die Option automatisch auf eigenen Seiten, je eine Umgebung und damit Formel auf einer Seite, gesetzt werden. Da auch der Seitenstil ¨empty¨eingesetzt wird, erscheinen keine Seitenzahlen mehr.

Als Beispiel diene die Datei formeln.tex, die die folgenden drei Formeln enth¨alt:

\documentclass{report}
\usepackage{german}
\usepackage[separate]{ochem}
\begin{document}

\begin{chemistry}[indigo]
  formula(L,R,"Indigo",HR,24)
  {
    ring(,,H,,5,0){
      vertex(,2){};
      0: bond(r,=C,L) ring(,0,H,,5,r){
                       vertex(,2){};
                       1: atom("N") bond(r) atom("H");
                       4: bond(r,=C) atom("O");
                     };
      1: atom("N") bond(r) atom("H");
      4: bond(r,=C) atom("O");

10 Kapitel 1 — Einführung

    }
  }
\end{chemistry}

\begin{chemistry}[purpur]
  formula(L,R,"\shortstack{6,6’-Dibrom-indigo\\Purpur}",HR,24)
  {
    ring(,,H,,5,0){
      vertex(,2){ 3: bond(r) atom("Br"); };
      0: bond(r,=C,L) ring(,0,H,,5,r){
                       vertex(,2){ 3: bond(r) atom("Br"); };
                       1: atom("N") bond(r) atom("H");
                       4: bond(r,=C) atom("O");
                     };
      1: atom("N") bond(r) atom("H");
      4: bond(r,=C) atom("O");
    }
  }
\end{chemistry}

\begin{chemistry}[carmin]
  formula(L,R,"Carmins""aure",HR,24)
  { ring(,,H){
      vertex(,1,4){ 0: bond(r,,Ln) atom("C",C,R) atom("OOH",L);
                     1: bond(r) atom("O",C,L) atom("H",R);
                     3: bond(r) atom("C",C,R) atom("H$_3$",L);
                   };
      vertex(,4,1){ 0: bond(r) atom("O",C,R) atom("H",L);
                     3: bond(r) atom("O",C,R) atom("H",L);
                     4: bond(r) atom("CO(CHOH)$_4$CH$_3$",L);
                     5: bond(r) atom("O",C,R) atom("H",L);
                   };
      0: bond(r,=C) atom("O");
      3: bond(r,=C) atom("O");
    }
  }
\end{chemistry}

\end{document}

(Beachten Sie in diesem Beispiel, daß die einzelnen chemistry-Umgebungen jeweils Namen indigo, purpur und carmin erhalten haben!)

Dieses Dokument, das ausschließ lich noch Formeln enth¨alt, ist die Grundlage, um aus den einzelnen Formeln einzelne EPS-Dateien zu erstellen, die anschließ end in ein beliebiges anderes Graphikformat konvertiert werden k¨onnen, wie JPG oder PNG, um die Formeln in einer Webpr¨asentation einzusetzen.

Um zum Beispiel die Formel der Carmins¨aure (dritte Formel oder Formel auf der dritten Seite des Dokumentes) in das PNG-Format zu konvertieren, kann die Befehlsfolge

dvips -E -n 1 -p 3 -o carmin.eps formeln.dvi
convert carmin.eps carmin.png

eingesetzt werden. Die Option -E von dvips, um EPS-Dateien zu erzeugen, ist auf jedem Rechner vorhanden, das Programm convert¨ ı st Bestandteil der OpenSource-

1.4 — Ausgabeformate 11

Graphikbearbeitungsprogramme Image Magick, die vor allem unter Linux oft installiert sind oder f¨ur verschiedene Betriebssysteme leicht aus dem Internet geladen werden k¨onnen.

Um eine ganze Reihe von Formeln zu konvertieren, kann das mitgelieferte Perl- Script makePic.pl sehr hilfreich sein:

makePic.pl --n=3 --format=png --outfile=formeln formeln

Es erstellt mit diesem Aufruf die Dateien farben1.eps und farben1.png bis farben3.eps und farben3.png. Noch einfacher wird die Konversion dadurch, daß die Namen der einzelnen chemistry-Umgebungen in eine spezielle Datei formeln.names geschrieben werden. Diese Namen k¨onnen durch den einfachen Aufruf

makePic.pl --namefile --format=png formeln

als Namen der einzelnen Graphikdateien verwendet werden. F¨ur das gegebene Beispieldokument erhalten Sie die Dateien indigo.eps und indigo.png, purpur.eps und purpur.png sowie carmin.eps und carmin.png.

Die PNG-Graphik kann zur Internet-Publikation eingesetzt, die EPS-Graphik dagegen nach Belieben in ein L^ATEX-Dokument eingebunden werden:

\fbox{
  \includegraphics{formeln3.eps}
}

Beachten Sie bitte, daß das Paket Image Magick auf Ihrem Rechner installiert sein muß .

1.4 Ausgabeformate

Es ist leicht m¨oglich, anstelle der PostScript-Ausgabe andere Formate generieren zu lassen. S¨amtliche Ausgaben werden aus graphischen Primitiven zusammengesetzt, die vom Modul be.pm in konkrete Anweisungen ¨ubersetzt werden. Diese Primitive umfassen Linien, Kreise, Text, verschiedene Bindungen sowie Reaktionspfeile. Gelingt es Ihnen, nur f¨ur diese Grundelemente eine Umsetzung im gew¨unschten Format zu fi nden, k¨onnen sie k¨unftig alle, auch komplexe Formeln und Schemata in dieses Format compilieren lassen. Als Grundlage f¨ur solche Eigenentwicklungen k¨onnen die bereits vorhandenen Ausgabeformate dienen, die durch eine eindeutige Zahl unterschieden werden:

$be::BE PSLATEX : Texte werden in L^ATEXin einer picture-Umgebung gesetzt, die Graphikausgabe erfolgt in PostScript. Die Defi nition der verschiedenen Bindungsarten und Reaktionspfeile ist in ochem.sty als PostScript-Headerdatei enthalten. Die Dimensionen der Textboxen werden ermittelt, indem alle auftretenden Texte in eine tempor¨are L^ATEX- Datei geschrieben werden, die durch den Compiler ¨ubersetzt wird und f¨ur jeden Text die Gr¨oß e (Breite, H¨ohe, Tiefe) liefert. In diesem Modus k¨onnen Sie die besten Ergebnisse erwarten.

$be::BE PS : Dieses Format erzeugt ein reines PostScriptfi le im EPS-Format. Achtung, die Gr¨oß en von Text werden derzeit nicht ermittelt und stets auf Null gesetzt, was in falschen und seltsamen Verschiebungen der Texte resultiert. Dieses Format dient im Moment vor allem zum raschen Testen w¨ahrend der Entwicklung, ohne L^ATEX benutzen zu m¨ussen, da die reinen Formelbilder korrekt dargestellt sind. Wenn Sie keinen Text in der Formel haben, k¨onnen Sie so fertige EPS-Ausgabedateien erzeugen.

$be::BE LATEX : Dies ist eine reine Ausgabe mit graphischen L^ATEX-Primitiven wie Linien, Pfeilen und Kreisen. Entsprechend den sparsamen M¨oglichkeiten, werden nicht alle Bindungsformen, Winkel und Kreisradien unterst¨utzt.

Abh¨angig von den Ausgabeformaten, muß auch die Ermittlung der Textgr¨oß en auf unterschiedliche Weisen geschehen. Da dieser Vorgang unmittelbar nach dem Parsen des Programmcodes erfolgt, fi nden Sie die entsprechenden Zeilen nicht in be.pm, sondern in streambuf.pm. Auch in diesem Modul wird anhand des Formattyps entschieden, ob L^ATEX zur Ermittlung der Textgr¨oß e aufgerufen werden muß . F¨ur eine reine PostScript-Ausgabe existiert derzeit keine Gr¨oß enermittlung.

1.5 Makrodefinitionen

Beim Compilieren des von L^ATEX erzeugten .chm-Files kann durch den Schalter -pp eine Pr¨aprozessorbearbeitung veranlaß t werden, um Makros aufzul¨osen oder bedingte Bearbeitung zuzulassen. Hierf¨ur m¨ussen Sie in chemie.pl bei der Auswertung der Variablen $preproc innerhalb der Backticks den Aufruf Ihres bevorzugten Pr¨aprozessors einf¨ugen. Der Pr¨aprozessor muß den Inhalt der Eingabedatei, deren Name in $infile steht, in bearbeiteter Form in die Zwischendatei $tmpfile ¨ubertragen. Wenn Sie den Gnu M4-Pr¨aprozessor benutzen, m¨ussen Sie keine Anpassung vornehmen.

M¨ochten Sie keinen Pr¨aprozessor einsetzen, brauchen Sie nichts anzupassen und d¨urfen den Compiler nicht mit -pp aufrufen.

Der M4-Präprozessor

Hier soll keine Manpage zu M4 gegeben, sondern nur kurz dargestellt werden, wie ein Makro defi niert und eingesetzt wird. Jede Makrodefi nition muß vor ihrer Benutzung innerhalb einer chemspecial-Umgebung auftreten; sowohl Name wie auch Ersetzungstext m¨ussen dabei durch einfaches Backtick respektive Hochkomma geklammert sein. Die aktuellen Argumente sind ¨uber die Variablen $1, $2 ...verf¨ugbar. Mit der Sequenz

1.5 — Makrodefinitionen 13

\begin{chemspecial}
  define(‘C3’, ‘bond(30) atom("$1") bond(-30)’)
\end{chemspecial}

\begin{chemistry}
  formula()
  { C3(O) C3(NH) }
    % ^-- aquivalent zu
    % bond(30) atom("O") bond(-30) ...
    % ... bond(30) atom("NH") bond(-30)
\end{chemistry}

wird die Struktur des Methoxymethyl-methylamins erzeugt.

Sie k¨onnen auch Makros schreiben, die mit Hilfe von ifelse Entscheidungen treffen. Als Beispiel dient hier die Reihe einiger f¨unfgliedriger Heterozyklen (Thiophen, Furan und Pyrrol), in der die Chalkogenvertreter keinen Wasserstoff tragen:

\begin{chemspecial}
  define(‘CP’,‘ring("cpentane",,1=3=)
    {0: atom("$1");
     ifelse($1,‘N’,‘0: bond(-90) atom("H");’)
    }’)
\end{chemspecial}

\begin{chemistry}
  formula(L,R,"Furan",HA,24){ CP(O) }
  formula(L,R,"Pyrrol",HA,24){ CP(N) }
  formula(L,R,"Thiophen",HA,24){ CP(S) }
\end{chemistry}

Das Makro vergleicht seinen Parameter mit dem Symbol f¨ur Stickstoff und erzeugt, falls der Vergleich positiv ausf¨allt, eine weitere Bindung zum Wasserstoffatom.

Haben Sie eine Anzahl Makros geschrieben, die Sie immer wieder ben¨otigen, k¨onnen Sie diese in einem Includefi le zusammenfassen und bei Bedarf nachladen – eine Art von Bibliotheksdatei auf der Ebene des Pr¨aprozessors. Ausf¨uhrliche Beispiele von Makros fi nden Sie in den Abschnitten 3.6 und 3.7. Sie m¨ussen das gew¨unschte Includefi le mit der Folge

\begin{chemspecial}
include(‘natur.inc’)
\end{chemspecial}

in Ihr Dokument einbinden. include ist ein weiterer Pr¨aprozessorbefehl von M4, der die angegebene Datei l¨adt und die darin enthaltenen Makrodefi nitionen ausf¨uhrt.

Um maximalen Nutzen aus Makros zu ziehen, m¨ussen sie m¨oglichst fl exibel geschrieben werden. Beim Terpenger¨ust TERPEN lag zum Beispiel die ¨Uberlegung zugrunde, daß die nat¨urlichen Terpene dieses Typs meistens an den Kohlenstoffatomen C^1, C^2, C^3 und C^8 substituiert sind. Der seltener auftretende Substituent an C^2 wird als (eventuell leerer) Parameter ¨ubergeben. Die anderen Positionen werden gespeichert und k¨onnen mit restoreXY zum Startpunkt von Substituenten werden. Mit Hilfe einiger ifelse-Befehle wird das STEROID-Makro so gestaltet, daß es die ¨ahnlichen Ger¨uste wichtiger Steroidgrundk¨orper erzeugen kann, wobei die Beschreibung der vier Ringe gemeinsam verwendet wird.

Sie sollten beim Schreiben Ihrer Makros ¨ahnliche ¨Uberlegungen anstellen, um zu einer Anzahl von wichtigen Schl¨usselatomen zu gelangen, an denen Modifi kationen m¨oglich sind. Kleine ¨Anderungen k¨onnen als Parameter ¨ubergeben werden, sind komplizierte Formeln zu erwarten, werden diese Parameter leicht un¨ubersichtlich und es ist besser, im Makro solche Schl¨usselpositionen zu speichern und sp¨ater mit

restoreXY wieder anzuspringen. Variable Ger¨ustteile k¨onnen mit ifelse einge-

setzt werden und auf einer gemeinsamen Struktur aufbauen.

1.6 Ausblick

Ich habe in der vorliegenden Version die Ideen bearbeitet, die mir spontan zum Thema Formelsatz in den Sinn gekommen sind; folgen weitere, werden diese in sp¨ateren Versionen ebenfalls zur Verf¨ugung stehen. Da sich viele ¨Anderungen oder Erweiterungen erst beim intensiven Nutzen des Pakets herauskristallisieren, w¨are ich ¨uber Anregungen oder Kritik aus Nutzerkreisen interessiert.

Da jedem automatisch der Sourcecode zug¨angig ist, kann unter ” Nutzung“ auch die Weiterentwicklung verstanden werden *smile*! Zun¨achst auf den Satzmechanismus konzentriert, habe ich hier ein weites Feld zum Bestellen hinterlassen; einige Punkte, die verbessert werden m¨ussen, sind zum Beispiel:

• Verbesserte Ausgabe nur mit Mitteln von L^ATEX (Moduskennung $be::BE_LATEX).

• Ermittlung der Textgr¨oß e auch bei reiner PostScript-Ausgabe mit der Moduskennung

$be::BE_PS.

• Vielleicht kann auch das Programmdesign insgesamt und von Grund auf besser oder einfacher gestaltet werden.

Ge¨anderter Code sollte analog dem Vorgehen bei TEX/L^ATEX durch eine Namens¨anderung kenntlich gemacht werden, das heiß t es darf unter allen Umst¨anden nur eine einzige Version OCHEM Version 3-0e 2001-04-10 existieren! Damit nach konstruktiven ¨Anderungen alle in den Genuß der Bem¨uhungen und weiterentwickelter Versionen kommen k¨onnen, schlage ich vor, Verbesserungen in Form des Sourcecodes oder detaillierter Bemerkungen an mich zu mailen, damit sie in die derzeit aktuelle Version eingepfl egt werden k¨onnen. Meine EMail-Adresse ist

kloeckl@vkcmzd.chemie.uni-mainz.de
ingo.kloeckl@2k-software.de

2. Tutorial

Dieses Kapitel soll Ihnen gen¨ugend Material an die Hand geben, um erfolgreich mit dem Formelsatz beginnen zu k¨onnen. Es enth¨alt eine Anleitung f¨ur die ersten Schritte, zeigt aber auch versteckte Aspekte einzelner Befehle und besondere Einsatzm¨oglichkeiten sowie Beispiele aus der Praxis. Eine ausf¨uhrliche Beschreibung der Kommandos wird in der Befehlsreferenz in Kapitel 3 gegeben.

2.1 Grundlegende Elemente

2.1.1 Einfache Formeln

Bei der Beschreibung der Formeln wird ein geometrischer Ansatz verfolgt. Um eine gr¨oß tm¨ogliche Zahl an Strukturen beschreiben zu k¨onnen, existiert nur eine minimale Anzahl an Grundelementen wie die Bindung in beliebiger Richtung in verschiedener Erscheinung, das Atomsymbol, Verzweigungen sowie einfache (monozyklische) Ringger¨uste. Diese Elemente kombinieren Sie, indem Sie an einem Atom, zum Beispiel eines Ringes, eine Bindung in der gew¨unschten Richtung anf¨ugen, die zu einem bestimmten Ringatom eines weiteren Ringes f¨uhrt und so fort. M¨ochten Sie komplexere Ringe wie zum Beispiel Inden verwenden, liegt Ihnen kein Makro f¨ur Indexger¨uste vor, sondern Sie kombinieren einen Sechs- und einen F¨unfring. Der Vorteil liegt gegen¨uber einer Sammlung von Makros f¨ur jede Struktur klar auf der Hand: es kann eine gr¨oß ere Vielfalt von Strukturen dargestellt werden, obwohl im Compiler selber nur eine Handvoll implementiert werden muß . Auch k¨onnen wenige Befehle rascher erlernt werden als zahlreiche unterschiedliche.

Der Nachteil: bei manchen Strukturen ben¨otigen Sie mehr Grundelemente als bei Einsatz eines spezialisierten Makros. Da die Bindungen auch durch einen Winkel beschrieben werden, m¨ussen Sie mehr geometrisch als chemisch denken (aber die meisten Strukturen arbeiten ohnehin mit einer immer gleichen Menge von Winkeln). Einige symbolische Winkelangaben im Zusammenhang mit Ringsystemen helfen Ihnen aber, komplizierte Winkelberechnungen zu vermeiden.

Jede Formel wird mit dem Befehl formula gesetzt, dessen Parameter weiter unten besprochen werden. Im Rumpf des Befehls steht die geometrische Beschreibung der Struktur. Nachfolgend einige Beispiele f¨ur sehr einfache Strukturen, die jedoch die Grundz¨uge des Verfahrens deutlich machen. Die Formel besteht aus einer einfachen Folge von Bindungen, woraus eine aliphatische Verbindung resultiert:

formula(L,R)
{ bond(30) bond(-30) bond(30,=) bond(-30) }

F¨ur die Doppelbindung ist die Angabe eines optionalen Parameters erforderlich.

Einfachbindungen entsprechen dem Defaultwert und m¨ussen nicht gesondert notiert werden. Beachten Sie die stets wiederkehrenden Winkelangaben 30^o, 150^o sowie -30^o und -150^o f¨ur schr¨age, aber fl ach verlaufende Bindungen. Diese Winkelwerte harmonieren mit den Kanten von Sechsringen, die aufrecht auf einer Ecke stehen. Bevorzugen Sie auf Kanten liegende Sechsringe, werden Sie die Winkel 60^o, 120^o sowie -60^o und -120^o einsetzen. Dar¨uberhinaus ben¨otigen Sie bis auf wenige Ausnahmen nur noch die Hauptachsenwinkel 0^o, 90^o, 180^o und -90^o.

Da Ketten einzelner Bindungen sehr h¨aufi g auftreten, ist hierf¨ur eine Kurzform erlaubt, in der mehrere Bindungsbeschreibungen, durch Semikola getrennt, zusammengefaß t sind. Obige Formel l¨aß t sich auch schreiben als

formula(L,R){ bond(30;-30;30,=;-30) }

Der Hauptnachteil der vorgestellten Beschreibung ist, daß die konkreten Bindungswinkel es erschweren, eine identische Alkylkette in anderer Orientierung, zum Beispiel gedreht, darzustellen. Sie m¨ussten alle Winkel umrechnen. Eine gr¨oß ere Abstraktion wird durch symbolische Winkelangaben erreicht, die vor allem im Zusammenhang mit Ringsystemen von Bedeutung sind und weiter unten genauer vorgestellt werden. Der Beispiel-Aliphat kann damit wie folgt formuliert werden:

  formula(L,R){ bond(r; r/; r,=; r/) }

r ist ein spezieller Winkel, der normalerweise der radialen Bindung an Ringen an der

aktuellen Ringposition entspricht. Liegt kein Ring in der Beschreibung der Formel vor, wird ein Defaultwert von 30^o f¨ur r, von 90^o f¨ur t und von -90^o f¨ur b eingesetzt. Die (komprimierte) Beschreibung kann damit wie folgt gelesen werden: bilde eine (radiale) Bindung; bilde abknickende Bindung; bilde Bindung in der urspr¨unglichen (radialen) Richtung als Doppelbindung; bilde wieder eine abknickende Bindung. Eine solche Beschreibung ist unabh¨angig von konkreten Winkeln und kann somit an verschiedenen Stellen der Formel eingesetzt werden. ring setzt selbst¨andig die Winkel auf sinnvolle Werte, bei reinen Alkylketten m¨ussen Sie in seltenen F¨allen mit

set die Winkel manuell setzen (Beispiele in der Referenz zu bond). Mit konkreten

Winkeln k¨onnen Sie jederzeit, auch bei intensiver Verwendung der symbolischen Angaben, die gew¨unschte Richtung erzwingen.

Die zweite und alle weiteren Formeln zeigen, wie Atomsymbole in das Ger¨ust aus Bindungen integriert werden. Beachten Sie die Positionierungen der einzelnen Atomgruppen: einzelne Atome an terminalen Positionen werden mit der Voreinstellung

C,C von atom zumeist zentriert zum Bindungsende gesetzt, w¨ahrend l¨angere Tex-

te b¨undig mit einer seitlichen Kante anschließ en (ein solcher Kantenabschluß w¨are aus Gr¨unden der Einheitlichkeit auch f¨ur einzelne Buchstaben wie ” O“ anzuraten, ist mir aber meistens zu umst¨andlich :-( ).

HOOC O CH_2CH_2 COOH formula(L,R)
{ atom("HOOC",L,R) bond(30;-30) atom("O")
  bond(30;-30) atom("CH$_2$CH$_2$",L,R)
  bond(30;-30) atom("COOH",L,R)
}

Ein besonderes Problem tritt auf, wenn Bindungen von 90^o oder -90^o in langen Texten enden, zum Beispiel in der Carboxyl- oder Hydroxylgruppe. Um diese lange Gruppe nicht seitlich, sondern (sch¨oner) zentriert zum C anzusetzen, wird der Text in einen zentrierbaren (C) und einen seitlich ansetzbaren (OOH) Teil aufgespalten

(Formel links). In der Formel rechts k¨onnen Sie erkennen, welches Aussehen die COOH-Gruppe bes¨aß e, w¨are sie einfach in einem St¨uck mit L positioniert worden (Bindung oben). Eine Zentrierung des Textes h¨atte katastrophale Folgen (COOH- Gruppe unten).

COOH

C HOO

COOH

COOH formula(L,R)
{ ring()
  { 0: bond(r) atom("C",C,R) atom("OOH",L);
    3: bond(r) atom("C",C,L) atom("HOO",R);
  }
}
formula(L,R)
{ ring()
  { 0: bond(r) atom("COOH",C);
    3: bond(r) atom("COOH",L); }
}

Das Problem tritt nicht in Erscheinung, wenn die Bindungen schr¨ag auf einen langen Textabschnitt treffen und seitlich zum gesamten Text positioniert werden, wie Sie bereits an der Etherkette mit den zwei Carboxylgruppen gesehen haben.

Ringstrukturen werden mit dem Befehl ring erzeugt. In seiner einfachsten Form erhalten Sie einen Benzenring, mit verschiedenen Parametern k¨onnen Sie jedoch auch n-gliedrige Ringe sowie andere wichtige Ger¨uste (Campher) darstellen. Mit Ringen begegnen Ihnen erstmal Elemente, von denen mehrere Substituenten abzweigen k¨onnen. Diese Substituenten werden wie normale Formeln beschrieben, m¨ussen jedoch jeweils durch ein ; angeschlossen werden. Das Ringatom, von dem aus der Substituent abzweigt, wird durch eine Nummer, bei Null beginnend, bezeichnet. Um nicht f¨ur jede Ringposition den genauen Winkel, unter dem die Bindungen abzweigen, lernen zu m¨ussen, k¨onnen Sie bei bond eine symbolische Winkelangabe r f¨ur radiale Bindungen angeben (eine ausf¨uhrlichere Diskussion symbolischer Winkel fi nden Sie im n¨achsten Abschnitt):

COOH formula(L,R)

{ ring()
  { 0: bond(r);
    3: bond(r) atom("C",C,R) atom("OOH",L);
    4: bond(r; r/; r); }
}

Eine weitere Form der Verzweigung kann bei Ketten auftreten. branch h¨alt die aktuelle Position in anonymer Form fest und erlaubt es, Substituenten anzugeben, die mit ; abgeschlossen werden und an diesem Punkt beginnen:

HO CH_2 CH_2 OH O C formula(L,R)
{ atom("HO",L,R) bond(0)
  atom("CH$_2$",L,R) bond(0)
  atom("CH$_2$",L,R) bond(0)
  branch { bond(-45) atom("OH", L);
           bond(45,=) atom("O",L);
           atom("C");
         }
}

Da dieser Punkt nach Abarbeitung des Befehlsrumpfes noch erhalten bleibt, kann von ihm aus die Kette weitergef¨uhrt werden. Sie k¨onnen daher eine Hauptkette formulieren und die von dieser abzweigenden Seiten¨aste mit branch beschreiben:

OH formula(L,R)
{ bond(30)
  branch { bond(120,t);
           bond(60,o) atom("O",C,R) atom("H",L);
         }
  bond(-30; 30)
}

Die Befehle saveXY und restoreXY bieten die M¨oglichkeit, die aktuelle Position unter einer Nummer zu speichern und an beliebiger Stelle darauf zuzugreifen. Die rekursive Schachtelung, die bei Verzweigungen mit branch auftritt, kann so in einen linearen Ablauf umgewandelt werden. Besonders bei tief geschachtelten Verzweigungen kann diese Form vorteilhaft sein.

HO CH_2 CH_2 C OH O formula(L,R)
{ atom("HO",L,R) bond(0)
  atom("CH$_2$",L,R) bond(0)
  atom("CH$_2$",L,R) bond(0)
  saveXY(#1) atom("C")
  restoreXY(#1) bond(-45) atom("OH", L)
  restoreXY(#1) bond(45,=) atom("O",L)
}

2.1.2 Formeln anordnen

Haben Sie erst einmal eine Formelbeschreibung erstellt, kann diese innerhalb des

formula-Befehls als Einheit betrachtet und automatisch plaziert werden. An dieser

Stelle kommen die Parameter des Befehls ins Spiel: sie geben an, wie die Formel relativ zu einem Startpunkt plaziert werden soll (zentriert, linksb¨undig, mit einer Ecke oder Kante) und an welcher Stelle der Formel der Anschluß punkt f¨ur nachfolgende Formeln liegt (ebenfalls zentriert, eine Kante oder Ecke). Dieser Anschluß punkt wird zum Startpunkt f¨ur folgende Formeln. Daraus folgt, daß Sie f¨ur beide Parameter beliebige Werte, zum Beispiel C,C, einsetzen k¨onnen, wenn es Ihnen nur darum geht, ein Formelbild in den Text einzuf¨ugen, da beide Parameter erst im Zusammenspiel mehrerer Formeln ihren Sinn erhalten.

Betrachten Sie als Beispiel das Schema 2– 1, in dem in der linken H¨alfte Formeln Seite an Seite erscheinen, auf eine gemeinsame vertikale Mitte ausgerichtet mit dem Positionierungsparameter L,R. Diesselben Formeln mit T,B gesetzt, reihen sich rechts an einer gemeinsamen Mittelkante, aber vertikal auf.

Die beiden Varianten, Formeln aneinanderzuf¨ugen, d¨urften die meistgenutzten sein, da mit ihnen Reaktionsketten aufgebaut werden k¨onnen. (In einem sp¨ateren Abschnitt werden Sie noch lernen, wie neben dem bislang linearen Ablauf Verzweigungen hinzuf¨ugen.) Es sind jedoch zwei wichtige Sonderanordnungen zu erw¨ahnen:

• Eine regelm¨aß ige tabellarische Anordnung kann mit den bisherigen Mitteln nicht erreicht werden, da die Position jeder folgenden Formel von der Gr¨oß e der vorhergehenden abh¨angig ist. Der Befehl →gotoXY setzt den Startpunkt einer Formel

\begin{chemistry}[intro2a]
  formula(L,R){ bond(30; -30; 30,=; -30) }

  formula(L,R)
  { atom("HOOC",L,R) bond(30; -30; 30) atom("COOH", L) }

  formula(L,R)
  { ring()
    { 0: bond(r);
      4: bond(r; r/; r); }
  }
\end{chemistry}
\hfil
\begin{chemistry}[intro2b]
  formula(T,B){ bond(30; -30; 30,=; -30) }

  formula(T,B)
  { atom("HOOC",L,R) bond(30; -30; 30) atom("COOH", L) }

  formula(T,B)
  { ring()
    { 0: bond(r);
      4: bond(r; r/; r); }
  }
\end{chemistry}

HOOC

COOH

HOOC

COOH

Schema 2–1 Die Positionierungsparameter vermitteln den Zusammenhang zwischen den einzelnen Formeln. In der linken H¨alfte von links nach rechts angeordnete Formeln, in der rechten H¨alfte ¨ubereinandergestellte Formeln (von oben nach unten).

unabh¨angig vom Kontext auf einen bestimmten Punkt und bietet somit die Grundlage einer regelm¨aß igen Anordnung. Ein Beispiel fi nden Sie in der Befehlsreferenz zu gotoXY.

• Eine freie Anordnung von Formeln kann sinnvoll sein, wenn zwei oder mehr Reaktanden zu einem Produkt f¨uhren und die Reaktanden nicht hintereinander, durch Pluszeichen verbunden, Schlange vor ihrem Reaktionspfeil stehen sollen, sondern

” wolkenartig“ verteilt sein sollen. Dies ist mit dem Befehl →shiftXY realisierbar, der den aktuellen Punkt um einen bestimmten Betrag verschiebt. Als Bezugspunkt wird dann in der Regel der Startpunkt der ersten Formel gew¨ahlt, alle anderen beziehen sich dann auf diese. Ein Beispiel ist im Tutorial in Abschnitt 2.6 gegeben.

2.2 Komplexere Formeln

Dieser Abschnitt stellt komplexere Formeln sowie die zu ihrer Beschreibung n¨otigen Mittel vor. Einige der verwendeten Tricks werden in der alphabetischen Referenz erw¨ahnt, andere nicht, in jedem Falle k¨onnen Sie in diesem Abschnitt eine ausf¨uhrlichere Diskussion wichtiger Darstellungsmittel fi nden, aus welchem Grunde dieser Abschnitt die Lekt¨ure der Referenz nicht ersetzen mag. Haben Sie eine Formel zu setzen, zu deren Beschreibung Ihnen anscheinend die Mittel fehlen, schauen Sie ¨uber die hier gezeigten Strukturen, vielleicht erhalten Sie die richtigen Anregungen. Manchmal geht’s auch einfach mit Pfusch und wilden Linien:-)

2.2.1 Symbolische Winkel

Der ring-Befehl liefert ¨uber die symbolische Angabe r hinaus f¨ur jedes Ringatom drei Kardinalwinkel, die f¨ur radiale Bindungen und Bindungen parallel zu den beiden angrenzenden Kanten verwendet werden. Diese werden von bond ausgewertet, wenn Sie einen der ” Winkel“ r, t oder b angeben. Sie k¨onnen so Bindungen erzeugen, ohne sich um den konkreten Winkel, unter dem sie vom Ring abzweigen, Gedanken zu machen. Die n¨achste Verbindung zeigt die Verwendung aller drei Symbole:

top bottom formula(L,R)
{ ring(,,H,,5,90)
  { 0: bond(r);
    2: bond(t) atom("top",L);
    2: bond(b) atom("bottom",L); }
}

formula(L,R)
{ ring(,,H1=)
  { 0: bond(t,t); 0: bond(b,o);
    3: bond(r);
    4: bond(r); }
}

Bei den Ringger¨usten mit r¨aumlicher Intention (chair, bc222o und so weiter) wird die Bedeutung der t- und b-Winkel als ” top“ und ” bottom“ deutlich, Beispiel fi nden Sie bei der Vorstellung der Bibliotheksmodule und des →ring-Befehls.

M¨undet eine Bindung mit symbolischem Winkel in einem weiteren Ring, kennen Sie den Winkel nicht, um den der zweite Ring gedreht werden muß , damit die Bindung auch bez¨uglich dieses zweiten Ringes in einer der drei Relationen r, t oder b steht. Dies ist auch nicht erforderlich, da anstelle des Drehwinkels des zweiten Ringes ebenfalls eine dieser symbolischen Angaben stehen kann. Sie m¨ussen nur die Nummer des Ringatoms angeben, an dem die Bindung endet:

4 formula(L,R)
{ ring(){4: bond(r)
            ring(,4,H,,5,r){ 0: bond(r); };
        }
}

formula(L,R)
{ ring(){ 4: bond(r)
             ring(,0,H,,5,r){ 0: bond(t,t); }; }
}
formula(L,R)
{ ring(){ 4: bond(r,o)
             ring(,0,H,,5,b){ 0: bond(t,t); }; }
}

Das Acoran zeigt, daß auch Spiroverbindungen gezeichnet werden k¨onnen, bei denen die beiden Ringe nicht durch eine Bindung verbunden sind. Der symbolische Winkel r ist ein f¨ur jede Ringposition spezifi scher Default f¨ur Drehwinkel:

Acoran formula(L,R,"Acoran",HR,24)
{ ring(,,H)
  {1: bond(r);
   4: ring(,0,H,,5,r)
      { 1: bond(r);
        4: bond(r) branch { bond(r+);
                            bond(r-); };
      };
  }
}

Ein nachgestelltes + oder - vergr¨oß ert beziehungsweise verkleinern den Winkel um 60^o. Aus der Sicht der zu zeichnenden Bindung entspricht dies einer Drehung um 60^o nach links beziehungsweise rechts. Sie k¨onnen so ohne Kenntnis des genauen Winkels verzweigte Alkyl-, zum Beispiel Isopropylsubstituenten beschreiben:

formula(L,R)
{ ring(,,H1=)
  { 3: bond(r) branch { bond(r+); bond(r-); };
    1: bond(r) branch { bond(r+); bond(r-); }; }
}

Weist eine der Ketten der Verzweigung mehr als eine Methyleneinheit auf, ist es geschickter, die Verzweigung mit rt oder r/t zu formulieren, siehe unten.

22 Kapitel 2 — Tutorial

Mit den +- und --Angaben k¨onnen auch l¨angere Alkylketten ohne Wissen um den genauen Winkel formuliert werden. Da die Drehungen relativ zum augenblicklichen Drehwinkel erfolgen, k¨onnen Sie alle identischen Alkylsubstituenten, an welcher Ringpositionen sie auch auftreten, durch Angabe ein und derselben Beschreibung setzen. Beachten Sie, daß bei diesem Verfahren alle Ketten durch Drehung aus derselben Grundform hervorgehen und die Ketten so in Abh¨angigkeit ihres Startatoms unterschiedlich orientiert sind:

formula(T,B)
{ ring(,,H1=,,7,90)
  { 1: bond(r; r-; r; r-; r);
    2: bond(r; r-; r; r-) ;
    5: bond(r; r-; r; r-) ; }
}

formula(T,B)
{ ring(,,H1=,,6,90)
  { 1: bond(r; r-; r; r-; r);
    2: bond(r; r-; r; r-) ;
    5: bond(r; r-; r; r-) ; }
}

formula(L,R)
{ ring(,,H,,5,90)
  { vertex(,1,4,H,6)
    { 0: bond(r);
      1: bond(r; r-; r,=U; r+; r);
    };
    vertex(,3,1,H,7)
    { 0: bond(r);
      4: bond(r; r-; r,=U; r+; r);
      5: bond(r; r-; r,=U; r+; r);
    };
  }
}

Einige dieser Ketten werden ¨ublicherweise horizontal orientiert dargestellt. W¨unschen Sie eine solche Anordnung, k¨onnen Sie den symbolischen Winkel r/ anstelle der festen Rechtsdrehung r- benutzen. Diese Angabe legt keine bestimmte Drehrichtung fest, sondern bestimmt diese anhand der augenblicklichen Bindungsrichtung: weist die Bindung nach oben rechts (zum Beispiel 30^o), wird die n¨achste Bindung nach unten rechts (-30^o) gezogen. Auf Bindungen nach unten links folgen solche nach links oben und so fort. Diese Anordnung f¨allt am angenehmsten bei sechsgliedrigen Ringen auf:

formula(T,B)
{ ring(,,H1=,,7,90)
  { 1: bond(r; r/; r; r/; r) ;
    2: bond(r; r/; r; r/) ;
    5: bond(r; r/; r; r/) ; }
}
formula(T,B)
{ ring(,,H1=,,6,90)
  { 1: bond(r; r/; r; r/; r) ;
    2: bond(r; r/; r; r/) ;
    5: bond(r; r/; r; r/) ; }
}

Wiederholte Anwendung der +- und --Symbole erh¨oht oder vermindert jeweils um 60^o, sodaß auch geschachtelte Alkylketten ohne Kenntnis der genauen Winkel der aktuellen Ringposition gesetzt werden k¨onnen:

formula(L,R)
{ ring(,,H,,5,90)
  { 1: bond(r)
       branch { bond(r+); }
       bond(r-,=)
       branch { bond(r--; r---; r--); }
       bond(r) bond(r-); }
}

Das folgende Beispiel diskutiert eine M¨oglichkeit der Kettenverzweigungen genauer. Durch die Festlegung einer Linksdrehung f¨ur die Methylverzweigung (r+) resultieren auf der ” linken“ Seite des Ringes andere Ketten als auf der rechten, was die Orientierung der Methylgruppen angeht. Die Ketten entstehen durch strenge Rotation:

formula(L,R)
{ ring()
  { 1: bond(r) branch { bond(r+); } bond(r-;r);
    2: bond(r) branch { bond(r+); } bond(r-;r);
    4: bond(r) branch { bond(r+); } bond(r-;r);
    5: bond(r) branch { bond(r+); } bond(r-;r); }
}

Die symbolischen Winkel rt und r/t versuchen, eine gef¨alligere, das heiß t symmetrischere und an den Koordinatenachsen orientierte Anordnung solcher Querbindungen zu erzielen, indem Links- oder Rechtsdrehungen zugelassen und auch automatisch bestimmt werden:

formula(L,R)
{ ring()
  { 1: bond(r) branch { bond(rt,,s); }
       bond(r/) branch { bond(r/t,,s); } bond(r);
    2: bond(r) branch { bond(rt,,s); }
       bond(r/) branch { bond(r/t,,s); } bond(r);
    4: bond(r) branch { bond(rt,,s); }
       bond(r/) branch { bond(r/t,,s); } bond(r);
    5: bond(r) branch { bond(rt,,s); }
       bond(r/) branch { bond(r/t,,s); } bond(r); }
}

24 Kapitel 2 — Tutorial

Zur Darstellung der Stereochemie an Doppelbindungen muß zu den bislang genannten noch der symbolische Winkel r\ hinzutreten, mit dem eine Bindung entgegen dem Knickwinkel r/ gezeichnet wird. Die Auswirkungen sowie die korrekte Formulierung von winkelunabh¨angigen E/Z-Isomeren ist in der folgenden Formel gezeigt:

(E)-Isomer

(Z)-Isomer formula(T,B,"(E)-Isomer",HR,24)
{ ring()
  { 1: bond(r; r/; r,=; r/; r; r/);
    2: bond(r; r/; r,=; r/; r; r/);
    4: bond(r; r/; r,=; r/; r; r/);
    5: bond(r; r/; r,=; r/; r; r/); }
}
formula(T,B,"(Z)-Isomer",HR,24)
{ ring()
  { 1: bond(r; r/; r,=; r\; r; r/);
    2: bond(r; r/; r,=; r\; r; r/);
    4: bond(r; r/; r,=; r\; r; r/);
    5: bond(r; r/; r,=; r\; r; r/); }
}

2.2.2 Lineare Systeme

Viele Formelbilder lassen sich als lineare Abfolge von einfachen Bindungen oder zumindest identischen oder ¨ahnlichen Einheiten auffassen und daher linear beschreiben.

Der durchaus komplizierte Chlorinring kann trotz seines Ringcharakter als lineares System aufgefaß t werden, das aus vier Pyrrol-/Methyleneinheiten besteht. Jede diese Methylenbr¨ucken endet am Pyrrolring der n¨achsten Einheit. Der Ring kann somit durch eingeschachtelte nahezu identische Beschreibungen der Grundeinheit beschrieben werden. Der Ringschluß ergibt sich durch korrekte Wahl der Winkel aus Symmetriegr¨unden von selber. (Da die Anschluß punkte einer Struktureinheit stets von der vorhergehenden abh¨angig sind, erlangt man hier keinen Gewinn durch den Einsatz einer Linearisierung mit saveXY und restoreXY. Dieser w¨are nur gegeben, wenn die Startpunkte der Partialformeln allesamt zu Beginn verf¨ugbar w¨aren.)

N H N N H N

Chlorin formula(L,R,"Chlorin",HR,24)
{ ring(,,H1=3=,,5,-45)
  { 0: atom("N") bond(-45) atom("H");
    4: bond(30) bond(-30,=)
       ring(,1,H2=4=,,5,-135)
       { 0: atom("N");
          4: bond(-60) bond(-120,=)
             ring(,1,H2=,,5,135)
             { 0: atom("N") bond(135) atom("H");
               4: bond(-150,=) bond(150)
                  ring(,1,H0=,,5,45)
                  { 0: atom("N");
                    4: bond(120,=) bond(60);
                  };
             };
       };
  }
}

2.2.3 Linear anellierte Ringe

Der einfachste Fall anellierter Ringsysteme liegt dann vor, wenn alle Ringe ein gemeinsames Atom aufweisen. Sie k¨onnen sich zunutze machen, daß ring den aktuellen Startpunkt nicht verschiebt und so, wenn der aktuelle Punkt genau diesem gemeinsamen Atom entspricht, alle Ringe in einem Rutsch dargestellt werden. Alle Ringe liegen dann in derselben Schachtelungsebene, was die Formel m¨oglicherweise ¨ubersichtlicher gestaltet als die Angabe von Ringen als Substituenten weiterer Ringe. Damit beide Ringe die gleiche Kantenl¨ange haben und somit nahtlos zusammenpassen, muß ihre L¨ange mit #N (auf Wunsch auf #L oder andere L¨angen, wesentlich ist das #-Zeichen) gleichgesetzt:

Guajan formula(L,R,"Guajan",HR,24)
{ ring(,4,H,#N,7,0){ 1: bond(r) branch { bond(r+); bond(r-); };
                     5: bond(r);}
  ring(,2,H,#N,5,180){4: bond(r); }
}

H

H

β-Cadinen formula(L,R,"$\beta$-Cadinen",HR,24)
{ ring(,5,H1=){ 1: bond(r); }
  ring(,1,H3=){ 0: bond(r,t) branch{ bond(r+);
                                     bond(r-);};
                1: bond(t,t) atom("H");
                2: bond(b,o) atom("H");
                3: bond(r);
              }
}

Haben nicht alle Ringe ein gemeinsames Atom, m¨ussen Sie einen Basisring ausw¨ahlen, der die weiteren Ringe als Substituenten tr¨agt. Um die ¨Ubersicht zu bewahren, kann es sinnvoll sein, den am kompliziertesten substituierten Ring als Basis zu w¨ahlen, um einfachere Substitution an den Ringen zu erhalten, die selbst Substituent dargestellen. Im Falle des Hirsutans wird der zentrale F¨unfring als Basis gew¨ahlt, da auf diese Weise zwei einfache Ringe als Substituenten auftreten. W¨are einer der endst¨andigen Ringe gew¨ahlt worden, w¨are der zentrale F¨unfring dessen Substituent, der wiederum einen Ring als Substituent tr¨uge, was die Formel unn¨otig komplizieren w¨urde. In diesen und den folgenden Formeln ist der Basisring durch gr¨oß ere Strichst¨arke hervorgehoben:

Hirsutan formula(L,R,"Hirsutan",HR,24)
{ ring(,,H,,5,90)
  { vertex(,1,2,H,5){ 0: bond(t); 0: bond(b); };
    vertex(,3,3,H,5){ 0: bond(r); };
    3: bond(t); }
}

Die vertex-Syntax ¨ubernimmt die ansonsten komplizierten Berechnungen der korrekten Startpunkte und Drehwinkel aller anelierter Ringe. Ein weiteres Beispiel zeigt,

26 Kapitel 2 — Tutorial

wie einfach zusammengesetzte Ringsysteme beschreibbar sind. Die beiden gezeigten Varianten demonstrieren, daß die Wahl des Basisringes auch die Gesamtdrehung der Formel beeinfl uß t: im ersten Falle ist das nicht gedrehte Zyklooktan die Basis, der F¨unfring erscheint an der schr¨ag verlaufenden [a]-Kante und verleiht dem Gesamtmolek¨ul so einen gedrehten Eindruck. W¨ahlen Sie, wie im zweiten Falle, dagegen den F¨unfring als Basis, entsteht der Eindruck einer horizontal liegenden Formel. Beachten Sie die variable Kantenl¨ange der Ringe. Sie k¨onnen jederzeit mit # eine bestimmte Kantenl¨ange f¨ur einen Ring festlegen.

formula(L,R)
{ ring(,,H,,8,0)
  { vertex(,0,1,H0=,5){ 0: bond(r); };
    0: bond(t);
  }
}

formula(L,R)
{ ring(,,H1=,,5,0)
  { vertex(,2,0,H,8){ 0: bond(t); };
    1: bond(r);
  }
}

Mehrere Substituenten bauen komplizierte Systeme auf, wobei die ankondensierten Ringe ihrerseits weitere Ringe tragen k¨onnen, sodaß Systeme entstehen, in denen nicht alle Ringe direkt an einer Kante des Basisringes liegen. Beides wird durch die folgenden Beispiele illustriert:

formula(L,R)
{ ring(,,H,,5,90)
  { vertex(,1,0,,6){ };
    vertex(,3,0,H2=4=,6){ };
    vertex(,4,0,H,6){ };
  }
}

formula(L,R)
{ ring(,,H,,8,0)
  { vertex(,1,1,,6)
    { 0: bond(r);
      vertex(,3,0,H2=4=,7){};
    };
    0: bond(r);
  }
}

H H_3CNH

NH_2

Buxenin-G formula(L,R,"Buxenin-G",HR,24)
{ ring(,,H4=,,7,0)
  { 0: bond(b,t) atom("H");
    vertex(,3,1,H,6){ 3: bond(t); 3: bond(b);
                      4: bond(r,t) atom("H$_3$CNH",R); };
    vertex(,6,4,H5=,6){ 2: bond(t,t);
                        3: bond(-90,o);
                        vertex(,2,2,H,5)
                        { 4: bond(r)
                             branch { bond(r+) atom("N",C,R)
                                               atom("H$_2$",L); }
                             bond(r-);
                        };
                      };
  }
}

Nicht immer muß die beiden Ringen gemeinsame Bindung sichtbar sein. Das Humulen besteht aus einem Neunring, kondensiert mit einem Vierring. Der Eindruck der quadratischen Aussparung wird einfach durch je eine unsichtbare Bindung in jedem der Ringe erreicht:

Humulen formula(L,R,"Humulen",HR,24)
{ ring(,,H4s,,9,0)
  { 3: bond(r);
    vertex(,4,3,H3s,4,-45)
       { 2: bond(100,<.); 2: bond(170,<<); };
    8: bond(45);
  }
}

2.2.4 Manuelle Erstellung anellierter Systeme

Sie haben im letzten Abschnitt gesehen, wie mit vertex mehrkernige Systeme aus streng polygonalen Ringen aufbaut werden k¨onnen. Diese Syntaxerweiterung ist jedoch nur f¨ur den Typ ring, also n-Ecke, defi niert und kann daher nicht mit anderen Typen wie cpentane verwendet werden. M¨ussen Ringe aus solchen anderen Typen aufgebaut werden, m¨ussen Sie die Berechnungen der korrekten Drehwinkel selber vornehmen. Es folgt eine kleine Anleitung, in der eine allgemeine Formel f¨ur den Drehwinkel entwickelt wird, die auf Polygonen basiert, aber auf andere Typen ¨ubertragbar ist, wenn Sie deren typische Winkel kennen. Die Basispolygone sind dabei nicht gedreht, <p2> besitzt den Wert Null.

Den einzelnen Kanten sind gem¨aß Tabelle 2.1 Winkel θ_i zugeordnet, die in die allgemeine Formel einzusetzen sind. In der folgenden allgemeinen Formel bedeuten ϕ_{soll} den gew¨unschten Drehwinkel des Grundk¨orpers, ϕ den korrekten Drehwinkel des Ringes, der als Substituent dargestellt wird, θ_1 und θ_2 die Kantenwinkel der zusammentreffenden Kanten von Basis- und Substituentenring:

ϕ = ϕ_{soll} + (θ_1 − θ_2) + 180

Die folgende Verbindung basiert auf dem f¨unfgliedrigen Pyrrolidinring und wird zur Verdeutlichung mit den Drehwinkeln 0^o, 45^o und 90^o dargestellt. Die Nullpunk-

Tabelle 2.1 Zuordnung des Kantenwinkels zu den einzelnen Kanten f¨ur verschiedene Polygone (Zyklopentan, n = 5 bis Zyklooktan, n = 8). Die Kantennummer ist gleichzeitig die Nummer eines Substituenten.

Kante Kantennummer n = 5 n = 6 n = 7 n = 8

a 0 54 60 64 67 b 1 -18 0 13 23 c 2 -90 -60 -38 -22 d 3 -162 -120 -90 -67 e 4 -234 -180 -141 -112 f 5 -240 -193 -157 g 6 -244 -202 h 7 -247

te der einzelnen Furanringe werden vom Sauerstoffatom eingenommen, der Nullpunkt des Pyrrolidins vom Stickstoff.

Konstruieren wir zun¨achst den Grundk¨orper ohne Drehung des Pyrrolidinkerns. Durch folgende Kantenkorrespondenzen ergeben sich die angegebenen Winkel:

Pyrrolidin Furan Winkel Kante Substituent i Kante Startatom j

b 1 e 4 + 1 = 5 0 + −18 − −180 + 180 = −18 d 3 b 1 + 1 = 2 0 + −162 − 0 + 180 = 18 e 4 e 4 + 1 = 5 0 + −234 − −180 + 180 = −234

Durch die Wahl der Ringkanten, die ¨ubereinstimmen sollen, ist die Lage des Nullpunktes des gedrehten Ringes festgelegt und ebenfalls in der Tabelle angegeben. Es gilt dabei, daß die Substituentenposition i des Basisringes die Nummer der Kante ist (Tabelle 2.1). Das Startatom j, das in der Beschreibungsliste des Substituentenringes eingesetzt werden muß , ist um Eins h¨oher als die Kantennummer.

O

OO N formula(L,R)
{ ring(,,H,#N,5,0)
  { 1: ring(,5,H,#N,6,-18){ 0: atom("O");};
    3: ring(,2,H,#N,6,18){ 0: atom("O");};
    4: ring(,5,H,#N,6,-234){ 0: atom("O");};
    0: atom("N"); }
}

Bei den gedrehten Systemen muß nun einfach der Drehwinkel ϕ_{soll} zu allen Winkeln addiert werden.

O OO N formula(L,R)
{ ring(,,H,#N,5,90)
  { 1: ring(,5,H,#N,6,72){ 0: atom("O");};
    3: ring(,2,H,#N,6,108){ 0: atom("O");};
    4: ring(,5,H,#N,6,-144){ 0: atom("O");};
    0: atom("N"); }
}

Und schließ lich noch ein System aus F¨unf-, Sieben- und Achtringen:

formula(L,R)
{ ring(,,H,#N,7,0)
  { 4: ring(,3,H,#N,5,129){ 0: bond(r);}; % 0 + -141 - -90 + 180
    1: ring(,6,H,#N,8,350){ 0: bond(r,<<);}; % 0 + 13 - -157 + 180
    0: bond(r); }
}

Die typischen Winkel beim Ringtyp cpentane entsprechen denen des Sechsecks, wobei Sie die fortfallende Ecke ber¨ucksichtigen m¨ussen.

2.2.5 Angular anellierte Ringe

Ringsysteme, in denen die Ringe mehr als eine gemeinsame Bindung haben, k¨onnen mitunter schwierig zu beschreiben sein, wenn Sie sie aus Polygonen aufbauen, da die Bindungen zum Beispiel eines F¨unfringen andere Winkel aufweisen als die der Sechs- oder Siebenringe. Bei kondensierten Sechsringen tritt dieses Problem nicht auf; f¨ur den h¨aufi gen Fall des Zusammentreffens von Sechs- und F¨unfringen kann zur Darstellung der F¨unfringe der Typ "cpentane" gew¨ahlt werden, dessen Winkel nahtlos zu denen der Sechsringe passen.

HO O H

O

(+)-Hibiscon A formula(L,R,"($+$)-Hibiscon A",HR,24)
{ ring(,4,H){ 1: bond(r,o) atom("HO",R);
              2: bond(r,=C) atom("O");
              5: bond(b,o) atom("H"); }
  ring(,2,H){ 0: bond(r,o) branch{ bond(r+); bond(r-);}; }
  ring("cpentane",,0=3=){ 2: atom("O");}
}

Die Formel des Crassanins zeigt, daß auch das Ger¨ust des Bizyklo[2.2.2]oktans mit einem Sechsring kombinieren werden kann, da die Bindungsl¨angen im Bizyklus denen der sonstigen Bindungen entspricht. Die Kantenl¨angen des F¨unfrings muß mit der Gr¨oß enangabe #N etwas verringert werden, damit der Benzenring nicht erheblich gr¨oß er als der Zyklohexanring wirkt:

O N H

O H_3C O H_3C

COOCH_3 H N

Crassanin formula(L,R,"Crassanin", HR,24)
{ ring("bc222o",,,N)
  { 2: bond(30) bond(-30);
    0: ring(,,H,N)
       { 1: ring(,4,H,#N,5,-70)
            { 0: bond(-70,=C) atom("O");
              1: atom("N") bond(r) atom("H");
              vertex(,2,1){ 4: bond(r) atom("O",C,L) atom("H$_3$C",R);
                            5: bond(r) atom("O",C,L) atom("H$_3$C",R); };
            };
       };
    0: bond(-90,o,NN) atom("C",C,R) atom("OOCH$_3$",L);
    4: bond(-60,<<) atom("H");
    7: atom("N"); }
}

Treten andere Ringgr¨oß en hinzu, ist es am einfachsten, ein Grundger¨ust aus F¨unfund Sechsringen aufzubauen und die gr¨oß eren oder kleineren unregelm¨aß igen Ringe in Form von Kohlenstoffbr¨ucken hinzuzuf¨ugen, was im n¨achsten Abschnitt angesprochen wird.

2.2.6 (Kohlenstoff-) Brücken

Br¨ucken k¨onnen in verschiedenen Formen in Erscheinung treten. In den folgenden

F¨allen (a) (b) (c) werden die gestrichelten Bindungen als Br¨ucke betrachtet.

Fall (a) zeichnet sich dadurch aus, daß die Br¨ucke eine Kopie eines Teilst¨uckes des Grundk¨orpers ist, das nur verschoben wurde. Es treten keine Positionierungsprobleme auf, wenn Sie die entsprechenden Teile des Grundk¨orpers exakt ¨ubernehmen (Parallelverschiebungen sind winkeltreu).

Fall (b) ist problematisch, weil die Br¨ucke kein Pendant im Grundger¨ust hat. Wenn Sie die Br¨ucke von einem Br¨uckenkopf her aufbauen, wissen Sie sp¨atestens bei der Bindung, die den Ringschluß herstellt (die letzte der Br¨ucke), nicht, welchen Winkel und welche L¨ange Sie w¨ahlen m¨ussen, um exakt auf den Br¨uckenkopf zu treffen. Hier empfehle ich, die Br¨ucke von beiden Br¨uckenk¨opfen her aufzubauen und die letzte Position eines Teilst¨uckes zu speichern. Diese Position dient als Ziel einer wilden Linie, die vom anderen Teilst¨uck her mit bond(#n) gezeichnet wird. Als wilde Linie w¨ahlen Sie am besten diejenige Bindung des Br¨uckenzuges aus, von der Sie am wenigsten wissen, wie Winkel und L¨ange zu w¨ahlen sind.

Fall (c) ist nochmals problematischer, weil die Br¨ucke so kurz ist, daß keine wilde Linie das unbekannte Teilst¨uck ¨uberbr¨ucken kann. Hier kann helfen, wenn Sie durch eine unsichtbare Bindung einen Punkt anspringen, der die Br¨ucke symmetrische teilt, dessen Position speichern und dann, wiederum mit bond(#n), zwei wilde Linien von den beiden Br¨uckenk¨opfen her ziehen. Welchen Punkt Sie als Zielpunkt ausw¨ahlen, h¨angt leider etwas von ihrer geometrischen Phantasie ab.

Fall (a)

Ein Beispiel f¨ur den einfachen Falle, daß die Br¨ucke einem bereits vorhandenen Teilst¨uck entspricht, das nur verschoben ist, ist das Cytisin. Die kopierte und verschobene Br¨ucke ist gestrichelt dargestellt (was in diesem Falle keine stereochemische Aussage sein soll!!!):

O N H

H N H

(−)-Cytisin formula(L,R,"($-$)-Cytisin",HR,24)
{ ring(,5,1=3=){ 0: bond(r,=C) atom("O");
                 5: atom("N");
               }
  ring(,1,H){ 3: bond(r,t) atom("H");
              5: bond(r,t) atom("H");
              3: bond(30,o; -30,o) atom("N")
                 branch { bond(30,o) atom("H"); }
                 bond(-90,o; -150,o);
            }
}

Auch das Morphin weist eine Verbr¨uckung auf, die einem Ausschnitt des Sechsringes gleicht und nach der Verschiebung wieder exakt auf ein Ringatom des Grundk¨orpers trifft. Im Gegensatz zum Cytisin werden hier aber kurze Bindungen mit ” krummen“ Winkeln f¨ur die Verschiebung sowie dickere Bindungen f¨ur die Br¨ucke benutzt, um einen perspektivischen Eindruck zu erreichen. Der Furanring ist durch drei anellierte Sechsringe sehr verzogen und wird in diesem Falle asymmetrisch wiedergegeben, wobei der Ringschluß durch eine wilde Linie erfolgt:

HO O HO H H N

Morphin formula(L,R,"Morphin",HR,24)
{ ring(,,H1=3=5=)
  { 2: bond(r) atom("HO",R);
    1: bond(-135,,L) atom("O") saveXY(#1);
  }
  ring(,2,H)
  { 1: ring(,3,H5=)
       { 1: bond(r,<.) atom("HO",R);
         2: bond(#1,<.);
         4: bond(t,<<) atom("H");
       };
    5: bond(t,<.) atom("H");
    1: bond(-70,t,S; -30,p; 30,t)
       saveXY(#2)
       atom("N") bond(110,t,S)
       restoreXY(#2) bond(-30);
  }
}

Die Br¨ucke im Lycopodin ist ein weiteres Beispiel f¨ur verschobene Repliken des Grundk¨orpers:

32 Kapitel 2 — Tutorial

H N

O H

Lycopodin formula(L,R,"Lycopodin",HR,24)
{ ring(,2,H){}
  ring(,4,H)
  { 3: bond(b,<<) atom("H");
    5: atom("N");
  }
  ring(,0,H)
  { 4: bond(r,=C) atom("O");
    0: bond(-170,t,S; 150,p) saveXY(#1)
       bond(90,t; 10,t,S)
       restoreXY(#1) bond(-150,<.)
       restoreXY(#1) bond(150,<<) atom("H");
  }
}

Fall (b)

Die folgenden beiden Beispiele weisen eine Br¨ucke auf, die durch Ann¨aherung von beiden Br¨uckenk¨opfen her aufgebaut wird, der Ringschluß erfolgt durch eine wilde Linie (Br¨ucke wiederum gestrichelt gezeichnet):

formula(L,R)
{ ring(,2,H)
  { 1: bond(-120,o) saveXY(#1);
    vertex(,5,2,H,5){ 1: bond(-150,o; #1,o); };
  }
}

H H

(−)-Kauran formula(L,R,"($-$)-Kauran",HR,24)
{ ring(,,H)
  { 3: bond(-90,t) atom("H");
    2: bond(b,o);
    1: bond(b,t) atom("H");
    vertex(,1,1,H,6){3: bond(-120,t); 3: bond(-60,o); };
    vertex(,3,3,H,6){ 1: bond(-40,o) saveXY(#1) bond(30,o);
                     3: bond(-20,o) bond(#1,o);
                   };
  }
}

Die Darstellung einer dem Humulen nahe verwandten Struktur bringt Schwierigkeiten, wenn die E/Z-Isomerie, die aus der Doppelbindung im Neunring resultiert, korrekt wiedergegeben werden soll. Am besten fassen Sie den Neunring als Br¨ucke mit einem unbekannten Teilst¨uck auf, das durch eine wilde Linie wiedergegeben wird. (Im Beispiel entsteht eine Asymmetrie, da diese wilde Linie nicht exakt in der vertikalen Mitte des Vierringes endet; wollen Sie eine perfekt symmetrische Formel, m¨ussen Sie L¨osungen analog dem Falle (c) suchen):

(E)-Isomer formula(L,R,"(E)-Isomer",HR,24)
{ ring(,,H,#N,4,-45)
  { 3: bond(70; 10; -50; -110,=) saveXY(#1);
    0: bond(-70; -10; 50; #1);
  }
}

Fall (c)

Die L¨osung dieses Problems wurde bereits gegeben, im Beispiel ist es sogar einfach, den Zielpunkt zu bestimmen, da er in radialer Verl¨angerung des Ringatoms, das sich zwischen den beiden Br¨uckenk¨opfen befi ndet, liegt (hier Position 5).

(c) formula(L,R,"(c)",HR,24)
{ ring(,,H){ 5: saveXY(#1,-30,N);
             0: bond(#1,o);
             4: bond(#1,o); }
}

Wilde Linien

Aus den vorangegangenen Beispielen ist deutlich geworden, wie n¨utzlich wilde Linien sind (vor allem zum Pfuschen). Sind die Br¨ucken sehr kompliziert und m¨ussen Sie daher un¨ubersichtliche Befehlsketten schreiben, kann es sinnvoll sein, die Position eines oder beider Br¨uckenk¨opfe mit saveXY zu speichern und sp¨ater, auß erhalb der geschachtelten Beschreibungen des Grundk¨orpers, mit restoreXY wieder aufzugreifen.

2.3 Reaktionsketten

Das Konzept der automatischen Positionierung von Formeln zueinander kann mit dem weiteren Grundelement des Reaktionspfeils so erweitert werden, daß nicht nur Formeln nebeneinander gestellt, sondern mit zwischengesetzten Pfeilen zu ganzen Reaktionsschemata oder -ketten verbunden werden. Neben einer einfachen linearen Abfolge sind daneben auch Verzweigungen und deren Gegenteil, die Zusammenf¨uhrung sowie mehrzeilige Schemata m¨oglich.

2.3.1 Horizontale lineare Anordnung

Anstelle der bereits bekannten beiden Positionierungsparameter einer Formel verf¨ugt das Pfeilelement ¨uber die Angabe einer Richtung und L¨ange, um den Anschluß - punkt zu bestimmen. Der Pfeil beginnt dabei stets am aktuellen Startpunkt.Im einfachsten Falle einer horizontal von links nach rechts gerichteten Kette notieren Sie eine Folge von Formeln mit den Positionierungen (L,R) sowie von Pfeilen mit einem Winkel von 0^o (Voreinstellung), wie Schema 2– 2zeigt:

34 Kapitel 2 — Tutorial

\begin{chemistry}[hor1]
  formula(L,R){ ring(){ 4: bond(r); 5: bond(r); } }

  arrow()
  { text(T,C){ formula(C,C){ atom("Br$_2$") }}
    text(B,C){ formula(C,C){ atom("h$\nu$") }}
  }

  formula(L,R)
  { ring(){ 4: bond(r) atom("CH$_2$Br",L);
            5: bond(r) atom("CH$_2$Br",L); }
  }

  arrow()
  { text(T,C){ formula(C,C){ atom("Na$_2$S") }}
    text(B,C){ formula(C,C){ atom("CH$_3$CH$_2$OH") }}
  }

  formula(L,R)
  { ring(,5){}
    ring("cpentane",1){ 4: atom("S"); }
  }

  arrow()
  { text(T,C){ formula(C,C){ atom("NaIO$_4$") }}
    text(B,C){ formula(C,C){ atom("H$_2$O, O $^o$C") }}
  }

  formula(L,R)
  { ring(,5){}
    ring("cpentane",1)
    { 4: atom("S") bond(r,=C) atom("O"); }
  }
\end{chemistry}

Br_2 hν CH_2Br

CH_2Br

Na_2S

CH_3CH_2OH S

NaIO_4

H_2O, O^oC S O

Schema 2–2 Horizontales Reaktionsschema: Darstellung eines Benzothiophens.

Die einzelnen Formeln werden mit dieser Positionierung auf eine gemeinsame Mittellinie, auf der die Reaktionspfeile liegen, zentriert. Es ist auch denkbar, die Ketten von rechts nach links aufzubauen, wobei die Positionierung (R,L) benutzt wird, f¨ur die Pfeile ein Winkel von 180^o.

Eine Erweiterung der Formelsyntax l¨aß t die Beschriftung der Komponenten zu und ist speziell f¨ur den Satz von Formeln in horizontalen Ketten geeignet. Sie m¨ussen dazu den Typ HA (horizontal, absoluter Abstand) w¨ahlen und den Abstand der Textunterkante von der Mittellinie angeben, wie Schema 2– 3 zeigt. Der Befehl \shortstack erlaubt es, Zeilenumbr¨uche innerhalb des Textes vorzunehmen:

\begin{chemistry}[hor2]
  formula(L,R,"o-Xylen",HA,36)

  { ring(){ 4: bond(r); 5: bond(r); } }

  arrow()
  { text(T,C){ formula(C,C){ atom("Br$_2$") }}
    text(B,C){ formula(C,C){ atom("h$\nu$") }}
  }

  formula(L,R,"$\alpha,\alpha’$-o-Xylen",HA,36)
  { ring(){ 4: bond(r) atom("CH$_2$Br",L);
            5: bond(r) atom("CH$_2$Br",L); }
  }

  arrow()
  { text(T,C){ formula(C,C){ atom("Na$_2$S") }}
    text(B,C){ formula(C,C){ atom("CH$_3$CH$_2$OH") }}
  }

  formula(L,R,"\shortstack{Dihydro-\\benzo[b]thiophen}",HA,36)
  { ring(,5){}
    ring("cpentane",1){ 4: atom("S"); }
  }
\end{chemistry}

o-Xylen

Br_2 hν CH_2Br

CH_2Br

α, α^′-o-Xylen

Na_2S

CH_3CH_2OH S Dihydrobenzo[b]thiophen

Schema 2–3 Horizontales Reaktionsschema mit Formelnamen, die auf eine gemeinsame Linie unterhalb des Reaktionspfeils ausgerichtet sind (fester Abstand des Textes zur Mittellinie).

Bei diesem Verfahren ist es erforderlich, den ben¨otigten Abstand der Texte von der Mittellinie abzusch¨atzen oder durch Versuche zu ermitteln. In beiden F¨allen k¨onnen Sie nur schwer f¨ur mehrere Schemata einen gleichm¨aß igen Abstand erreichen. Der Befehl multiline, der speziell f¨ur den Satz n-zeiliger horizontaler Reaktionsketten konzipiert ist, tr¨agt dem jedoch Rechnung, indem er in jeder Zeile die am tiefsten liegende Formelunterkante bestimmt und zu dieser einen konstanten Abstand

rTextSep einh¨alt, um die H¨ohe der gesamten Textzeile zu berechnen. Alle Formel-

beschriftungen in dieser Zeile liegen somit auf gleicher H¨ohe und halten einen identischen Mindestabstand von der gr¨oß ten Formel ein. W¨ahlen Sie n = 1, k¨onnen Sie das angesprochene Problem l¨osen (Schema 2– 4). Beachten Sie, daß Formeln innerhalb von multiline keine Positionierungsparameter f¨ur die Beschriftung (HR und ¨ahnliche) besitzen d¨urfen, da diese Werte ja selbst¨andig bestimmt werden:

\begin{chemistry}[hor3]
multiline(1)
{ formula(L,R,"o-Xylen")
  { ring(){ 4: bond(r); 5: bond(r); } }

  arrow()
  { text(T,C){ formula(C,C){ atom("Br$_2$") }}
    text(B,C){ formula(C,C){ atom("h$\nu$") }}
  }

  formula(L,R,"$\alpha,\alpha’$-o-Xylen")
  { ring(){ 4: bond(r) atom("CH$_2$Br",L);
            5: bond(r) atom("CH$_2$Br",L); }
  }

  arrow()
  { text(T,C){ formula(C,C){ atom("Na$_2$S") }}
    text(B,C){ formula(C,C){ atom("CH$_3$CH$_2$OH") }}
  }

  formula(L,R,"\shortstack{Dihydro-\\benzo[b]thiophen}")
  { ring(,5){}
    ring("cpentane",1){ 4: atom("S"); }
  };
}
\end{chemistry}

o-Xylen

Br_2 hν CH_2Br

CH_2Br

α, α^′-o-Xylen

Na_2S

CH_3CH_2OH S Dihydrobenzo[b]thiophen

Schema 2–4 Horizontales Reaktionsschema mit Formelnamen, die automatisch in konstantem Abstand auf einer gemeinsamen Linie unterhalb des Reaktionspfeils ausgerichtet sind.

Zur Bestimmung des korrekten Abstands der Texte von der Mittellinie werden nur diejenigen Formeln herangezogen, die auch eine Beschriftung aufweisen. Der Fall, daß ein sehr ausgedehntes Zwischenprodukt (zum Beispiel ein vertikal dargestelltes Isomerengemisch ohne Text) die H¨ohe der Beschriftungen der gesamten Zeile bestimmt, tritt somit nicht ein, wie Sie in Schema 2– 8 sehen k¨onnen.

2.3.2 Mehrzeilige Schemata

Die Diskussion ging bislang von einzelnen Reaktionszeilen aus. Es ist jedoch m¨oglich, mehrere solcher Zeilen ¨ubereinandersetzen zu lassen, wobei automatisch zwischen den einzelnen Zeilen ein konstanter Abstand rMultilineSep gesetzt wird, unter Ber¨ucksichtigung eventueller Beschriftungen (Schema 2– 5).

\begin{chemistry}[multli1]
set("rArrowExtend",12)
multiline(2,L)
{ % line 1
  formula(L,R,"Naphthalin")
  { ring(,5){} ring(,1){} }

  arrow(){ text(B,C){ formula(C,C){ atom("AlCl$_3$") } } }

  formula(L,R,"Succinylnaphthalin")

  { ring(,5){}
    ring(,1)
    { 3: bond(r)
         branch { bond(150,=) atom("O"); }
         bond(30; -30; -90) atom("C",C,R) atom("OOH",L); }
  }

  arrow()
  { text(T,C){ formula(C,C){ atom("ZnHg") } }
    text(B,C){ formula(C,C){ atom("HCl") } } }

  formula(L,R,"Naphthylbutters""aure")
  { ring(,5){}
    ring(,1){ 3: bond(r; 30; -30; -90) atom("C",C,R) atom("OOH",L); }
  }
  ;
  %line 2
  arrow()
  { text(T,C){ formula(C,C){ atom("H$_3$PO$_4$") } }
    text(B,C){ formula(C,C){ atom("Zyklisierung") } } }

  formula(L,R)
  { ring(,5){}
    ring(,1){ 3: ring(,1,H){ 5: bond(r,=) atom("O"); }; }
  }

  arrow()
  { text(T,C){ formula(C,C){ atom("ZnHg") } }
    text(B,C){ formula(C,C){ atom("HCl") } } }

  formula(L,R,"Tetrahydrophenanthren")
  { ring(,5){}
    ring(,1){ 3: ring(,1,H){}; }
  }

  arrow(){ text(T,C){ formula(C,C){ atom("DDQ") } } }

  formula(L,R,"Phenanthren")
  { ring(,5){}
    ring(,1){ 3: ring(,1){}; }
  }
  ;
}
\end{chemistry}

Sie k¨onnen auf diese Weise eine lange Reaktionskette in mehrere Zeilen brechen oder auch eine Schar von unabh¨angigen kurzen Ketten ¨ubereinander setzen. Dar¨uberhinaus fi ndet der Befehl Anwendung beim Zusammenfassen mehrerer einzelner Formeln oder gr¨oß eren Reaktionsausschnitten, Beispiele hierzu sind im Abschnitt 2.3.5 gegeben.

Naphthalin

AlCl_3 O COOH

Succinylnaphthalin

ZnHg HCl COOH

Naphthylbuttersäure

H_3PO_4

Zyklisierung O ZnHg HCl

Tetrahydrophenanthren

DDQ

Phenanthren

Schema 2–5 Synthese von Phenanthren, mehrzeilig gesetzt mit multiline.

2.3.3 Vertikale Anordnung

Ebenso einfach wie horizontale Ketten k¨onnen vertikal verlaufende Abl¨aufe gesetzt werden, wobei die Hauptrichtung von oben nach unten mit der Positionierung (T,B), seltener von unten nach oben mit (B,T) erreicht wird, wie man anhand des Schemas 2– 6 (links) sieht. Die Pfeile weisen nun unter einem Winkel von -90^o nach unten:

\begin{chemistry}[vert1]
  formula(T,B)
  { ring(){} }

  arrow(-90)
  { text(T,C){ formula(C,C){ atom("Cl") bond(30)
                branch{ bond(90,=C) atom("O"); }
                        bond(-30; 30) }}
    text(B,C){ formula(C,C){ atom("AlCl$_3$") }}
  }

  formula(T,B)
  { ring()
    { 4: bond(r) branch { bond(r+,=C) atom("O"); }
         bond(r-; r); }
  }

  arrow(-90)
  { text(T,C){ formula(C,C){ atom("S/Morpholin") }} }

  formula(T,B)
  { ring()
    { 4: bond(r; r-; r) atom("COOH",L,R); }
  }

\end{chemistry}

Cl O AlCl_3

O

S/Morpholin

COOH

Benzen

Cl O AlCl_3

O

S/Morpholin

COOH

3-Phenylpropansäure

Schema 2–6 Vertikales Reaktionsschema: Darstellung der 7-Oxononans¨aure. Rechts dasselbe Schema mit Bennenung der Komponenten. Die Texte weisen einen festen Abstand zur Formelunterkante auf.

Auch hierbei ist eine Beschriftung der Formeln m¨oglich, wobei in jedem Falle der Abstand der Textunterkante von der Formelunterkante angegeben werden muß (Typ

V), da keine horizontale Mittellinie existiert. Auch in diesem Falle k¨onnen Zeile-

numbr¨uche mit \shortstack innerhalb des Textparameters auftreten (Schema 2– 6 rechts):

\begin{chemistry}[vert2]
  formula(T,B,"Benzen",V,24)
  { ring(){} }

  arrow(-90)
  { text(T,C){ formula(C,C){ atom("Cl") bond(30)
                branch{ bond(90,=C) atom("O"); }
                bond(-30; 30) }}
    text(B,C){ formula(C,C){ atom("AlCl$_3$") }}
  }

  formula(T,B)
  { ring()
    { 4: bond(r) branch { bond(r+,=C) atom("O"); }
         bond(r-; r); }
  }

  arrow(-90)
  { text(T,L){ formula(C,C){ atom("S/Morpholin") }} }

  formula(T,B,"\shortstack{3-Phenyl-\\propans""aure}",V,24)
  { ring()
    { 4: bond(r; r-; r) atom("COOH",L,R); }
  }
\end{chemistry}

Bislang wurde davon ausgegangen, daß die Hauptrichtung der Kette stets dieselbe ist – von links nach rechts oder von oben nach unten. Dies muß nicht immer der Fall sein, es steht Ihnen frei, die Positionierungsparameter einer Formel zum Beispiel auf

L,B zu setzen, um den Verlauf an dieser Stelle von links-rechts nach oben-unten zu

¨andern und einen Knick zu erreichen.

2.3.4 Kontext und Verzweigung

Es ist nun an der Zeit, das Konzept des Kontextes zu verdeutlichen. Implizit haben Sie es mit dem Zeichnen des ersten Schemas bereits angewandt, ohne es zu wissen. Ein Kontext ist nichts weiter als die Bounding Box einer Formel. Er wird eingesetzt, um die Koordinaten der neuen Kardinalpunkte der Formel (die vier Ecken, die vier Seitenmitten sowie der Mittelpunkt) anhand der Positionierungsparameter <pos> und <Cpos> zu berechnen. Jeder formula-Befehl setzt den aktuellen Kontext auf die Bounding Box der durch ihn gezeichneten Formel.

Unmittelbar sichtbar kann der Kontext durch die Befehle fbox oder bracket gemacht werden: sie nutzen den aktuellen Kontext, um die zugeh¨orige Formel einzurahmen oder mit eckigen Klammern einzuschließ en.

Ein wichtiges Merkmal von Reaktionsketten, das grundlegend auf dem Kontext aufbaut, ist die Verzweigung, durch die Sie von einer Formel ausgehend, mehrere Ketten weiterlaufen lassen k¨onnen. Dabei speichern Sie mit savecontext den Kontext und damit die geometrische Ausdehnung der Formel unter einer bestimmten Nummer und k¨onnen so immer, wenn Sie ihn erneut ben¨otigen, mit setcontext unter dieser Nummer darauf zugreifen. Dies ist zu Beginn jeder weiteren Reaktionskette erforderlich, da durch die folgenden Formeln der Kontext der Verzweigungsformel ¨uberschrieben ist.

Eine einfache Verzweigung der Reaktionskette wird in der Beschreibung zum Befehl

setcontext in Schema 3– 4 gegeben. Hier soll anhand der Reaktionsm¨oglichkeiten

eines Borans die Mehrfachverzweigung dargestellt werden. Ebenso wie bei der einfachen Verzweigung speichern Sie den Kontext der Formel, die als Verzweigungspunkt dienen soll, ein einziges Mal ab, da er sich nicht ¨andert. Zum Unterschied vom einfachen Falle verwenden Sie ihn aber mehrmals, um mit setcontext einen Anschluß punkt zu berechnen:

\begin{chemistry}[cont2]
  formula(L,R)
  { atom("R") bond(30; -30,=) }

  arrow()
  { text(T,L){ formula(C,C){ atom("H") bond(0) atom("B")

        branch{bond(90; 30); bond(-30; 30);} }} }

  formula(L,R,"Alkylboran",HR,24)
  { atom("R") bond(30; -30; 30)
    atom("B")
    branch{bond(90; 30); bond(-30; 30);}
  }
  savecontext(#1)

  arrow()
  { text(T,L){ formula(C,C){ atom("R’-COOD") }} }
  formula(L,R)
  { atom("R") bond(30; -30; 30) atom("D") }

  setcontext(#1,TR)
  arrow(45)
  { text(T,L){ formula(C,C){ atom("AgNO$_3$/OH$^\ominus$") }} }
  formula(BL,R)
  { atom("R") bond(30; -30; 30; -30; 30) atom("R") }

  setcontext(#1,BR)
  arrow(-45)
  { text(T,L){ formula(C,C){ atom("H$_2$O$_2$/OH$^\ominus$") }} }
  formula(TL,R)
  { atom("R") bond(30; -30; 30) atom("OH",L) }

  setcontext(#1,B)
  arrow(-90)
  { text(T,L){ formula(C,C){ atom("Br$_2$/OH$^\ominus$") }} }
  formula(T,R)
  { atom("R") bond(30; -30; 30) atom("Br",L) }
\end{chemistry}

Weitere M¨oglichkeiten der Verzweigungen, wobei mehrere Formeln einen gemeinsamen Kontext bilden und so als Einheit agieren, werden im folgenden Abschnitt vorgestellt.

2.3.5 Reaktionsabschnitte als Einheit betrachten

So kann der Wunsch bestehen, innerhalb einer horizontalen Kette ein Zwischenprodukt mit mehreren mesomeren Zust¨anden vertikal ¨ubereinander darzustellen, wie dies in Schema 2– 8 zu sehen ist. Hier besteht die L¨osung darin, einen multiline- Befehl mit einem kleinen vertikal orientierten Schema als einzeiligen Inhalt zu verwenden. Die beiden Grenzstrukturen k¨onnen so als Einheit betrachet werden, die durch die Positionierung L,R zur Reaktionskette zentriert ist. Die innere Ausrichtung ist bei einzeiligem Inhalt von multiline sinnlos und kann beliebig gew¨ahlt werden. Zus¨atzlich wurde eine multiline-Umgebung benutzt, um die Formelbeschriftung auf gleiche H¨ohe zu bringen:

\begin{chemistry}[multli2]
multiline(1)
{ formula(L,R,"Benzen"){ ring(){} }

  arrow()

42 Kapitel 2 — Tutorial

R H B R B Alkylboran

R’-COOD R D

AgNO_3/OH^⊖ R R

H_2O_2/OH^⊖

R OH

Br_2/OH^⊖

R Br

Schema 2–7 Was passiert mit dem Boran? Mehrfachverzweigungen, von derselben Formel ausgehend, k¨onnen es zeigen. Der Kontext der Verzweigungsstelle, hier des Borans, muß daf¨ur gespeichert werden.

  { text(T,L){ formula(C,C){ atom("Cl") bond(30)
      branch{bond(90,=C) atom("O");} bond(-30) } }
  }

  multiline(1,C)
  { formula(T,B)
    { ring(,,H0=2=)
      { 4: bond(30) branch{ bond(90,=C) atom("O");} bond(-30);
        4: bond(b) atom("H");
        5: bond(r,s,S) atom("$\oplus$"); }
    }
    arrow(-90,,<=>){}
    formula(T,B)
    { ring(,,H2=5=)
      { 4: bond(30) branch{ bond(90,=C) atom("O");} bond(-30);
        4: bond(b) atom("H");
        1: bond(r,s,S) atom("$\oplus$"); }
    };
  }

  arrow(){}

  formula(L,R,"Azetylbenzen")
  { ring()
    { 4: bond(r) branch { bond(r+,=C) atom("O");} bond(r-); }
  }
;}
\end{chemistry}

Benzen

Cl O

O H ⊕

O H ⊕

O

Azetylbenzen

Schema 2–8 Ein unabh¨angiges vertikales Schema wird als einzeiliges multiline- Objekt gesetzt und kann so als Einheit in den horizontalen Formelverlauf eingef¨ugt werden.

Sie k¨onnen mit diesem Verfahren ebensogut horizontal angeordnete Grenzstrukturen oder Reaktionsabschnitte innerhalb von vertikalen Ketten anordnen, wobei

T,B als Positionierung eingesetzt werden muß . Dies ist beim Gemisch verschiede-

ner Chlorbutene im Schema 2– 22 benutzt worden.

2.3.6 Kontexte zusammenfassen

multiline kann Ihnen auch weiterhelfen, wenn Sie einige Formeln zusammen-

fassen wollen, um eine Klammerung aller beteiligten Formeln erzielen m¨ochten, zum Beispiel um die mesomeren Zwischenstrukturen des Schemas 2– 8. Der Befehl bracket klammert jedoch nur die letzte gezeichnete Formel. Eine einzeilige

multiline-Einheit liefert nach auß en hin einen Superkontext, das heiß t einen Kon-

text, der alle eingeschachtelten Kontexte enth¨alt und es daher erlaubt, eine Kettenverzweigung oder Einrahmung auf alle beteiligten Formeln zu beziehen. Betrachten Sie als Beispiel f¨ur eine Einrahmung Schema 2– 9:

\begin{schema}
\begin{chemistry}[multli3]
multiline(1)
{ formula(L,R,"Benzen"){ ring(){} }

  arrow()
  { text(T,L){ formula(C,C){ atom("Cl") bond(30)
      branch{bond(90,=C) atom("O");} bond(-30) } }
  }

  multiline(1,C)
  { formula(T,B)
    { ring(,,H0=2=)
      { 4: bond(30) branch{ bond(90,=C) atom("O");} bond(-30);
        4: bond(b) atom("H");
        5: bond(r,s,S) atom("$\oplus$"); }
    }
    arrow(-90,,<=>){}

44 Kapitel 2 — Tutorial

    formula(T,B)
    { ring(,,H2=5=)
      { 4: bond(30) branch{ bond(90,=C) atom("O");} bond(-30);
        4: bond(b) atom("H");
        1: bond(r,s,S) atom("$\oplus$"); }
    }
    ;
  }
  bracket()
  arrow(){}

  formula(L,R,"Azetylbenzen")
  { ring()
    { 4: bond(r) branch { bond(r+,=C) atom("O");} bond(r-); }
  }
;}
\end{chemistry}

Benzen

Cl O

O H ⊕

O H ⊕

O

Azetylbenzen

Schema 2–9 Ein Superkontext mit multiline erlaubt die Klammerung mehrerer Formeln. Er wird durch ein multiline-Objekt gebildet.

Anstelle des bracket-Befehls k¨onnen Sie ebensogut mehrfache Verzweigungen mit

savecontext und setcontext aufbauen, wobei der Superkontext als Grundlage

verwendet wird. Unabh¨angig von diesem k¨onnen Sie aber auch auf die Einzelkontexte zugreifen, sofern diese innerhalb von multiline gespeichert werden, wie das Beispiel des Schemas 2– 10 zeigt:

\begin{chemistry}[multli4]
formula(L,R)
{ bond(-30,=) bond(30)
  branch {bond(90,=C) atom("O");} bond(-30);
}

arrow()
{ text(T,C){ formula(C,C){ atom("+ H$^\oplus$") } } }

multiline(1,C)
{ formula(T,B)
  { bond(-30,=; 30) branch{ bond(-90,s,S) atom("$\oplus$");
                            bond(90) atom("O",C,R) atom("H",L);
                          }

                bond(-30)
  }
  savecontext(#1)
  arrow(-90,,<=>){}
  formula(T,B)
  { atom("$\oplus$") bond(-90,s,S)
    bond(-30; 30,=)
    branch { bond(90) atom("O",C,R) atom("H",L); }
    bond(-30)
  }
  savecontext(#2)
  ;
}
bracket()

setcontext(#1,R)
arrow(){}
formula(L,R,"1,2-Addukt",HR,24)
{ bond(-30,=; 30)
  branch{ bond(150);
          bond(90,=C) atom("O");
        }
  bond(-30) atom("R")
}
setcontext(#2,R)
arrow(){}
formula(L,R,"1,4-Addukt",HR,24)
{ atom("R") bond(-90)
  bond(-30; 30)
  branch { bond(90,=C) atom("O"); } bond(-30)
}
\end{chemistry}

O + H^⊕ ⊕ OH

⊕ OH

OH R 1,2-Addukt

R O

1,4-Addukt

Schema 2–10 Unabh¨angig vom Superkontext k¨onnen Verzweigungen auf die einzelnen Teilformeln ausgerichtet werden, wenn deren Kontexte einzeln gespeichert werden.

Das Speichern der Kontexte einzelner Formeln eines multiline-Zusammenhanges ist von allgemeiner Bedeutung, wenn zwar multiline zum Ausrichten der Formelbeschriftungen benutzt werden soll, aber dennoch die Formeln Verzweigungspunkte darstellen. Hierzu folgt ein weiteres Beispiel (Schema 2– 11):

\begin{chemistry}[multli5]

46 Kapitel 2 — Tutorial

multiline(1)
{ formula(L,R,"Zimts""aure")
  { ring(){5: bond(r;r/,=;r) atom("COOH",L); } }
  savecontext(#1)

  arrow(){}
  formula(L,R,"p-Cumars""aure")
  { ring(){5: bond(r;r/,=;r) atom("COOH",L);
           2: bond(r) atom("HO",R); }
  }
  savecontext(#2)

  arrow(){}
  formula(L,R,"Kaffees""aure")
  { ring(){5: bond(r;r/,=;r) atom("COOH",L);
           2: bond(r) atom("HO",R);
           1: bond(r) atom("HO",R); }
  }
;}
setcontext(#1,B) arrow(-90){}
formula(T,B,"Cumarin",V,24)
{ ring(,,H5=)
  { vertex(,1,1){};
    3: atom("O");
    4: bond(r,=C) atom("O"); }
}
setcontext(#2,TR) arrow(45){}
setcontext(#2,BR) arrow(-45){}
\end{chemistry}

COOH

Zimtsäure

COOH HO

p-Cumarsäure

COOH HO HO Kaffeesäure

O O

Cumarin

Schema 2–11 Mit multiline angeordnete Formelbeschriftungen und Verzweigungen, erm¨oglicht durch individuelle Kontexte.

F¨ur jede Formel, die als Verzweigungspunkt dient, wird ein Individualkontext gespeichert. Mit multiline erzeugte Formelbeschriftungen gehen dabei in den Kontext mit ein, sodaß Sie bei nach unten f¨uhrenden Pfeilen wie gezeigt den korrekten Anschluß punkt erhalten.

Da der multiline-Block als Ganzes betrachtet wird, ergeben sich Positionierungsprobleme, wenn auß erhalb des Block eine Formel ohne Beschriftung folgen soll. Deren Mittellinie l¨age mittig zum gesamten vorangegangegen Block samt Beschriftung:

COOH HO HO Kaffeesäure HO HO O O multiline(1)
{ formula(L,R,"Kaffees""aure")
  { ring(){5: bond(r;r/,=;r) atom("COOH",L);
           2: bond(r) atom("HO",R);
           1: bond(r) atom("HO",R); }
          }
          arrow(,12){}
        ;}
        formula(L,R)
{ ring(,,H5=)
  { vertex(,1,1){ 4: bond(r) atom("HO",R);
                  5: bond(r) atom("HO",R); };
    3: atom("O");
    4: bond(r,=C) atom("O"); }
}

Die L¨osung besteht wiederum darin, den Kontext der letzten Formel zu speichern und nach Abschluß des multiline-Blockes zu setzen:

COOH HO HO Kaffeesäure HO HO O O multiline(1)
{ formula(L,R,"Kaffees""aure")
  { ring(){5: bond(r;r/,=;r) atom("COOH",L);
           2: bond(r) atom("HO",R);
           1: bond(r) atom("HO",R); }
          }
          savecontext(#1)
        ;}
        setcontext(#1,R)
arrow(,12){}
formula(L,R)
{ ring(,,H5=)
  { vertex(,1,1){ 4: bond(r) atom("HO",R);
                  5: bond(r) atom("HO",R); };
    3: atom("O");
    4: bond(r,=C) atom("O"); }
}

2.3.7 Zusammenführung mehrerer Zweige

Reaktionsketten k¨onnen in mehrere Teilzweige aufgespalten werden, aber auch das Gegenteil, die Zusammenf¨uhrung mehrerer Zweige zu einem einzigen, etwa im Zuge einer konvergenten Synthese, ist oft erforderlich. Je nachdem, ob die ¨Aste horizontal oder vertikal zusammenlaufen sollen, k¨onnen Sie zwischen den Befehlen

joinh und joinv w¨ahlen.

Als Beispiel f¨ur ein horizontal ausgerichtetes Schema dient die Synthese eines Kro-

48 Kapitel 2 — Tutorial

nenethers.

\begin{chemistry}[joinh1]
joinh(2,L)
{ % line1
  formula(L,R,"Bis-chlorethylether")
  { atom("Cl") bond(45; 0; -45)
    atom("O") bond(45; 0; -45) atom("Cl")
  }
  arrow(){}
  formula(L,R)
  { ring()
    { 5: bond(r) atom("OH",L);
      4: bond(30) atom("O") bond(90; 30; -30)
         atom("O")
         bond(30; -30; -90) atom("O")
         bond(-30)
         ring(,2){ 1: bond(r) atom("HO",R); }; }
  };

  % line 2
  formula(L,R,"2,6-Bishydroxymethyl-pyridin")
  { ring()
    { 0: atom("N");
      1: bond(r; r+) atom("O",C,L) atom("H",R);
      5: bond(r; r-) atom("O",C,R) atom("H",L); }
  }
  arrow(){}
  formula(L,R)
  { ring()
    { 0: atom("N");
      1: bond(r; r+) atom("Br");
      5: bond(r; r-) atom("Br"); }
  };
}

arrow(){}

formula(L,R,"Dibenzopyridino[18]krone-6",HR,24)
{ ring()
  { 5: bond(-30) atom("O")
       bond(-90; -30; 30) atom("O")
       bond(-30; 30; 90) atom("O") bond(30);
    4: bond(30) atom("O") bond(90; 30)
       ring(,1)
       { 5: bond(-30; -90) atom("O") bond(-30)
            ring(,2) {};
         0: atom("N");
       };
  }
}
\end{chemistry}

Auch mehrfache horizontale Zusammenf¨uhrungen sind m¨oglich, wenn Sie in einen oder mehrere der Zweige weitere joinh-Befehle schreiben (Schemata 2– 13):

\begin{chemistry}[joinhh]

Cl O Cl

Bis-chlorethylether OH O O O HO

N OH OH

2,6-Bishydroxymethyl-pyridin

N Br Br

O O O O O N

Dibenzopyridino[18]krone-6

Schema 2–12 Die Synthese eines Kronenethers als Beispiel f¨ur die Zusammenf¨uhrung zweier horizontaler Ketten mit joinh.

joinh(2,R)
{ joinh(2,R)
  { formula(L,R,"Anilin")
    { ring(){ 4: bond(r) atom("N",C,R) atom("H$_2$",L);} }
    ;
    formula(L,R,"Salpetrige S""aure")
    { atom("HNO$_2$") }
    ;
  }
  arrow(){}
  formula(L,R,"Diazoniumion")
  { ring(){ 4: bond(r) atom("N",C,R) atom("$_2^+$",L);} }
  ;
  formula(L,R,"Anilin")
  { ring(){ 5: bond(r) atom("N",C,R) atom("H$_2$",L);} }
  ;
}
arrow(){}
formula(L,R,"p-Amino-diazobenzen",HR,24)
{ ring(){ 4: bond(r) atom("N=N",L,R) bond(r/)
             ring(,0,,,,r)
             { 3: bond(r) atom("N",C,R) atom("H$_2$",L);};
        }
}
\end{chemistry}

Analog k¨onnen Sie auch vertikale Schemata zusammenf¨uhren. Das Schema 2– 14 zeigt dies mit zwei ineinandergeschachtelten joinv-Einheiten:

\begin{chemistry}[joinv1]
set("rArrowExtend",18)
joinv(2,B)
{ joinv(2,B)
  { formula(T,B)
    { bond(30,=; -30; 30)
      branch { bond(150); bond(30) atom("OH",L); } bond(-30)

50 Kapitel 2 — Tutorial

NH_2

Anilin

HNO_2

Salpetrige Säure

N^+_2

Diazoniumion

NH_2

Anilin

N=N NH_2

p-Amino-diazobenzen

Schema 2–13 Homogene geschachtelte Zusammenf¨uhrungen mehrerer Reaktionszweige.

  }
  arrow(-90){ text(T,L){formula(C,C){atom("HBr")}}}
  formula(T,B)
  { atom("Br") bond(30; -30; 30,=)
    branch { bond(90); } bond(-30)
  };

  formula(T,B,"Tosylchlorid",V,24)
  { ring(){ 2: bond(r); 5: bond(r) atom("SO$_2$Cl",L); }
  }
  arrow(-90)
  { text(T,L){formula(C,C){atom("NaOH")}}
     text(B,L){formula(C,C){atom("Zn")}}
  }
  formula(T,B)
  { ring(){ 2: bond(r); 5: bond(r) atom("SO$_2$Na",L); }
  };
}
arrow(-90){} % 80%

formula(T,B)
{ ring(){ 2: bond(r);
          5: bond(r) atom("SO$_2$",L,R) bond(30; -30; 30,=)
             branch{bond(90);} bond(-30); }
};

formula(T,B,"3-Methylbutens""aure",V,24)
{ bond(30) branch{ bond(90); } bond(-30,=; 30) atom("COOH",L)
}
arrow(-90){ text(T,L){formula(C,C){atom("CH$_3$OH")}}}
formula(T,B)
{ bond(30) branch{ bond(90); } bond(-30,=; 30) atom("CO$_2$CH$_3$",L)

  };
}

arrow(-90){ text(T,L){formula(C,C){atom("NaOCH$_3$")}}}

formula(T,B)
{ ring()
  { 3: bond(r);
    0: bond(r) atom("S")
       branch { bond(180,=C) atom("O");
                bond(0,=C) atom("O"); }
       bond(-90)
       saveXY(#1)
       bond(-150) bond(150,=)
         branch { bond(-150); bond(90); }
       restoreXY(#1)
       bond(-30) branch { bond(-150);
                          bond(-30); }
       bond(30)
       branch{ bond(90,s,S) atom("$\ominus$"); }
       bond(-30) atom("CO$_2$CH$_3$",L);
  }
}
bracket()

arrow(-90,10){}
arrow(-90,10){ text(T,L){formula(C,C){atom("KOH")}}}

formula(T,B,"\textit{trans}-Chrysanthemums""aure",V,24)
{ ring(,,H,,3,90)
  { 1: bond(r,<.) atom("COOH",L);
    2: bond(150,<<) bond(-150,=) branch{ bond(-90); bond(150); };
    0: bond(t);
    0: bond(b);
  }
}
\end{chemistry}

Bei Bedarf k¨onnen Sie auch mehrfach geschachtelte gemischte Zusammenf¨uhrungen formulieren, zum Beispiel als horizontales Schema, das ein vertikales Teilschema enth¨alt (Schema 2– 15):

joinh(2,R)
{ joinv(2,B)
  { formula(T,B,"Anilin",V,24)
    { ring(){ 4: bond(r) atom("N",C,R) atom("H$_2$",L);} }
    ;
    formula(T,B,"Salpetrige S""aure",V,24)
    { atom("HNO$_2$") }
    ;
  }
  arrow(-90,12,-){} nospace arrow(){}
  formula(L,R,"Diazoniumion")
  { ring(){ 4: bond(r) atom("N",C,R) atom("$_2^+$",L);} }
  ;
  formula(L,R)
  { ring(){ 5: bond(r) atom("N",C,R) atom("H$_2$",L);} }

OH

HBr

Br

SO_2Cl Tosylchlorid NaOH Zn

SO_2Na

SO_2

COOH 3-Methylbutensäure CH_3OH

CO_2CH_3

NaOCH_3

S O O ⊖ CO_2CH_3

KOH

COOH trans-Chrysanthemumsäure Schema 2–14 Synthese der Chrysanthemums¨aure, gesetzt mit joinv.

  ;
}
arrow(){}
formula(L,R,"p-Amino-diazobenzen",HR,24)
{ ring(){ 4: bond(r) atom("N=N",L,R) bond(r/)
             ring(,0,,,,r)
             { 3: bond(r) atom("N",C,R) atom("H$_2$",L);};
        }
}

NH_2

Anilin

HNO_2

Salpetrige Säure

N^+_2

Diazoniumion

NH_2

N=N NH_2

p-Amino-diazobenzen

Schema 2–15 Synthese einer Diazoverbindung. Horizontale und vertikale Zusammenf¨ugung gemischt.

Das Pendant mit vertauschten Rollen ist ein vertikales Schema mit horizontalem Anteil (Schema 2– 16):

\begin{chemistry}[joinvh]
joinv(2,T)
{ joinh(2,L)
  { formula(L,R,"Anilin")
    { ring(){ 4: bond(r) atom("N",C,R) atom("H$_2$",L);} }
    ;
    formula(L,R,"Salpetrige S""aure")
    { atom("HNO$_2$") }
    ;
  }
  arrow(,12,-){} nospace arrow(-90){}
  formula(T,B,"Diazoniumion",V,24)
  { ring(){ 4: bond(r) atom("N",C,R) atom("$_2^+$",L);} }
  ;
  formula(T,B,"Anilin",V,24)
  { ring(){ 5: bond(r) atom("N",C,R) atom("H$_2$",L);} }
  ;
}
arrow(-90){}
formula(T,B,"p-Amino-diazobenzen",V,24)

{ ring(){ 4: bond(r) atom("N=N",L,R) bond(r/)
             ring(,0,,,,r)
             { 3: bond(r) atom("N",C,R) atom("H$_2$",L);};
        }
}
\end{chemistry}

NH_2

Anilin

HNO_2

Salpetrige Säure N^{+2}

Diazoniumion

NH_2

Anilin

N=N NH_2

p-Amino-diazobenzen

Schema 2–16 Gemischte joinv/joinh-Befehle.

2.3.8 Mischung der Grundelemente

Es ist m¨oglich, die bislang vorgestellten M¨oglichkeiten zum Satz komplexer Formeleinheiten (mehrzeilige Bl¨ocke und Zusammenf¨uhrungen) miteinander zu mischen. Die homogene Mischung der Zusammenf¨uhrung wurde bereits vorgestellt, hier soll gezeigt werden, wie eine Zusammenf¨uhrung zu Beginn eines einzeiligen multiline- Rumpfes eingebettet wird (Schema 2– 17):

\begin{chemistry}[joinhml]
multiline(1)
{ joinh(2,L)
  { formula(L,R,"Anilin")
    { ring(){ 4: bond(r) atom("N",C,R) atom("H$_2$",L);} }
    ;
    formula(L,R,"Salpetrige S""aure")
    { atom("HNO$_2$") }

    ;
  }
  arrow(){}
  formula(L,R,"Diazoniumion")
  { ring(){ 4: bond(r) atom("N",C,R) atom("$_2^+$",L);} }

  arrow()
  { text(T,L){formula(L,R)
    { ring(){ 5: bond(r) atom("N",C,R) atom("H$_2$",L);} } } }

  formula(L,R,"\shortstack{p-Amino-\\ diazobenzen}")
  { ring(){ 4: bond(r) atom("N=N",L,R) bond(r/)
               ring(,0,,,,r)
               { 3: bond(r) atom("N",C,R) atom("H$_2$",L);};
          }
  }
;
}
\end{chemistry}

Weitere Zeilen k¨onnen bei Bedarf der multiline-Umgebung hinzugef¨ugt werden.

NH_2

Anilin

HNO_2

Salpetrige Säure

N^+_2

Diazoniumion

NH_2 N=N

NH_2 p-Amino-diazobenzen

Schema 2–17 Zusammenf¨uhrung horizontaler Reaktionsketten und multiline gemischt.

2.4 Zyklohexan und Saccharide

In der Literatur sind verschiedene Darstellungen von Sacchariden ¨ublich. Textformeln in der Darstellung nach FISCHER k¨onnen mit atom gesetzt werden, die Verzweigungen nach links und rechts sind mit branch nachzubilden:

56 Kapitel 2 — Tutorial

CHO HO H C H OH C H OH C CH_2OH

D-Arabinose formula(L,R,"D-Arabinose",HR,24)
{ branch { atom("C",C,R) atom("HO",L); } bond(-90)
   branch { bond(180) atom("HO",R); bond(0) atom("H",L); }
   atom("C",C,C) bond(-90)
   branch { bond(180) atom("H",R); bond(0) atom("OH",L); }
   atom("C",C,C) bond(-90)
   branch { bond(180) atom("H",R); bond(0) atom("OH",L); }
   atom("C",C,C) bond(-90)
   atom("C",C,R) atom("H$_2$OH",L)
}

Bei h¨aufi gem Einsatz solcher Formeln ist ein Makro empfehlenswert, zum Beispiel:

\begin{chemspecial}
define(‘TOP_’, ‘branch { atom("C",C,R) atom("$1",L); } bond(-90)’)
define(‘MID_’, ‘branch { bond(180) atom("$1",R);
                         bond(0) atom("$2",L); }
                atom("C",C,C) bond(-90) ’)
define(‘BOTTOM_’, ‘atom("C",C,R) atom("$1",L); }’)
\end{chemspecial}
\begin{chemistry}
  formula(L,R,"D-Arabinose",HR,24)
  { TOP_(‘HO’)
     MID_(‘HO’, ‘H’)
     MID_(‘H’, ‘OH’)
     MID_(‘H’, ‘OH’)
     BOTTOM(‘H$_2$OH’)
  }
\end{chemistry}

Eine schematische r¨aumliche Darstellung (Formel nach HAYWORTH) von Furanosen kann mit dem Ringtyp furanose erreicht werden, eventuell mit Angabe einer Bindungsliste zur Verst¨arkung der perspektivischen Bindungen:

O OH

OH OH OH

β-D-Xylose formula(C,C,"$\beta$-D-Xylose",HR,24)
{ ring("furanose",0,,L)
  { 0: atom("O");
    1: bond(90) atom("O",C,R) atom("H",L);
    2: bond(-90) atom("O",C,R) atom("H",L);
    3: bond(90) atom("O",C,R) atom("H",L);
    4: bond(90;180) atom("O",C,L) atom("H",R);
  }
}

O OH

OH OH OH

β-D-Xylose formula(C,C,"$\beta$-D-Xylose",HR,24)
{ ring("furanose",0,1<<2b3>>,L)
  { 0: atom("O");
    1: bond(90) atom("O",C,R) atom("H",L);
    2: bond(-90) atom("O",C,R) atom("H",L);
    3: bond(90) atom("O",C,R) atom("H",L);
    4: bond(90;180) atom("O",C,L) atom("H",R);
  }
}

Innerhalb des Ringes kann die Darstellung von vic-Hydroxylgruppen in cis-Stellung aus Platzgr¨unden problematisch sein. Sie k¨onnen jedoch das Wasserstoffsymbol mit TEX-Befehlen n¨aher an das Sauerstoffsymbol r¨ucken. Damit Teile des ” O“ nicht ¨uberdeckt werden, muß beim Setzen des ” H“ das L¨oschen des Hintergrundes abgeschaltet werden:

O OH OH OH OH

β-D-Lyxose formula(C,C,"$\beta$-D-Lyxose",HR,24)
{ ring("furanose",0,,L)
  { 0: atom("O");
    1: bond(90) atom("O",C,R) atom("H",L);
    2: bond(90) atom("O",C,L) atom("H\kern-.2em",R,,0);
    3: bond(90) atom("O",C,R) atom("\kern-.2emH",L,,0);
    4: bond(90;180) atom("O",C,L) atom("H",R);
  }
}

Furanoseringe k¨onnen wie normale Ringe kombiniert werden, um Oligosaccharide aufzubauen. Im folgenden Beispiel wird der Bindungstyp ~ verwendet, um ein Anomer zu kennzeichnen:

O O O OH OH OH OH OH OH

D-Xylofuranosyl-(1→4)-α-D-arabinose formula(C,C,
"D-Xylofuranosyl-(1$\to$4)-$\alpha$-D-arabinose",HR,24)
{ ring("furanose",0,1<<2b3>>,L)
  { 0: atom("O");
    1: bond(0,~) atom("O",C,C) bond(0;-90)
    ring("furanose",4,1<<2b3>>,L)
    { 0: atom("O");
      1: bond(-90) atom("O",C,R) atom("H",L);
      2: bond(90) atom("O",C,L) atom("H",R);
      3: bond(-90) atom("O",C,R) atom("H",L);
    }
    ;
    2: bond(-90) atom("O",C,R) atom("H",L);
    3: bond(90) atom("O",C,R) atom("H",L);
    4: bond(90;180) atom("O",C,L) atom("H",R);
  }
}

Auch Pyranosen k¨onnen in der HAYWORTH-Darstellung wiedergegeben werden. Hierzu ist kein spezieller Ringtyp erforderlich, es kann ein normaler groß er Sechsring verwendet werden, eventuell mit Angabe einer Bindungsliste zur Erh¨ohung des r¨aumlichen Eindrucks:

O OH OH OH OH

CH_2OH

α-D-Glucopyranose formula(L,R,"$\alpha$-D-Glucopyranose",HR,24)
{ ring(,,H,L,,0)
  { 5: atom("O");
    0: bond(-90) atom("O",C,R) atom("H",L);
    1: bond(-90) atom("O",C,R) atom("H",L);
    2: bond(90) atom("O",C,R) atom("H",L);
    3: bond(-90) atom("O",C,L) atom("H",R);
    4: bond(90) atom("C",C,R) atom("H$_2$OH",L);
  }
}

58 Kapitel 2 — Tutorial

O OH OH OH OH

CH_2OH

α-D-Glucopyranose formula(L,R,"$\alpha$-D-Glucopyranose",HR,24)
{ ring(,,H0<<1b2>>,L,,0)
  { 5: atom("O");
    0: bond(-90) atom("O",C,R) atom("H",L);
    1: bond(-90) atom("O",C,R) atom("H",L);
    2: bond(90) atom("O",C,R) atom("H",L);
    3: bond(-90) atom("O",C,L) atom("H",R);
    4: bond(90) atom("C",C,R) atom("H$_2$OH",L);
  }
}

Mit Hilfe von wilden Linien k¨onnen Derivate der Saccaride gesetzt werden, wie die 1,2-3,4-Di-O-sec-butyliden-β-D-arabino-pentopyranose zeigt:

O O O

O O formula(L,R)
{ ring(,,H0t1t2t,L,,0)
  { 5: atom("O");
    1: bond(90) atom("O") saveXY(#1);
    0: bond(90) atom("O") bond(20,,L)
         branch{bond(-30); bond(90);} bond(#1);
    2: bond(-90) atom("O") saveXY(#2);
    3: bond(-90) atom("O") bond(-120,,L)
         branch{bond(150); bond(-90);} bond(#2);
  }
}

F¨ur die genauere Darstellung der sterischen Verh¨altnisse der Pyranosen kann der Ringtyp chair genutzt werden, der einen Zyklohexanring in Stereodarstellung (” Sesselform“ ) zeigt. Hilfreich sind hierbei die symbolischen Winkelangaben t und b f¨ur axiale und ¨aquatoriale Substituenten. Aus Platzgr¨unden muß die Hydroxylgruppe an Position 5 um 20^o aus ihrer normalen t-Position gedreht werden:

OH

OH

OH OH O OH formula(L,R)
{ ring("chair",,,L,-1)
  { 0: bond(b) atom("O",C,L) atom("H",R);
    5: bond(20+t) atom("O",C,L) atom("H",R);
    4: bond(b) atom("O",C,R) atom("H",L);
    3: bond(b) atom("O",C,R) atom("H",L);
    2: atom("O");
    1: bond(t; t-) atom("OH",L);
  }
}

HO HO OH OH O OH formula(L,R)
{ ring("chair",,,L)
  { 0: bond(b) atom("HO",R);
    5: bond(t) atom("HO",R);
    4: bond(b) atom("O",C,L) atom("H",R);
    3: bond(b) atom("O",C,R) atom("H",L);
    2: atom("O");
    1: bond(t; t-) atom("OH",L);
  }
}

Das Zyklohexanger¨ust wurde so konstruiert, daß die 1→6-glykosidische Bindung von Di- und Polysacchariden mit mit einem bond-Befehl doppelt-normaler L¨ange (NN) oder – mit Ethersauerstoff – zwei aufeinanderfolgenden Bindungen normaler L¨ange erzeugt werden kann. Der zweite Ring liegt dann auf der gleichen H¨ohe wie der erste, sodaß Oligosaccharide horizontal verlaufen:

O O O formula(L,R)
{ ring("chair",,,L)
  { 2: atom("O");
    3: bond(t,,L) atom("O") bond(t,,L)
       ring("chair",,,L){ 2: atom("O"); };
  }
}

Durch die Symmetrie des chair-Ger¨ustes k¨onnen mehrere Ringe aneinander gesetzt werden, um zum Beispiel ein Steroidger¨ust r¨aumlich zu veranschaulichen. F¨ur den f¨unfgliedrigen D-Ring m¨ussen Sie eine ” wilde“ Verbindungslinie ziehen:

OH 3-α-Hydroxysteran formula(L,R,"3-$\alpha$-Hydroxysteran",HR,24)
{ ring("chair",,,L,-1)
  { 0: bond(b)
       atom("O",C,L) atom("H",R);
       vertex("chair",3,0,3-4-5-,-1)
       { vertex("chair",2,5,3-4-5-,-1)
         { 3: bond(b,,L) saveXY(#1);
           4: bond(t,,L; -t|,,L; #1);
         };
       };
  }
}

OH 3-β-Hydroxysteran formula(L,R,"3-$\beta$-Hydroxysteran",HR,24)
{ ring("chair",,,L,-1)
  { 0: bond(t)
       atom("O",C,L) atom("H",R);
       vertex("chair",3,0,3-4-5-,-1)
       { vertex("chair",2,5,3-4-5-,-1)
         { 3: bond(b,,L) saveXY(#1);
           4: bond(t,,L; -t|,,L; #1);
         };
       };
  }
}

Im Falle des Valerans ergibt sich durch das Verschmelzen zweier Ringkanten mit normalerweise unterschiedlicher L¨ange eine kleine Unsch¨onheit, da der eine Ring entsprechend vergr¨oß ert werden muß , um die Ringkanten anzugleichen:

60 Kapitel 2 — Tutorial

Valeran formula(L,R)
{ ring(,4,H){}
  ring(,2,H){ 1: bond(t,o);
              2: bond(b,o);
              5: bond(r,t) branch { bond(r+); bond(r-); }; }
}

formula(L,R,"Valeran",HR,24)
{ ring("chair",,,,-1)
  { 3: bond(b);
    vertex("chair",3,0,5p)
    { 4: bond(0) branch{bond(0++); bond(0+-);}; };
    4: bond(-90); }
}

Sie k¨onnen das Formelbild durch manuelle Bearbeitung etwas verbessern:

Valeran formula(L,R,"Valeran",HR,24)
{ ring("chair",,,,-1)
  { 3: bond(b);
    4: saveXY(#1);
    3: bond(t; 30; -120)
       saveXY(#2) bond(-90,p; #1)
       restoreXY(#2) bond(0) branch{bond(0++); bond(0+-);};
    4: bond(-90); }
}

Der variable Parameter des Zyklohexanger¨ustes erlaubt es, zwischen den beiden denkbaren Hauptkonformeren auszuw¨ahlen. Schema 2– 18 kann damit den Verlauf einer Eliminierungsreaktion erkl¨aren:

\begin{chemistry}[conf]
  formula(L,R,"3-Menthyltosylat",HA,36)
  { ring("chair",,,L,-1)
    { 0: bond(t) branch{ bond(t-); bond(t+);} ;
      0: bond(b) atom("H");
      5: bond(130,,L) atom("O",C,R) atom("Tos",L);
      5: bond(b) atom("H");
      4: bond(t) atom("H",L);
      3: bond(b); }
  }

  arrow(,,<=>){}

  formula(L,BR)
  { ring("chair",,,L)
    { 0: bond(t) branch{ bond(t+); bond(t-);} ;
      0: bond(b) atom("H");
      2: bond(t) atom("H");
      3: bond(b);
      5: bond(b) atom("O",C,R) atom("Tos",L);
      5: bond(t) atom("H"); }
  }
  savecontext(#1)

  arrow(-45,,-|>){}

2.5 — Änderung von Parametern 61

  formula(TL,C,"p-Menth-3-en",HA,54)
  { ring(,,H5=){ 0: bond(r) branch { bond(r-); bond(r+); };
                 3: bond(r,<.); }
  }

  setcontext(#1,BL)
  arrow(-135,,->){ text(T,L){ formula(C,C){atom("E$_2$") } } }
  formula(TR,C,"p-Menth-2-en",HA,54)
  { ring(,,H4=){ 0: bond(r) branch { bond(r-); bond(r+); };
                 3: bond(r,<.); }
  }
\end{chemistry}

H

OTos H H 3-Menthyltosylat

H H

OTos H

p-Menth-3-en

E_2

p-Menth-2-en

Schema 2–18 Das Gleichgewicht zwischen den beiden Hauptkonformeren des Zyklohexans nutzt die verschiedenen Konformeren des Zyklohexans.

2.5 Änderung von Parametern

Als Beispiel soll im Schema 2– 19 der Hauptreaktionsweg durch verst¨arkte Linien hervorgehoben werden. Die L¨osung besteht darin, den Programmcode f¨ur diesen Weg in ein save/restore-Befehlspaar einzuschachteln und innerhalb dieser Klammer den Parameter $rLW f¨ur die Linienst¨arke zu ¨andern. Der seitlich abzweigende Nebenweg darf nicht innerhalb dieser Klammerung notiert werden:

\begin{chemistry}[param1]
  save(#1)
  set("rLW", 1.5)
  formula(L,BR)
  { ring(){ 3: bond(r) atom("O",C,R) atom("H",L); } }
  savecontext(#1)
  arrow(-45){}
  formula(L,R)
  { ring(){ 3: bond(r) atom("O",C,R) atom("H",L);

            0: bond(r) atom("N=N",L,R) bond(0)
               ring(,3,,,,r){}; }
  }

  restore(#1)

  setcontext(#1,BL)
  arrow(-135){}
  formula(R,R)
  { ring(){ 3: bond(r) atom("O",C,R) atom("H",L);
            4: bond(r) atom("N",C,R) atom("=N",L,R) bond(0)
               ring(,3,,,,r){}; }
  }
\end{chemistry}

OH

OH

N=N

OH N=N

Schema 2–19 Hauptreaktionen k¨onnen durch h¨ohere Linienbreite von Nebenwegen abgehoben werden. Die Linienbreite wird mit set eingestellt.

Wollen Sie alle Parameter mit einem gleichen Faktor multiplizieren, k¨onnen Sie anstelle mehrerer set-Befehle auch →scale einsetzen.

2.6 Elektronendarstellungen

Ein wesentlicher Bestandteil vor allem mechanistischer Formeldarstellungen ist die Wiedergabe der Elektronenkonfi gurationen und der Verschiebungen von Elektronen.

2.6.1 Elektronenkonfiguration

Die Wiedergabe der Elektronenkonfi guration an einem Atom ist zwar nicht unmittelbarer Bestandteil von OCHEM, sie l¨aß t sich jedoch mit dem L^ATEX-Paket echem.sty, das in der OCHEM-Distribution enthalten ist, erreichen. Es enth¨alt die beiden Makros \vd{<Atom>} und \vdd{<Atom>}, um neben dem genannten Atom ein resp. zwei Elektronen vertikal ¨ubereinanderzustellen. Die Elektronen werden durch Punkte symbolisiert. Das Atomsymbol dient nur zur Festlegung des Raumes, um den die Elektronen herum angeordnet werden sollen, es wird selber nicht dargestellt und

2.6 — Elektronendarstellungen 63

muß , wie die Beispiele zeigen, nach dem Makro folgen. Je nach der Reihenfolge des Atoms und der Makros erhalten Sie die Elektronen vor oder nach dem Atomsymbol:

Li, Li, Be, Be, B

\vd{Li}Li,
Li\vd{Li},
\vdd{Be}Be,
Be\vdd{Be},
\vdd{B}B\vd{B}

F¨ur eine horizontale Anordnung stehen vier Makros zur Verf¨ugung: f¨ur Elektronen oberhalb des Atomsymbols die Makros \hdu{<Atom>} und \hddu{<Atom>},f¨ur Elektronen unterhalb des Atomsymbols die Makros \hdl{<Atom>} und \hddl{<Atom>}:

Li, Li, C

\hdu{Li}Li,
\hdl{Li}Li,
\hddu{C}\hddl{C}C

F¨ur die Einbettung in eine OCHEM-Formel muß das Paket in einer package-Anwei-sung innerhalb einer chemspecial-Umgebung sowie im Dokument selber geladen werden:

O O

\usepackage{ochem,echem}
\begin{chemspecial}
package("echem")
\end{chemspecial}
...
\begin{chemistry}[elekt]
formula(L,R)
{
   bond(30;-30;30)
   branch { bond(90,=C)
            atom("\vdd{O}O\vdd{O}");
          }
   bond(-30;30;-30)
   atom("\hdl[\echhbar]{O}\hdu[\echhbar]{O}O",L,R)
   bond(30;-30)
}
\end{chemistry}

2.6.2 Elektronenverschiebungen

Als wichtiges Detail bei mechanistischen Betrachtungen treten kleine Pfeile innerhalb von Formeln oder -schemata auf, die die Wanderung von Elektronen veranschaulichen. Zwar ist dies fast ein Anwendungsfall f¨ur chemische CAD-Programme, ich habe jedoch versucht, einen sinnvollen Befehl f¨ur genau diese Elektronenverschiebungen zu implementieren. Da Start- und Zielpunkt nicht einfach durch eine Gerade verbunden werden k¨onnen, sind einige Parameter zur genauen Beschreibung des Bogens erforderlich. Grundlage des Verschiebungspfeils ist ein kubischer Spline, der neben Start- und Zielpunkt durch die Winkel, die die Tangenten an diesen beiden Punkten bilden, sowie zwei Kontrollpunkte beschrieben wird. Da die Aus- und Einfallswinkel meist feststehen, sind es vor allem die Kontrollpunkte (beziehungsweise deren Entfernung von den Zielpunkten), mit denen der Kurvenverlauf gesteuert werden kann. Hier ist jedoch Erfahrung und Gef¨uhl n¨otig, um nicht zu verzweifeln, weil die Kurve immer falsch geknickt ist :-)

Da die Zielpunkte der Verschiebungsbewegung meist in der Mitte von Bindungen liegt, die Bindungen selber aber in ganzer L¨ange dargestellt werden sollen, kann zum Speichern dieser mittleren Position, vom Startpunkt der Bindung ausgehend,

saveXY in seiner erweiterten Syntax mit Winkel und L¨angenangabe eingesetzt wer-

den. Diese Syntax gestattet es, die Koordinaten des Punktes zu speichern, der erreicht w¨urde, z¨oge man vom aktuellen Punkt eine Bindung mit dem angegebenen Winkel und der L¨ange. Da der aktuelle Punkt dabei nicht verschoben wird, kann sofort nach saveXY ein bond-Kommando folgen. Bei saveXY kommen entsprechend der Mittelposition h¨aufi g die Kleinbuchstaben als L¨angenangaben (n) in Betracht. Das Beispiel zeigt, wie Sie eine Mittelposition samt Bindung speichern und zeichnen:

saveXY(#1,30,n) % Winkel 30 Grad, halbe Normallaenge
bond(30) % Winkel 30 Grad, volle Normallaenge

Die Nummern, unter der die Punkte gesichert werden, sind Eingabeparameter f¨ur

emove. Start- und Zielpunkt k¨onnen innerhalb einer Formel liegen, m¨ussen aber

nicht, entscheidend ist in jedem Falle der aktuell unter der betreffenden Nummer gespeicherte Punkt (heiß t also, daß Sie Ihre saveXY-Kommandos konsequent fortlaufend numerieren sollten).

Betrachten Sie als Beispiel den kettenverl¨angernden Schritt der Isoprenoid-Biosynthese in Schema 2– 20. Es werden vier Punkte gespeichert entsprechend einer intramolekularen Verschiebung und einer intermolekularen Elektronenbewegung:

formula(L,R)
{ bond(30) branch { bond(90); } bond(-30,=; 30)
  saveXY(#2,-30,n) bond(-30)
  atom("O",L,R) saveXY(#4,0,n) bond(0) atom("PO(OH)$_2$",L)
}
emove(#2,-150,20,#4,-90,20)
space(R)
formula(L,R)
{ saveXY(#3,30,n)
  bond(30,=) branch { bond(90);} bond(-30; 30; -30)
  atom("OPO(OH)$_2$",L)
}
emove(#3,150,20,#2,30,20)

arrow(){}

formula(L,R)
{ bond(30) branch { bond(90); } bond(-30,=; 30; -30; 30)
  branch { bond(90); }
  bond(-30,=; 30; -30) atom("OPO(OH)$_2$",L)
}

O PO(OH)_2 OPO(OH)_2 OPO(OH)_2

Schema 2–20 Der kettenverl¨angernde Schritt der Biosynthese der Isoprenoide wird durch Elektronenverschiebungen verdeutlicht.

Ein weiteres Beispiel, die Friedel-Crafts-Acylierung, ist in Schema 2– 21 dargestellt. Auch hier werden intra- und intermolekulare Verschiebungen illustriert. Die shiftXY- Befehle dienen dazu, die drei Reaktanden in der richtigen relativen Lage zu positionieren. Durch die Einschachtelung in den Rumpf von multiline kann der anschließ ende Reaktionspfeil auf dieses Formeltriplet als Einheit bezogen werden:

multiline(1)
{ formula(C,C)
  { ring(,,H1=3=5=){ 3: saveXY(#1,-30,n); }
  }
  shiftXY(50,30)
  formula(C,C)
  { bond(30) saveXY(#2)
    branch { bond(90,=C) atom("O"); }
    saveXY(#3,-30,n) bond(-30) atom("Cl");
  }
  shiftXY(0,-30)
  formula(C,C)
  { saveXY(#4) atom("Al")
    branch{ bond(-45,,L) atom("Cl");
            bond(-90,,L) atom("Cl");
            bond(-135,,L) atom("Cl"); }
  }
  emove(#1,80,10,#2,150,15)
  emove(#3,-120,10,#4,90,10)
  ;
}

arrow(){}

formula(L,R)
{ ring(,,H1=5=)
  { 1: saveXY(#5,90,n);
    3: saveXY(#6) bond(90,s,n) atom("$\oplus$");
    4: bond(r) branch{ bond(r+,=C) atom("O"); }
       bond(r-);
    4: bond(t) atom("H");
  }
}
emove(#5,0,10,#6,-90,10)
bracket([)

arrow(0,10,<->){}

O Cl Al Cl Cl Cl ⊕ OH

Schema 2–21 Die Elektronenverschiebung zeigt deutlich den Verlauf der Friedel- Crafts-Acylierung eines Aromaten.

2.7 Beispiele

Sie haben nun alle Elemente, die in Formeln verwendet werden k¨onnen, kennengelernt. Zum Abschluß des Tutorials werden sie kombiniert, um zwei komplexe Reaktionsschemata zu programmieren.

2.7.1 Nylonsynthese

Schema 2– 22 zeigt die Darstellung der Ausgangsprodukte f¨ur die Nylon/Perlonsynthese. Die Idee dahinter ist, eine m¨oglichst lange Folge von Formeln und Pfeilen zu fi nden, die einfach beschrieben werden kann und die Zahl der ben¨otigten Verzweigunsstellen minimiert. Hiermit kann die ¨Ubersicht gewahrt werden. Ein anderer Ansatz k¨onnte sein, die zusammenh¨angenden Reaktionsstr¨ange nach chemischen Gesichtspunkten zusammenzufassen, was im vorliegenden Falle jedoch unm¨oglich ist, da zwei Startpunkte existieren: einer beim Benzen und ein anderer beim Butadien. Es w¨are nicht m¨oglich, die beiden Str¨ange in der gezeigten Anordnung zu einer gemeinsamen Formel zu f¨uhren. (Eine solche Zusammenfassung kann von joinh oder joinv geleistet werden, erfordert aber eine komplette Umgestaltung des ganzen Schemas.)

Der Hauptstrang beginnt beim Benzen und endet beim Butadien – aber halt, erfolgen die letzten Schritte nicht ” gegen die Fahrtrichtung“ ? Nicht aus der Sicht des Compilers, f¨ur den das Erscheinungsbild eines Pfeiles unerheblich ist. Reaktionen

” gegen den Strich“ werden eben nicht mit dem ->-Pfeil, sondern mit dem <--Pfeil formuliert.

Den Richtungs¨anderungen, die im Verlaufe des Hauptstranges auftreten, wird durch eine entsprechende Positionsparametrisierung der Formeln und der Pfeilwinkel Rechnung getragen.

Die Chlorierung des Butadiens liefert ein Gemisch zweier Dichlorbutene, die mit Hilfe eines einzeiligen multiline-Blockes zur Achse der vertikalen Reaktionsfolge zentriert werden k¨onnen. Das Pluszeichen ist eine verkappte ” Formel“ , die lediglich den Text ” +“ wiedergibt. Der Name, der unter beiden Formeln zentriert stehen soll, wird durch eine leere Formel erzeugt, die nur die Beschriftung enth¨alt.

Der Hauptstrang pr¨asentiert sich als Programm wie folgt:

\begin{chemistry}[nylon]
  formula(L,R,"Benzen",HR,24){ ring(){} }

  arrow()
  { text(T,C){ formula(C,C){ atom("H$_2$/Pt, Al$_2$O$_3$") } }
    text(B,C){ formula(C,C){ atom("36 bar") } }
  }

  formula(L,R){ ring(,,H){} }
  savecontext(#1)
arrow(,36)
{ text(T,C){ formula(C,C){ atom("O$_2$") } }
  text(B,C){ formula(C,C){ atom("Co(OAc)$_3$") } }
}

formula(L,B)
{ atom("HOOC",L,R) bond(30;-30;30;-30;30;-30)
  atom("COOH", L)
}
fbox

arrow(-90)
{ text(T,C){ formula(C,C){ atom("$\Delta$") } }
  text(B,C){ formula(C,C){ atom("NH$_3$") } }
}

formula(T,B)
{ atom("H$_2$NCO",L,R) bond(30;-30;30;-30;30;-30) atom("CONH$_2$", L) }

arrow(-90){}

formula(T,B)
{ atom("NC",L,R) bond(30;-30;30;-30;30;-30) atom("CN", L) }

arrow(-90)
{ text(T,C){ formula(C,C){ atom("H$_2$/Ni") } }
  text(B,C){ formula(C,C){ atom("130 $^o$C, 240 bar") } }
}

formula(T,L)
{ atom("H$_2$N",L,R) bond(30;-30;30;-30;30;-30) atom("NH$_2$", L) }
fbox savecontext(#2)

arrow(180,,<-){} nospace
arrow(-90,,-)
{ text(T,C){ formula(C,C){ atom("NH$_3$") } }
  text(B,C){ formula(C,C){ atom("H$_2$/Kat") } }
}

formula(T,B)
{ atom("NC",L,R) bond(30;-30;30,=;-30;30;-30) atom("CN", L) }

arrow(-90,,<-)
{ text(T,C){ formula(C,C){ atom("CuCN") } }
  text(B,C){ formula(C,C){ atom("CN$^\ominus$") } }
}

multiline(1,C,T,B)
{ formula(L,R)
  { atom("Cl") bond(30;-30;30,=;-30;30) atom("Cl")
  }
  formula(L,R){atom("+") }
  formula(L,R)
  { bond(30,=;-30) branch { bond(-90) atom("Cl"); }
    bond(30;-30) atom("Cl")
  };

68 Kapitel 2 — Tutorial

}
formula(T,B,"Chlorbuten",V,12){}

arrow(-90,,<-)
{ text(T,C){ formula(C,C){ atom("Cl$_2$") } } }

formula(T,B,"Butadien",V,24)
{ bond(30,=; -30; 30,=) }

Durch die Verwendung eines Hauptstranges verbleiben insgesamt zwei Verzweigungen, die zu den Endprodukten f¨uhren und keine Besonderheiten aufweisen:

  setcontext(#1,B)
  arrow(-90)
  { text(T,C){ formula(C,C){ atom("NOCl") } }
    text(B,C){ formula(C,C){ atom("HCl") } }
  }

  formula(T,B, "Zyklohexanonoxim",V,24)
  { ring(,,H){ 4: bond(r,=C) atom("NOH",L); } }

  arrow(-90)
  { text(T,C){ formula(C,C){ atom("H$_2$SO$_4$") } } }

  formula(T,B,"$\epsilon$-Caprolactam",V,24)
  { ring(,,H,,7,90)
    { 0: atom("N") bond(r) atom("H");
      1: bond(r,=C) atom("O"); }
  }
  fbox

  arrow(-90){}

  formula(T,B){ atom("Nylon 6") }

  setcontext(#2,B)
  arrow(-90)
  { text(T,C){ formula(C,C){ atom("270$^o$C") } }
    text(B,C){ formula(C,C){ atom("10 bar") } }
  }

  formula(T,B){ atom("Polyamid 66") }
\end{chemistry}

2.7.2 Terpen-Biosynthese

Die Biosynthese der Hauptvertreter bizyklischer Monoterpene ist als weiteres komplexes Beispiel vorgesehen und in Schema 2– 23 dargestellt. Es enth¨alt Verzweigungen, verschiedene Ringger¨uste sowie fett gedruckte Formeln.

save(#1) % save parameter
formula(L,R)
{ bond(30) saveXY(#1) bond(90,=; 150)

Benzen

H_2/Pt, Al_2O_3 36 bar

O_2 Co(OAc)_3

HOOC COOH

Δ NH3

H_2NCO CONH_2

NC CN

H_2/Ni 130^oC, 240 bar

H_2N NH_2 NH_3 H_2/Kat

NC CN

CuCN CN^⊖

Cl Cl + Cl Cl Chlorbuten

Cl_2

Butadien

NOCl HCl

NOH

Zyklohexanonoxim

H_2SO_4

N H O

ϵ-Caprolactam

Nylon 6

270^oC 10 bar

Polyamid 66

Schema 2–22 Die Nylonsynthese als Beispiel f¨ur ein komplexes Schema.

70 Kapitel 2 — Tutorial

  bond(90; 30) saveXY(#2) bond(-30,=)
  bond(-90; -45,s,n) atom("$\oplus$")
  restoreXY(#1) bond(-30) restoreXY(#2) bond(90)
}

arrow(){}

formula(L,R)
{ ring(,,H3=)
  { 0: bond(r) saveXY(#1) bond(r-)
       restoreXY(#1) bond(r+)
       restoreXY(#1) bond(r,s,n) atom("$\oplus$");
    3: bond(r); }
}
savecontext(#1)

arrow(){}
set("rLW",1.5)
formula(L,R,"Terpineol",HR,24)
{ ring(,,H3=)
  { 0: bond(r)
       branch{ bond(r-); bond(r+); }
       bond(r) atom("O",C,R) atom("H",L);
    3: bond(r); }
}
restore(#1)

% to thujene
setcontext(#1,B) arrow(-90,60){}
formula(T,R)
{ ring(,,H3=)
  { 0: bond(r) branch { bond(r-); bond(r+); };
    0: bond(r,s) atom("$\oplus$");
    3: bond(r); }
}

arrow(){}
formula(L,R)
{ ring(,,H)
  { 0: bond(r) branch { bond(r-); bond(r+); };
    3: bond(r);
    3: bond(45,s) atom("$\oplus$");
    4: saveXY(#2);
    0: bond(#2); }
}

arrow(){}
set("rLW",1.5)
formula(L,R,"3-Thujen",HR,24)
{ ring(,,H2=)
  { 0: bond(r) branch { bond(r-); bond(r+); };
    3: bond(r);
    4: saveXY(#2);
    0: bond(#2); }
}
restore(#1)

% to pinene
setcontext(#1,TR) arrow(60,36){}

formula(L,R)
{ ring(,,H)
  { 3: bond(r);
    3: bond(45,s) atom("$\oplus$");
    0: bond(90) branch { bond(30); bond(90,,n);
                         bond(-150,,n);};
  }
}

formula(L,R)
{ ring("bc311h")
  { 0: bond(t); 0: bond(b);
    1: bond(r); }
}

arrow(){}
set("rLW",1.5)
formula(L,R,"$\beta$-Pinen",HR,24)
{ ring(,,H)
  { 3: bond(r,=C);
    0: bond(90) branch { bond(30); bond(90,,n);
                         bond(-150,,n);};
  }
}
restore(#1)

% to camphene
setcontext(#1,T)
arrow(90,60){}

formula(B,R)
{ ring("bc221h")
  { 4: bond(r);
    3: bond(r,s,n) atom("$\oplus$");
    6: bond(t); 6: bond(b); }
}

arrow(){}
set("rLW",1.5)
formula(L,R,"Campher",HR,24)
{ ring("bc221h")
  { 4: bond(r);
    3: bond(r,=C) atom("O");
    6: bond(t); 6: bond(b); }
}
restore(#1)

% to carene
setcontext(#1,BR) arrow(-60,36){}

formula(L,R)
{ ring(,,H)
  { 3: bond(r);
    0: bond(t) saveXY(#1) bond(90; 45,s) atom("$\oplus$")

       restoreXY(#1) bond(-90) restoreXY(#1) bond(30);
  }
}

arrow(){}
set("rLW",1.5)
formula(L,R,"3-Caren",HR,24)
{ ring(,,H2=)
  { 3: bond(r);
    0: bond(t) saveXY(#1) bond(90)
       restoreXY(#1) bond(-90) restoreXY(#1) bond(30);
  }
}
restore(#1)

2.8 Weitere Verbindungsklassen

Viele Verbindungsklassen wurden in diesem Tutorial nicht behandelt. Die meisten von ihnen k¨onnen aber mit den besprochenen Mitteln erzeugt werden und bed¨urfen daher keiner Diskussion. Die Arbeit mit immer gleichen oder sehr ¨ahnlichen Formeln kann jedoch erheblich erleichtert werden, wenn man sich einen Satz von Makros schreibt, die auf diesen ¨Ahnlichkeiten basieren. Anregungen f¨ur solche Makros erhalten Sie im Abschnitt 3.6. Dort werden folgende Klassen von Stoffen n¨aher angesprochen:

• Steroide

• Isoprene

• Aminos¨auren und Oligopeptide

2.9 Entwicklung eigener Ringstrukturen

Um den Compiler nicht mit einer Vielzahl von denkbaren Ringstrukturen, die nur selten ben¨otigt werden, zu belasten, wurden nur die unter ring aufgelisteten Grundstrukturen in das Core aufgenommen. Es ist jedoch m¨oglich, in einem Perlmodul eigene Ringstrukturen zu defi nieren und dieses mit Hilfe von require in dem Dokument, das diese Strukturen ben¨otigt, nachzuladen.

Sie fi nden als Beispiel f¨ur den Aufbau eines solchen Moduls die beiden Dateien

mncyclib.pm und bicyclib.pm, die ¨uber den Grundwortschatz hinaus mono- und

bizyklische Ger¨uste bereitstellen. Sie k¨onnen diese Dateien, deren Listings in den Abschnitten ?? und ?? abgedruckt sind, als Templates benutzen; was Sie im wesentlichen in einer Strukturdefi nition bereitstellen m¨ussen, sind

• die Koordinaten der einzelnen Atome im Array @$XYa,

• die Bindungstypen (einfache, doppelte, projektive) im Array @$BT,

⊕ ⊕ OH

Terpineol

⊕

⊕

3-Thujen

⊕

β-Pinen

⊕ O Campher

⊕

3-Caren

Schema 2–23 Die Biosynthese der bizyklischen Monoterpene aus der geradkettigen Vorstufe Farnesol.

• eine Liste von Atomen, die als Startpunkt f¨ur Bindungen dienen, im Array @$BLa,

• eine Liste mit den Winkeln der Bindungen im Array @$rThetaa,

• eine Liste mit den Bindungsl¨angen im Array @$rLena,

• die Anzahl der Atome im Ger¨ust in $$iAnz,

• die Ringgr¨oß e, im allgemeinen die L¨ange einer typischen Ringkante in $$rLen,

• Information, ob ein Aromatenring gezeichnet werden soll, in $$bAromat sowie sein Radius in $$rRadius,

• gegebenenfalls Werte f¨ur die beiden freien Parameter.

F¨ur jedes Ger¨ust m¨ussen Sie zwei Funktionen bereitstellen:

<type> default : Diese Funktion gibt die Defaultwerte vor, die der Befehl ring nach auß en bereitstellt. Hierzu z¨ahlen vor allem die Anzahl der Atome und die Liste mit den Bindungstypen, da diese von einer eventuell folgenden Bindungsliste ¨uberschrieben werden k¨onnen sollen.

<type> : Diese Funktion enth¨alt die eigentliche Defi nition des Ger¨ustes. Alle Parameter, auch die beiden variablen Parameter <p1> und <p2> sind bekannt und k¨onnen benutzt werden.

Das Array @$XYa enth¨alt die Koordinaten der Atome in Punkt (pt), die das Ger¨ust aufbauen. Sie k¨onnen diese auf einen beliebigen Nullpunkt beziehen und entweder frei angeben oder – f¨ur geometrische Ger¨uste – berechnen lassen. Versuchen Sie dabei nach M¨oglichkeit, den L¨angenparameter in die Koordinaten einfl ieß en zu lassen, damit Ihre Struktur in der Gr¨oß e skalierbar ist! In den Grundger¨usten bezeichnet er die L¨ange einer Kante und bestimmt damit die Gesamtgr¨oß e.

Die Liste @$BLa enth¨alt eine Folge von Nummern der Atome, von denen Bindungen ausgehen. Die i-te Bindung ist mit der i-ten Atomnummer in dieser Liste verkn¨upft und wird durch die Daten beschrieben, die jeweils an i-ter Stelle in den einzelnen Eigenschaftenlisten stehen. Monozyklische Systeme werden durch eine einfache Liste @$BLa beschrieben, die aus den Zahlen von Null bis (Anzahl Atome−1) besteht, analog enthalten die Eigenschaftenlisten soviele Eintr¨age wie Atome und damit Bindungen zwischen ihnen vorhanden sind. Die Liste @$BT ist einfach aufzubauen und enth¨alt einfach die den Bindungstypen entsprechenden Nummern (k¨onnen in streambuf.pm bei der Funktion GetBond nachgelesen werden). Die anderen beiden Listen k¨onnen unter Umst¨anden Schwierigkeiten bei der erforderlichen Berechnung der Winkel und L¨angen bereiten. Sie k¨onnen jedoch die Funktionen

getangle(<p1>, <p2>)
getlength(<p1>, <p2>)

benutzen, um Winkel und L¨ange einer Bindung zwischen den Atomen mit den Koordinaten <p1> und <p2> zu bestimmen, wobei der Startpunkt der Bindung am Atom <p1> liegt.

getpos(<p1>, <angle>, <len>) kann dagegen den Endpunkt einer Bindung vom Punkt <p1> liefern, die eine gegebene Richtung und L¨ange hat. Je nachdem, ob Sie die Koordinaten der Ger¨ustatome kennen oder es anhand bekannten Bindungswinkel und -l¨angen aufbauen, kommen Sie mit einem der beiden Funktionens¨atze rasch zu den fehlenden Daten.

76

3. Alphabetische Befehlsreferenz

Dieses Kapitel enth¨alt die komplette Beschreibung aller Befehle und ihrer Syntax. F¨ur die einzusetzenden Werte gilt:

• Angaben wie Bindungstypen oder -l¨angen werden durch Symbole festgelegt, deren Aussehen der sp¨ateren Erscheinung m¨oglichst ¨ahnlich ist.

• Alle Winkel werden in Altgraden gemessen. Drehsinn ist der mathematisch negative (Uhrzeigersinn), Bezugsachse die positive x-Achse.

• Absolut angebbare L¨angen werden in Punkt (pt) ohne Einheit notiert.

Die meisten Parameter der Befehle sind optional und werden bei Auslassung mit einem Defaultwert belegt, der dem h¨aufi gsten Anwendungsfalle entspricht und die Schreibarbeit betr¨achtlich erleichern kann. Nach dem letzten konkret angegebenen Parameter kann bereits die schließ ende Klammer gesetzt werden, Notation der restlichen Parameter oder der trennenden Kommata ist unn¨otig. Soll ein sp¨ater in der Liste auftretender Parameter angegeben werden, ist es zur Kenntlichmachung seiner Position erforderlich, alle vorhergehenden Argumente ebenfalls zu notieren. M¨ochten Sie keine Werte einsetzen, ist die Angabe der Kommata ausreichend.

Es bedeuten in der Syntaxbeschreibung:

<fliste> := atom | bond | branch | orbital | saveXY | ring |
            save | restore | set

<sliste> := arrow | formula |
            savecontext | setcontext | gotoXY | shiftXY |
            emove |
            save | restore | set | scale
            bracket | fbox

3.1 Benutzerbefehle

Der folgende Abschnitt listet alle Befehl auf, die innerhalb der chemistry-Umge-bung (Einsatz im L^ATEX-Dokument) beziehungsweise des schema-Befehls (manuelle Erstellung des Programmfi les) auftreten k¨onnen.

3.1.1 arrow

arrow([ [<phi>] [, [<len>] [, [<type]]]])
{ [ text(T, <Tpos>){ <fliste> } ]
  [ text(B, <Tpos>){ <fliste> } ]
}
[ nospace ]

<Tpos> := L | C | R

Defaults: <phi> = 0
          <len> = rArrowExtend
          <type> = 1

Mit diesem Kommando erzeugen Sie Reaktionspfeile, mit denen die einzelnen Formeln zu Reaktionsketten verbunden werden. Die Gestalt des Pfeils wird von den Parametern <phi>, <len> und <type> bestimmt. Die Richtung des Pfeils wird mit den Winkel <phi> in Altgrad und im mathematisch positiven Sinne gegen¨uber der positiven x-Achse gemessen.

<type> legt die Art des Pfeils fest (Gleichgewichtspfeil, ” f¨uhrt zu“ -Pfeil, durchgestri-

chener Pfeil) und wird durch ein Symbol gem¨aß der folgenden Aufstellung spezifi - ziert:

-

-> -|>

<- <|-

<=> <->

Der Parameter <len> modifi ziert die L¨ange des Pfeils und gibt die Strecke an, die nach dem Start und vor der Spitze auf jeden Fall existieren muß . Die Mindestl¨ange eines Pfeils betr¨agt somit 2<len>. Die Angabe dieser Strecke ¨uber die halbe L¨ange wird verst¨andlich, wenn man die M¨oglichkeit, den Pfeil zu beschriften, ber¨ucksichtigt. Die Gesamtl¨ange eines Pfeiles bestimmt sich n¨amlich aus dem Anteil <len> vor der Beschriftung, der L¨ange der Beschriftung selbst und einem folgenden Anteil

<len>.

Die Beschriftung kann sowohl oberhalb wie auch unterhalb des Pfeils auftreten. Dem wird durch die Positionierungsparameter T und B im Befehl text Rechnung getragen. Da als Beschriftung sowohl Texte wie auch Formeln m¨oglich sind, wird anstelle von <fliste> eine formula-Beschreibung oder eine Reaktionssequenz erwartet. Normale Texte m¨ussen daher mit dem Befehl atom innerhalb einer Formeln gesetzt werden. Einige einfache Beispiele m¨ogen dies verdeutlichen:

Der Parameter <Tpos> wird bedeutsam, wenn Sie Formeln oder Texte sowohl ¨uber als auch unter den Pfeil setzen wollen. In diesem Falle k¨onnen Sie mit den Werten L,

C und R beide Beschriftungen linksb¨undig zueinander, zentriert oder rechtsb¨undig

ausrichten, wie dies im Schema 3– 1 zu sehen ist.

linksbündig

kurz zentriert

kurz rechtsbündig

kurz

Schema 3–1 Anordnung der oberen und unteren Beschriftung mit verschiedener Ausrichtung.

Im Schema 3– 2 sehen Sie, wie die Beschriftung bei verschiedenen Winkeln des Pfeils positioniert wird.

oben

unten oben

unten oben unten oben unten

oben

unten

oben unten oben unten oben unten

Schema 3–2 Die Anordnung der oberen und unteren Beschriftung bei verschiedenen Winkeln des Pfeils.

Wie bereits angesprochen, k¨onnen anstelle einfacher Texte auch Reaktionsschemata auf einem Pfeil aufgesetzt werden, wie das Beispiel des Schemas 3– 3 zeigt, in dem neben der Hauptreaktion auch die Darstellung des Reagenzes skizziert ist:

\begin{chemistry}[arrow2]
  formula(L,R)
  { ring(){} }

  arrow()
  { text(T,L)
    { formula(T,B){ bond(30; -30)atom("COOH",L) }
      arrow(-90)
      { text(T,L) { formula(C,C){ atom("SOCl$_2$") } } }
      formula(T,B){ bond(30; -30)atom("COCl",L) }
    }
  }

  formula(L,R)
  { ring()
    { 4: bond(r) branch { bond(r+,=C) atom("O"); }
         bond(r-; r); }
  }
\end{chemistry}

COOH

SOCl_2

COCl O

Schema 3–3 Komplette Reaktionsketten k¨onnen als ” Beschriftung“ auf Pfeile aufgesetzt werden.

Normalerweise wird nach einem Pfeil in der Richtung des Fortsetzungspunktes ein kleiner Zwischenraum gesetzt, um das Aneinanderpicken der einzelnen Elemente zu vermeiden. Ist dies aber beabsichtigt, k¨onnen Sie unmittelbar nach arrow den Befehl nospace einf¨ugen.

3.1.2 atom

atom("<Text>" [, [<pos>] [, [<Cpos>]]])

<pos> := C | L | R | T | B | TL | TR | BL | BR
<Cpos> := C | L | R | T | B | TL | TR | BL | BR

Defaults: <pos> = C
          <Cpos> = C

Mit diesem Befehl k¨onnen Sie an der aktuellen Position einen beliebigen TEX-Text einf¨ugen, der gem¨aß der Positionierungsvorgabe <pos> zum aktuellen Punkt ausgerichtet wird. Der aktuelle Punkt kann durch die Dimension der Textbox beeinfl uß t werden, die Positionierung des Fortsetzungspunkte wird durch den Parameter

<Cpos> bestimmt.

Nachfolgend sind einige Formeln dargestellt, in denen die verschiedenen M¨oglichkeiten, Atomsymbole zu verwenden, illustriert sind. Im einfachsten Falle, einem terminalen Atom, k¨onnen Sie die Positionierungen C verwenden, um das Symbol zentriert zum Endpunkt der Bindung zu setzen. Bei l¨angeren Texten wie COOH ist es empfehlenswerter, je nach Richtung der Bindung eine links- oder rechtsb¨undige Anordnung vorzunehmen. Da in beiden F¨allen keine Fortsetzung von diesem Atom ausgeht, ist die Angabe <Cpos> unerheblich und kann fortgelassen werden. Erst wenn von diesem Atomsymbol aus weitere Bindungen abzweigen, gewinnt dieser Parameter seine Bedeutung, wie Sie an der Methylengruppe sehen, die von rechts erreicht wird (<pos> ist R) und von der aus eine Bindung nach links weiterf¨uhrt (<pos> ist L):

O O O Cl formula(L,R)
{ ring(,,H)
  { 3: bond(r,=C) atom("O");
    4: bond(r) atom("O") bond(r-; r)
       atom("O") bond(r-);
    5: bond(r) atom("Cl");
  }
}

COOH

HOOC

HOOC

HOOC COOH

COOH formula(L,R)
{ ring()
  { 0: bond(r) atom("COOH", L);
    1: bond(r) atom("HOOC", R);
    2: bond(r) atom("HOOC", BR);
    3: bond(r) atom("HOOC", R);
    4: bond(r) atom("COOH", L);
    5: bond(r) atom("COOH", TL);
  }
}

Sie sehen, daß lange Textst¨ucke wie ” COOH“ unsch¨on plaziert werden, im Tutorial auf Seite 16 k¨onnen Sie erfahren, wie Sie in solchen F¨allen besser vorgehen.

CH_2CH_2 formula(L,R)
{ ring()
  { 2: bond(r) atom("CH$_2$CH$_2$", R, L)
       bond(-150) ring(,4){};
  }
}

Da die Texte und Atomsymbole durch TEX gesetzt werden, sind Sie nicht auf Buchstaben beschr¨ankt, sondern k¨onnen auch mathematische Formatierungshilfsmittel einsetzen, um Indizes, die Symbole ⊕ und ⊖ sowie griechische Zeichen zu erhalten. Auch benutzerdefi nierte Befehle k¨onnen als Text auftreten, in diesem Fall m¨ussen die Defi nitionen allerdings in einer Paketdatei gesammelt werden. Das Vorgehen ist in Abschnitt 3.2.2 beschrieben.

3.1.3 bond

bond(<phi> [, [<btype>] [, [<len>]]])
bond([<phi>] [+|-] r|t|b|l|v [+|-] [, [<btype>] [, [<len>]]])
bond(#<n> [, [<btype>] ]])

bond(<args1...>; <args2...>; ... )
<btype> := - -> <- <-> <. << 3 t s >> >. p b~
<len> := S | N | L | s | n | l [...]

Defaults: <btype> = 0
          <len> = N

Dieses Kommando erzeugt das grundlegende graphische Primitiv ” Bindung“ . Die Syntaxvarianten decken verschiedene Anwendungsf¨alle ab.

Normalform

Sie k¨onnen dabei mit dem Parameter <phi> den Winkel der Bindung gegen¨uber der positiven x-Achse vorgeben, mit Hilfe von <btype> zwischen mehreren Bindungsformen w¨ahlen sowie mit <len> die L¨ange der Bindung festlegen. Die Bindung beginnt am aktuellen Startpunkt und verschiebt den aktuellen Endpunkt, der zum folgenden Startpunkt wird, um ihre eigene L¨ange und Richtung, sodaß Sie mehrere

bond-Befehle aneinanderh¨angen k¨onnen, um Atomketten zu bilden oder einfach

– in der komprimierten Syntax – mehrere Beschreibungen mit Semikola verketten k¨onnen.

Bindungstypen

In allen Syntaxvarianten werden die verschiedenen Typen von Bindungen gem¨aß der folgenden Aufstellung mit Hilfe von Symbolen f¨ur <btype> ausgew¨ahlt (alle Bindungen sind mit einem Winkel von 90^o gezeichnet):

- t p b s << <. >> >. o =C =U = 3 -> <-~

- : Diese Bindungsform entspricht der normalen Bindung, die meisten in den Formeln verwendet wird.

=, =C, =U : Doppelbindungen k¨onnen mit drei verschiedenen Typen gesetzt werden, die sich in der Lage der einzelnen Bindungen unterscheiden. Die normale Form ist = und wird eingesetzt, wenn die zweite Bindung im Uhrzeigersinn rechts von der Hauptbindung liegen soll. Innerhalb von (Benzen-) Ringen ist dies die Form, die ein ausgewogenes Aussehen aufweist, da die kleineren zweiten Bindungen innerhalb des Ringes liegen. =U erzeugt eine zweite Bindung links der Hauptbindung und wird innerhalb von Ringen benutzt, die gegen den Uhrzeigersinn gezeichnet werden, damit die Bindung wiederum im Inneren des Ringes liegt. =C schließ lich zeichnet zwei gleichwertige Bindungen, die zentriert zu einer gedachten Mittellinie liegen. Sie eignet sich nicht f¨ur Ringbindungen, wird aber benutzt, wenn eine Doppelbindung zu einem terminalen Atomsymbol f¨uhrt. Die Unterschiede der einzelnen Formeln werden in den folgenden Formeln sichtbar:

gotoXY(0,0)
  formula(C,C){ ring(,,H0=2=4=){} }
gotoXY(50,0)
  formula(C,C){ ring(,,H0=U2=U4=U){} }
gotoXY(100,0)
  formula(C,C){ ring(,,H0=C2=C4=C){} }

3 : Diese Form entspricht einer Dreifachbindunng.

t, p, << : Hiermit wird eine dickere Bindung (t), eine dickere Bindung mit weiß em Rand (p) oder eine sich verbreiternde Bindung (<<) gezeichnet. Die Formen t und << eignen sich gleichermaß en zur Darstellung von Bindungen aus der Papierebene heraus, wie das Beispiel zum n¨achsten Stichwort zeigt. Bindungen mit t sind dar¨uberhinaus in polyzyklischen Systemen geeignet zur Darstellung der r¨aumlich aus dem Papier ragenden Verbr¨uckungen. Zur Erh¨ohung dieses perspektivischen Eindrucks zeichnet

p noch einen weiß en Rand um die breitere Bindung. An der folgenden Verbindung

ist dies dargestellt, wobei p nur bei den Bindungen eine Wirkung zeigt, die andere Bindungen schneiden:

formula(L,R)
{ ring(,4,H){}
  ring(,2,H){}
  ring(,0,H)
  { 1: bond(180,t,S; -90,t; -30,t; 0,t,S); }
}
formula(L,R)
{ ring(,4,H){}
  ring(,2,H){}
  ring(,0,H)
  { 1: bond(180,t,S; -90,p; -30,t; 0,t,S); }
}

Wenn Sie diesen Trick anwenden, m¨ussen Sie beachten, daß die Reihenfolge der Bindungen entscheidend ist, Bindungen mit Aussparungen (Typ p) m¨ussen immer zuletzt gezeichnet werden! Im obigen Beispiel darf die Reihenfolge, in der die Ringe gezeichnet werden, nicht ge¨andert werden, wenn die perspektivische Wirkung erhalten bleiben soll.

<., o : Diese Bindungsformen entsprechen einer gestrichelten, sich verbreiternden Bindung beziehungsweise einer gestrichelten Bindungen und eignen sich alle zur Darstellung von Bindungen in die Papierebene hinein:

formula(C,C)
{ bond(30)
  branch { bond(120,<<); bond(60,<.); }
  bond(-30)
  branch { bond(-120,t); bond(-60,o); }
  bond(30)
}

Die fein gepunkteten Linie o eignet sich auch dazu, Bindungen in statu nascendi oder Wasserstoffbr¨uckenbindungen anzudeuten:

formula()
{ ring(,,H0=2=4=U3o5o)
  { 4: bond(r);
    5: bond(r); }
}

>>, >., b : Diese Bindungsformen entsprechen sich verj¨ungenden ausgef¨ullten, sich verj¨ungenden gestrichelten oder breiten Bindungen und eignen sich zur Darstellung von Bindungen eines Atom ¨uber der Papierebene in diese hinein. Die Breite der Bindung vom Typ b (einstellbar mit dem Parameter rBW) entspricht der Maximalbreite der sich verbreiternden oder verj¨ungenden Bindungen, sodaß alle drei Formen leicht kombiniert werden k¨onnen, ein Beispiel ist die r¨aumliche Wiedergabe nach HAY-

WORTH: O formula(C,C)
{ ring("furanose",0,1<<2b3>>,L)
  { 0: atom("O");
  }
}

~ : Diese Bindungsformen wird benutzt, wenn die Stereochemie an einem Atom nicht weiter spezifi ziert werden soll oder ein Anomerengemisch vorliegt, wie ein Beispiel aus der Zuckerchemie zeigt: O OH formula(C,C)
{ ring("furanose",,,L)
  { 0: atom("O");
    1: bond(0,~) atom("OH",L);
  }
}

s : Diese unsichtbare Bindung kann benutzt werden, um Punkte in der Umgebung eines Atoms zu erreichen. Ein Beispiel w¨are die Numerierung eines Atomger¨ustes:

O 1 2

4 formula(C,C)
{ ring(,,H)
  { 0: atom("O");
    0: bond(r,s) atom("1");
    1: bond(r,s) atom("2");
    3: bond(r,s) atom("4"); }
}

Weiterhin k¨onnen in einem Ringger¨ust einige Bindungen ausgespart werden, wie etwa das Zyklodekan zeigt:

formula(C,C)
{ ring(,4,H4s)
  { }
  ring(,2,H1s)
  { }
}

->, <- : Diese Bindungsform eignet sich zur Darstellung von Komplexbindungen mit Angabe des Atoms, das das Elektronenpaar bereitstellt: Cu formula(C,C)
{ ring(,,H0<-4->,,5,0)
  { 0: atom("Cu", L); }
}

Bindungslängen

Die L¨ange der Bindung wird durch die Buchstaben S (short), N (normal) und L (long) kodiert. Ein Kleinbuchstabe kodiert eine L¨ange, die halb so groß wie die des entsprechenden Groß buchstabens ist. Jeder h¨ohere Buchstabe bedeutet eine um 50% erh¨ohte L¨ange. Das mehrfache Erscheinen eines Buchstabens bedeutet, daß die Bindung um ein entsprechendes Vielfaches der L¨ange dieses Buchstabens verl¨angert wird, beispielsweise bedeutet NN die doppelte Normall¨ange. Als Folge hiervon k¨onnen einige L¨angen auf mehrere Arten dargestellt werden: nn ist dasselbe wie N, NNN dasselbe wie LL. Die tats¨achliche L¨ange der einzelnen Bindungen ist in den Variablen

rLenS, rLenN und rLenL gespeichert (→set). Die besondere L¨angenangabe 0 er-

zeugt eine Bindung der L¨ange Null, was sinnvoll sein kann, wenn Sie ein allgemeines Makro geschrieben haben und eine der Bindungen nicht immer vorhanden sein soll.

r/t/b/l/v-Form

Diese Syntax ist mit der Normalform in allen Punkten identisch, lediglich die Winkelangabe ist abweichend. Die Syntax gibt vor allem eine Hilfestellung beim Ansatz von Seitenketten an Ringsysteme und ist demzufolge innerhalb eines ring- oder

vertex-Befehls am sinnvollsten einsetzbar. Die Buchstaben stellen dann eine sym-

bolische Winkelangabe dar, deren konkreter Wert vom Ringsystem abh¨angig ist.

r : Eine radiale Bindung, also eine Bindung, deren Richtung durch die Gerade Zentrum-

Ringatom bestimmt ist.

t, b : Winkel f¨ur Bindungen ober- und unterhalb der Papierebene (top und bottom). Be-

sonders beim Ringtyp chair und verschiedenen Konformerenger¨usten ist diese Bedeutung offensichtlich, bei rein planaren Formeln (ebene Ringe) werden f¨ur diese Bindungen h¨aufi g die Verl¨angerungen der an das Ringatom anstoß enden beiden Kanten verwendet. Welche Winkel f¨ur diese Variante tats¨achlich benutzt werden, wird in Form von schematischen Formeln bei der Besprechung der einzelnen Ringe (Befehl ring und Vorstellung der Bibliotheken) gezeigt.

                                              formula(L,R)
                                              { ring(,,H,,5,90)
                                                { 0: bond(r);
                                                  2: bond(t,t);
                                                  2: bond(b,o); }
                                              }

v(<n>) : Winkel, unter dem die Bindung vom Ringatom mit der Nummer <n> zum n¨achsten

Ringatom gezeichnet wird, entspricht also dem Winkel der Ringkante <n>.

l : Entspricht dem Winkel, unter dem die letzte Bindung gezeichnet wurde.

Diese Grundwinkel k¨onnen durch verschiedene Symbole modifi ziert werden, die in dieser Reihenfolge auftreten und kombiniert werden k¨onnen:

/ : Diese Angabe verwandelt den Winkel eines r-, t- oder b-Winkels in seinen Kom-

plement¨arwinkel. Dies ist der Winkel, der dazu f¨uhrt, daß eine Kette mit der Winkelfolge r r/ r r/ r in Zickzackform verl¨auft und sich dabei an der horizontalen respektive vertikalen Koordinatenachse orientiert. Der Innenwinkel der Knickstellen entspricht dann stets 60^o.

                                        formula(L,R)
                                        { ring()
                                          { 1: bond(r; r/; r; r/);
                                            2: bond(r; r/; r; r/);
                                            3: bond(r; r/; r; r/);
                                            4: bond(r; r/; r; r/);
                                            5: bond(r; r/; r; r/); }
                                        }

t : Hiermit erzeugen Sie einen Winkel, der dem /-Winkel gerade entgegengesetzt ist

und so eine Verzweigungsbindung hervorbringt. Dieser Winkel tritt somit ¨ublicherweise innerhalb von branch auf. In der Kombination /t erhalten Sie den Abzweigwinkel zur ” anderen“ Seite der Zickzackkette:

                                          formula(L,R)
                                          { ring()
                                            { 1: bond(r) branch{ bond(rt); }
                                                 bond(r/) branch{ bond(r/t); }
                                                 bond(r);
                                              3: bond(r) branch{ bond(rt); }
                                                 bond(r/) branch{ bond(r/t); }
                                                 bond(r);
                                              4: bond(r) branch{ bond(rt); }
                                                 bond(r/) branch{ bond(r/t); }
                                                 bond(r); }
                                          }

+, - : Diese Zeichen k¨onnen mehrfach auftreten und erh¨ohen beziehungsweise vermin-

dern den Zeichenwinkel um jeweils 60^o. Dies ist der ¨ubliche Winkel von Alkanen in Zickzackform oder der Hauptwinkel bei Sechsringen. Die Wirkung entspricht einer Drehung aus der augenblicklichen Richtung nach links beziehungsweise nach rechts. Sie erzeugen damit ¨Anderungen in der Kettenrichtung oder Isopropylverzweigungen:

formula(L,R)
{ ring()
  { 2: bond(r) branch { bond(r+); bond(r-); };
    3: bond(r) branch { bond(r+); bond(r-); };
    5: bond(r) branch { bond(r+); bond(r-); };
  }
}

F¨ur l¨angere Ketten mit Methylverzweigung ist die Bestimmung des Abzweigwinkels mit rt oder r/t besser geeignet.

| : Dieses Zeichen spiegelt die Bindung an der vertikalen Achse. Sie k¨onnen so einfach

symmetrisch gebaute Molek¨ule erzeugen:

formula(L,R)
{ ring()
  { 4: bond(r; r+; r|)
       ring(,0,,,,r)
       { 3: bond(r); };
  }
}

Es ist m¨oglich, eine Summe aus einem konkreten Winkel <phi> und dem durch die aufgef¨uhrten symbolischen Angaben erhaltenen Winkel zu bilden, um zum Beispiel um 10^o von einer radialen Bindung abzuweichen. Der Operator + bildet die gew¨ohnliche Summe der Winkel, - dagegen liefert die Summe aus dem konkreten Winkel und der dem symbolischen Winkel entgegengesetzten Richtung.

- kann auch ohne konkrete Zahlangabe benutzt werden, um Bindungen entgegen

einer Richtung zu zeichnen:

N=N NH_2 formula(L,R)
{ ring()
  { 5: bond(r) atom("N=N",L,R) bond(-r|)
       ring(,0,,,,r)
       { 3: bond(r) atom("N",C,R)
                    atom("H$_2$",L); };
  }
}

Eine ausf¨uhrliche Besprechung der einzelnen Varianten sowie zahlreiche Beispiele sind im Tutorial, Abschnitt 2.2.1 gegeben. Es folgt hier eine kurze Zusammenfassung einiger Richtungen, die durch Kombination verschiedener Winkelsymbole erhalten werden k¨onnen.

                 r
          -r

          r|
                 -r|

             r

              r+
       -r+
         r-

             -r-

  r r/
r\ -r\

-r/

                          r

                               -r

                               r|

                        -r|

                          r

                         r+

                               -r+

                             r-

                             -r-

                                           r

                                           r/

                                         r\

                                         -r\

                                    -r/

Zu Beginn einer Formel werden diese symbolischen Angaben mit Defaultwinkeln assoziiert, und zwar 30^o f¨ur r, 90^o f¨ur t und -90^o f¨ur b. Sie k¨onnen diese Angaben somit auch auß erhalb eines Ringsystems verwenden:

formula(L,R)
{ bond(t; r; r/; r,=; r/; b) }

Um die Startwinkel zu ver¨andern, k¨onnen Sie →set einsetzen:

                                                              formula(L,R)
                                                              { set("iAngleR",30)
                                                                set("iAngleT",60)
                                                                set("iAngleB",-60)
                                                                bond(t; r; r/; r,=; r/; b)
                                                              }

#-Form

Die Form mit #<n> (auch als #-Syntax bezeichnet) dient zur Erzeugung einer Bindung, deren Richtung und L¨ange nicht spezifi ziert wird, sondern die zu einem zuvor mit saveXY gespeicherten Atom f¨uhrt. Sie k¨onnen damit das aktuelle Atom mit dem Atom <n> verbinden. Beispiele sind bei →saveXY und →restoreXY zu fi nden.

Verkettete Bindungslisten

Die Bindungsbeschreibungen mehrerer bond-Befehle k¨onnen, durch Semikola getrennt, in einem einzigen bond-Kommando zusammengefaß t werden und dadurch viel Schreibarbeit sparen. Es sind dabei alle Syntaxvarianten sowie Kombinationen m¨oglich. Einige ¨Aquivalenzen:

bond(30) bond(0,=,L) bond(-30) bond(30)
 == bond(30; 0,=,L; -30; 30)

bond(r) bond(r/) bond(r)
 == bond(r;r/;r)

3.1.4 bracket

{formula | multiline} bracket() | bracket([) | bracket(])

Mit Hilfe dieses Befehls f¨ugen Sie eckige Klammern um die letzte mit formula erzeugte Formel oder mit multiline erzeugten Formelblock. Wird dieser Befehl nach einem anderen als den angegebenen Kommandos ausgef¨uhrt, k¨onnen die Ergebnisse von den Erwartungen abweichen! Sie k¨onnen angeben, ob eine linke oder rechte Klammer gew¨unscht wird oder ob – mit einer leeren Parameterliste – beide Klammern erscheinen sollen. Die Klammern dienen vor allem zum Zusammenfassen mesomerer Grenzstrukturen oder eines Zwischenproduktes (-gemisches).

Um N¨aheres ¨uber die Klammersetzung in komplexen Zusammenh¨angen zu erfahren, lesen Sie bitte Abschnitt 2.3.5 des Tutorials.

3.1.5 branch

branch{ <aliste_1>;
        <aliste_2>;
        ...
      }

Mit diesem Befehl k¨onnen Sie ausgehend von der aktuellen Position, Verzweigungen aufbauen. Jede Formelkette <aliste> beginnt aufs Neue am Verzweigungspunkt. Als Beispiel m¨ogen folgende Formeln dienen, in denen vom zentralen Kohlenstoffatom je drei Ketten abzweigen. Da der Befehl nach seinem Abschluß den Startpunkt unver¨andert l¨aß t, m¨ussen nur zwei Ketten im Rumpf von branch formuliert werden:

Cl H D Br formula(L,R)
{ atom("Cl") bond(30,,L)
  branch { bond(90,,L) atom("H");
           bond(-20,<<,L) atom("D");
         }
  bond(-70,<.,L) atom("Br")
}

Cl H Br D formula(L,R)
{ atom("Cl") bond(30,,L)
  branch { bond(90,,L) atom("H");
           bond(-20,<<,L) atom("Br");
         }
  bond(-70,<.,L) atom("D")
}

Dieser Befehl eignet sich durch seine hierarchisch-rekursive Arbeitsweise hervorragend f¨ur strukturiert denkende Mitmenschen; geh¨oren Sie dagegen zu den Freigeistern, wird Ihnen die M¨oglichkeit, mit →saveXY Punkte zu speichern und sp¨ater darauf zur¨uckzukommen (→restoreXY, →bond), sehr willkommen sein, da so Verzweigungen ohne zus¨atzliche Schachtelungsebenen, daf¨ur mit hintereinander auftretenden linearen Abschnitten, simuliert werden.

3.1.6 joinh

joinh(<n>, <pos>)
{ <rliste_1>;
  ...
  <rliste_n>;
}

<pos> := L | C | R

Der Befehl dient zum Zusammenf¨uhren von <n> horizontal laufenden Reaktionsketten. Die einzelnen Ketten werden so angeordnet, daß sie in einem gleichen Abstand ¨ubereinander verlaufen. Der Parameter <pos> legt fest, ob sie dabei linksb¨undig, zentriert oder rechtsb¨undig zueinander stehen. Eventuell auftretende Formelbeschriftungen werden dabei mit einem konstanten Abstand von der Mittellinie gesetzt, sodaß der vertikale Mindestabstand dem Wert des Parameters rTextSep entspricht. Aus diesem Grunde d¨urfen beschriftete Formeln keine Abstandsparameter besitzen! Formeln, die keine Beschriftung aufweisen, beeinfl uß en durch ihre Gr¨oß e den Textabstand nicht.

Der Anschluß punkt f¨ur weiterf¨uhrende Reaktionspfeile liegt auf der rechten Seite in der Mitte des Formelblockes. Das nachfolgende Schema zeigt als Beispiel eine Synthese, bei der zwei getrennte Reaktionsketten zum Addukt zusammengef¨uhrt werden. Typischerweise folgt nach joinh ein arrow-Befehl, da joinh keine Pfeilspitzen zeichnet (die beim Ineinanderschachteln mehrerer joinh-Befehle hinderlich w¨are). Durch die Positionierung L beginnen alle Teilketten linksb¨undig:

joinh(2,L)
{ % upper line
  formula(L,R,"Propions""aure"){ bond(30;-30;30) atom("COOH",L) }
  arrow(,12){}
  formula(L,R){ bond(30;-30;30) atom("COCl",L) }
  ;
  % lower line
  formula(L,R,"Benzen"){ ring(){} }
  ;
}
arrow(,12){}
formula(L,R,"Propionylbenzen",HR,24)
{ ring()
  { 4: bond(r) branch{ bond(rt,=C) atom("O"); }
       bond(r/;r); }
}

COOH

Propionsäure

COCl

Benzen

O

Propionylbenzen

3.1.7 joinv

joinv(<n>, <pos>)
{ <rliste_1>;
  ...
  <rliste_n>;
}

<pos> := T | C | B

¨Ahnlich →joinh f¨ur horizontale Ketten, f¨uhrt dieser Befehl <n> vertikal von oben nach unten laufende unabh¨angige Reaktionsketten an einem Punkt in der Mitte der Unterkante des gesamten Blockes zusammen, von dem aus ein Reaktionspfeil nach unten angesetzt werden kann. Die einzelnen Ketten werden gem¨aß dem Parameter <pos> mit ihren Oberkanten, Mitten oder Unterkanten zueinander ausgerichtet. Die einfache Anwendung des Befehls erfolgt analog zu →joinh, das dortige Beispiel pr¨asentiert sich in der vertikalen Form verm¨oge T b¨undig zur oberen Kante:

COOH

Propionsäure

COCl

Propionylchlorid

Benzen

O

Propionylbenzen joinv(2,T)
{ % left line
  formula(T,B,"Propions""aure",V,24)
  { bond(30;-30;30) atom("COOH",L) }
  arrow(-90,12){}
  formula(T,B,"Propionylchlorid",V,24)
  { bond(30;-30;30) atom("COCl",L) }
  ;

  % right line
  formula(T,B,"Benzen",V,24)
  { ring(){} }
  ;
}

arrow(-90){}

formula(T,B,"Propionylbenzen",V,24)
{ ring()
  { 4: bond(r) branch{ bond(rt,=C) atom("O"); }
       bond(r/;r); }
}

3.1.8 emove

emove(#<i>, <iAngle>, <iCtrl>, #<j>, <jAngle>, <jCtrl>)

(Nur mit PostScript-Ausgabe verf¨ugbar!) Der Befehl erzeugt kleine gerade oder gebogene Pfeile, mit denen Wanderungsbewegungen von Elektronen dargestellt werden. Pfeile k¨onnen innerhalb einer Formel verlaufen oder Punkte in verschiedenen Formeln miteinander verbinden. Der Befehl emove selbst muß wie formula oder

arrow auß erhalb eines Formelrumpfes stehen.

Der Start- und Endpunkt der Pfeile wird von Punkten gebildet, die mit Hilfe zweier saveXY-Befehle gespeichert wurden, die zugeordnete Nummer tritt hier als Parameter <i> beziehungsweise <j> in Erscheinung. Es wird ein Abstand der Gr¨oß e

rEmove zwischen Pfeil und spezifi ziertem Punkt eingef¨ugt, damit der Pfeil nicht di-

rekt an der Bindung klebt. Die genaue Gestalt der Pfeile wird durch die beiden Winkelangaben <iAngle> und <jAngle> sowie zwei reelle Zahlen <iCtrl> und

<jCtrl bestimmt. Die Winkel geben an, unter welchem Winkel Pfeilende und -

spitze die spezifi zierten Punkten verlassen beziehungsweise erreichen. Diese Winkel sind den Angaben bei bond analog. Die Linien, die in Verl¨angerung die Startund Endpunkte unter den gegebenen Winkeln verlassen, k¨onnen als Tangenten an den Pfeilbogen angesehen werden. Auf ihnen liegen, in den Abst¨anden <iCtrl> beziehungsweise <jCtrl>, zwei Kontrollpunkte.

Die Menge der Start-, End- sowie der beiden Kontrollpunkte steuert dann den genauen Verlauf eines PostScript-Splines (kubischer Spline), an dessen Ende eine kleine Pfeilspitze aufgesetzt ist.

Will man eine konkrete Elektronenbewegung zeichnen, ist die Wahl der Winkel meist kein Problem, da sie durch die geplanten Aus- und Einfallwinkel bestimmt werden. Fingerspitzengef¨uhl und Erfahrung erfordert dagegen die Wahl der Kontrollparameter, da sehr leicht zu krumme oder zu gerade oder falsch gebogene B¨ogen zustandekommen.

Einige Beispiele sollen den Einfl uß der Kontrollparameter zeigen:

formula(L,R)
{ saveXY(#1,30,n) bond(30,=)
  saveXY(#3,-30,n) bond(-30)
  saveXY(#2,30,n) bond(30,=)
}
emove(#1,90,1,#2,120,1)
emove(#1,-60,1,#3,-120,1)

emove(#1,90,5,#2,120,5)
emove(#1,-60,5,#3,-120,5)

emove(#1,90,10,#2,120,10)
emove(#1,-60,10,#3,-120,10)

emove(#1,90,20,#2,120,10)
emove(#1,-60,20,#3,-120,20)

Sie erkennen an den Beispielen zweierlei: a) der Einfl uß der Kontrollparameter auf die Kurvenkr¨ummung ist umso st¨arker, je gr¨oß er diese sind, das heiß t je weiter entfernt die Kontrollpunkte sind. Die vier Steuerungspunkt bilden ein Viereck, innerhalb dessen die Splinekurve vollst¨andig liegt, wobei jeder Kontrollpunkt die Kurve gleichsam an sich ziehen m¨ochte. Je weiter er entfernt ist, desto mehr wird auch die Kurve in seine Richtung gebogen. Am vierten Beispiel k¨onnen Sie sehen, daß auf diese Weise sogar Verschlingungen entstehen k¨onnen. b) Die meisten Startpunkte von Elektronensplines liegen in der Mitte von Bindungen. Hier vereinfacht der Einsatz der erweiterten saveXY-Syntax die Formelbeschreibung erheblich, da mit nur zwei Befehlen der Bindungsmittelpunkt gesichert und die Bindung selbst gezeichnet werden kann. Zur Angabe der halben Bindungsl¨ange wird die L¨angenangabe n statt

N bei normalen Bindungen eingesetzt. Im Tutorial, Abschnitt 2.6, werden konkrete

Anwendungsf¨alle beschrieben.

3.1.9 fbox

{formula | multiline} fbox

Mit diesem Befehl erzeugen Sie einen Rahmen um die zuletzt mit formula gezeichnete Formel oder mit multiline erzeugten Formelblock.

N¨aheres ¨uber die Rahmung komplexer Zusammenh¨ange ist in Abschnitt 2.3.5 des Tutorials zu erfahren.

3.1.10 formula

formula(<pos>, <Cpos>, ["<Text>", HA|HR|V, <dy>])
{ <fliste> }
[ nospace ]

<pos> := C | L | R | T | B | TL | TR | BL | BR
<Cpos> := C | L | R | T | B | TL | TR | BL | BR

Defaults: <pos> = L
          <Cpos> = R

Dies ist der zentrale Befehl des Chemiepaketes. Mit ihm bilden Sie anhand grundlegender Elemente Formeln, die mit diesem Befehl zu einer festen Einheit verwandelt werden, die automatisch bez¨uglich weiterer Formeln plaziert wird und in Zusammenhang mit Pfeilen zu Reaktionsschemata f¨uhrt. Auch weitere abstraktere Plazierungsbefehle wie →multiline oder →joinh erfordern Formeln als Basis.

Die Formel selbst wird im Rumpf des Befehls <fliste> durch eine Folge von Grundelementen (→atom, →bond, →ring, →branch) gebildet. Diese Einheit wird so plaziert, daß sie in der Relation <pos> zum Startpunkt steht. Der Anschluß punkt f¨ur die folgenden Formeln wird durch den Parameter <Cpos> bestimmt. Die Bedeutung der Parameter wird im Tutorial detailliert gezeigt.

Normalerweise wird nach einer Formel in der Richtung des Anschluß punktes ein Zwischenraum gesetzt, damit die einzelnen Elemente nicht direkt aneinanderh¨angen. Ist ebendies beabsichtigt, f¨ugen Sie unmittelbar nach formula den Befehl nospace ein.

98 Kapitel 3 — Alphabetische Befehlsreferenz

3.1.11 gotoXY

gotoXY([<x>] [, [<y>]])

Defaults: aktuelle x- und y-Werte

Mit diesem Befehl kann der aktuelle Punkt auf einen bestimmten absoluten Punkt gesetzt werden. Dies widerspricht zun¨achst der Forderung nach einer selbst¨andigen Ausrichtung aller Elemente eines Schemas, kann aber sinnvoll sein, wenn Sie Formeln an festen Stellen erscheinen lassen wollen, unabh¨angig vom Zufall der jeweiligen Gr¨oß e, etwa in einer tabellarischen Anordnung:

N H Pyrrol N H Pyrrolin

N H N

Imidazol N H N H

Imidazolin

\begin{chemistry}[gotoxy1]
  gotoXY(0,0)
  formula(C,C,"Pyrrol",HA,36)
  { ring("cpentane",,1=3=)
    { 0: atom("N") bond(r) atom("H"); }
  }
  gotoXY(50,0)
  formula(C,C,"Pyrrolin",HA,36)
  { ring("cpentane",,1=)
    { 0: atom("N") bond(r) atom("H"); }
  }
  gotoXY(0,100)
  formula(C,C,"Imidazol",HA,36)
  { ring("cpentane",,1=3=)
    { 0: atom("N") bond(r) atom("H");
      2: atom("N"); }
  }
  gotoXY(50,100)
  formula(C,C,"Imidazolin",HA,36)
  { ring("cpentane",,3=)
    { 0: atom("N") bond(r) atom("H");
      2: atom("N") bond(r) atom("H"); }
  }
\end{chemistry}

Eine relative Anordnung wird durch den Einsatz des Befehls →shiftXY erm¨oglicht.

3.1.12 multiline

multiline(<n> [, [<iPos>] [, [<pos>] [, [<cPos>]]]] )
{ <rliste_1>;
  ...
  <rliste_n>;
}

<iPos> := L | C | R
<pos> := C | L | R | T | B | TL | TR | BL | BR
<Cpos> := C | L | R | T | B | TL | TR | BL | BR

Defaults: <iPos> = L
          <pos> = L
          <cPos> = R

Der Befehl dient dazu, <n> Zeilen mit Formeln oder Reaktionsketten ¨ubereinander zu setzen, wobei zwischen den einzelnen Zeilen ein konstanter Abstand verbleibt. Die einzelnen Zeilen werden gem¨aß <iPos> linksb¨undig, mittig oder rechtsb¨undig zueinander ausgerichtet. Der <n>-zeilige Block als Ganzes wird gem¨aß <pos> auf den aktuellen Punkt ausgerichtet, der Anschluß punkt f¨ur folgende Formeln gem¨aß

<cPos> bestimmt (¨ahnlich dem Positionierungsverfahren f¨ur formula). Beschriftun-

gen von Formeln werden in einem selbst¨andig berechneten, konstanten Abstand gesetzt, sodaß alle Texte einer Zeile in gleicher H¨ohe aufscheinen. Sie d¨urfen innerhalb der beschrifteten formula-Befehle keine Abstandsangaben f¨ur den Textparameter vorgeben!

Im einfachsten Falle wird der Befehl benutzt, um l¨angere Reaktionsschemata auf mehrere Zeilen aufzuteilen oder mehrere kurze Ketten ¨ubereinanderzusetzen:

\begin{chemistry}
  multiline(3,L)
  { formula(L,R){ bond(30; -30,=;30) }
    arrow(){}
    formula(L,R){ bond(30; -30;30) }
    ;
    formula(L,R){ ring(){} }
    arrow(){}
    formula(L,R){ ring(,,H0=2=){} }
    ;
    formula(L,R){ bond(0; 0,3; 0) }
    arrow(){}
    formula(L,R){ bond(30;0,=;-30) }
    ;
  }
\end{chemistry}

Der Spezialfall <n> gleich Eins kann jedoch eingesetzt werden, um mehrere Formeln als Einheit zu betrachten und zu plazieren, einzurahmen oder als Verzweigung dienen zu lassen. Lesen Sie hierzu Abschnitt 2.3.5 des Tutorials.

100 Kapitel 3 — Alphabetische Befehlsreferenz

3.1.13 orbital

orbital([<iAngle>] [, [<rWeight>]])

Default:
         <rWeight> = 1

(Derzeit nur mit einem PostScript-Ausgabeformat!) Mit Hilfe dieses Befehls k¨onnen Orbitallappen oder -keulen dargestellt werden, wie Sie gelegentlich zur Illustration der Vorg¨ange bei Zykloadditionen verwendet werden. Der Parameter <iAngle> ist der Winkel, den die L¨angsachse der Keule und die positive x-Achse einschließ en,

<rWeight> ist ein Skalierungsfaktor, der die Gr¨oß e des Orbitals angibt. Die obere

und die untere H¨alfte des Orbitals unterscheiden sich durch ihren Grauwert. Einige Beispiele:

formula(L,R)
{ orbital(90,2) bond(30)
  orbital(-90,1) bond(-30)
  orbital(90,0.5) bond(30)
  orbital(-90,1) bond(-30)
  orbital(90,2)
}

multiline(1)
{ formula(L,R)
  { orbital(90,1) bond(135,=)
    orbital(-90,1) bond(45)
    orbital(-90,1) bond(0,=) orbital(90,1)
  }
  shiftXY(12,0)
  formula(L,R)
  { orbital(90,1) bond(45,=C) orbital(90,1) }
;}
arrow(){}
formula(L,R)
{ ring(,,H1=)
  { 1: orbital(-110,1);
    2: orbital(-110,1); }
}

3.1.14 restore

restore(#<n>)

Mit diesem Befehl wird der unter der ganzen Zahl <n> gespeicherte Parametersatz reaktiviert (→save).

102 Kapitel 3 — Alphabetische Befehlsreferenz

3.1.15 restoreXY

restoreXY(#<n>)

Mit diesem Befehl wird der aktuelle Punkt auf die zuvor mit →saveXY unter der ganzen Zahl <n> gespeicherte Position gesetzt. Sie k¨onnen so Verzweigungen in einer Struktur erreichen, ohne wie bei →branch rekursiv denken zu m¨ussen (sozusagen

” wilde“ Verzweigungen:-) ). Beispiele:

Dimethyl-zyklopentan formula(C,C,"Dimethyl-zyklopentan",HA,36)
{ saveXY(#1) bond(170; -160; -10,t; 60,t,L)
  saveXY(#2) bond(#1)
  restoreXY(#2) bond(30) restoreXY(#2) bond(150)
}

Bizyklo[2.2.2]oktan formula(C,C,"Bizyklo[2.2.2]oktan",HA,36)
{ bond(30,t,L) saveXY(#1) bond(-30,t,L; 60; 150,,L)
  saveXY(#2) bond(-150,,L; -120)
  restoreXY(#1) bond(90,t,L; 60) bond(#2)
}

Die Beispiele des branch-Befehls k¨onnen mit saveXY und restoreXY folgendermaß en formuliert werden:

\begin{chemistry}
  formula(L,R)
  { atom("Cl") bond(30,,L)
    saveXY(#1)
    bond(90,,L) atom("H")
    restoreXY(#1) bond(-20,<<,L) atom("D")
    restoreXY(#1) bond(-70,<.,L) atom("Br")
  }

  formula(L,R)
  { atom("Cl") bond(30,,L)
    saveXY(#1)
    bond(90,,L) atom("H")
    restoreXY(#1) bond(-20,<<,L) atom("Br")
    restoreXY(#1) bond(-70,<.,L) atom("D")
  }
\end{chemistry}

Es sei noch erw¨ahnt, daß das Bizyklo[2.2.2]oktan als Grundger¨ust in bicyclib.pm vorhanden ist. Verschiedene Zyklopentankonformeren sind in mncyclib.pm enthalten.

3.1.16 ring

ring([ <typ> [, [<start>] [, [<bliste>]
     [, [[#]<len>] [, [<p1>] [, [<p2>]]]]]]])
{ [ <aliste_1>;
    <aliste_2>;
    ... ]

  % nur bei <typ> = "ring", "cpentane", "chair":
  [ vertex([<kante_1>] [, [<kante_2>]
           [, [<bliste_1>] [, [<p1>] ]]]])
    { [<aliste_11>;
       <aliste_12>;
       ... ]
    }
    ... ]
}

<start> := 0..<n-1>
<bliste> := [H|O] [ 0|1|...|<b><btype> <bliste>]
<aliste_i> := C|0|1|...|<n-1> : <fliste>

Defaults: <typ> = "ring"
          <start> = 0
          <bliste> = O
          <len> = N
          <p1> = 6
          <p2> = -90

Der Befehl dient zur Erzeugung vorgefertigter Ringstrukturen. Die erweiterte Syntax mit vertex ist nur bei den angegebenen Ringtypen m¨oglich.

Fixe Parameter

Der Parameter typ bestimmt gem¨aß nachfolgender Kodierung, welches Grundger¨ust gezeichnet wird:

ring cpentane chair newman

Obige Liste enth¨alt die im Compilercore defi nierten Ringstrukturen; mit Hilfe des Befehls →require k¨onnen Sie jedoch eine Bibliothek aus extern defi nierten Ringstrukturen aufbauen und bei Bedarf hinzuladen. M¨oglichkeiten, f¨unfgliedrige Ringe in Stereodarstellung ¨ahnlich chair zu erzeugen, sind bei den Steroiden in Abschnitt 2.4 des Tutorials sowie der Befehlsreferenz zu restoreXY gezeigt worden.

Die Anordnung der Bindungen bei den Richtungstypen r, t und b des Befehls bond wird in den nachfolgenden Formeln gezeigt. t-Bindungen werden durch dicke, b- Bindungen durch gestrichelte Linien symbolisiert.

radial top/bottom radial top/bottom radial top/bottom

radial, p=1 top/bottom, p=1

radial, p=0 top/bottom, p=0

radial, p=-1 top/bottom, p=-1

Einige der Parameter besitzen eine feststehende Bedeutung, bei anderen variiert diese je nach dem zugrundeliegenden Ger¨ust. Fixe Bedeutung kommt den Parametern <start> und <bliste> zu. Mit dem ersten Wert, der zwischen Null und der Anzahl der Atome im Ger¨ust minus Eins variieren kann, legen Sie die relative Lage des Ger¨ustes zum aktuellen Startpunkt fest, das heiß t, welches Atom am augenblicklichen Zeichenpunkt liegt. Ist die Ringstruktur das erste Element der Formel, ist dieser Parameter ohne Bedeutung, da der Ring das Basiselement der Formel wird. Haben Sie jedoch bereits Bindungen erzeugt, geben Sie hiermit das Zielatom der letzten Bindung an. Die Numerierung der einzelnen Ger¨uste ist im folgenden gezeigt:

0 1 2 3 4 5 01 2 3 4 0 1 2 3 4 5 0 1 2 3 4 5

Als Grundregel f¨ur einfache n-gliedrige Ringe und die meisten anderen Grundger¨uste gilt: die Nummern werden im Uhrzeigersinne gr¨oß er, das Atom mit der Nummer Null liegt auf der positiven x-Achse oder ist um einen Winkel, der typspezifi sch ist (- 90^o f¨ur den Typ ring), im Uhrzeigersinne gedreht. Bei einigen Grundger¨usten kann die Reihenfolge der Nummern abweichen.

Beachten Sie bei den folgenden Anwendungsm¨oglichkeiten, daß ring den aktuellen Startpunkt nicht ver¨andert, was in der letzten Formel ausgiebig verwandt wird, um die drei Ringe zu zeichnen, ohne einige von ihnen als Substituenten der anderen betrachten zu m¨ussen:

formula(C,C)
{ bond(150; -150) ring(,4){} }

formula(C,C)
{ bond(-150; -90) ring(,3){} }

formula(C,C)
{ bond(30,=C) ring("cpentane",1){} }

N formula(C,C)
{ ring(,4,H){} ring(,2,H){} ring(,0,H){}
  atom("N")
}

Die Numerierung dient auch zur Kenntlichmachung, welcher Substituent an welchem Ringatom beginnt (Nummern in <aliste>).

Mit der Bindungsliste <bliste> variieren Sie die Bindungsarten, die beim Aufbau des Ger¨ustes eingesetzt werden. Sie besteht aus einer Folge von Paaren aus Bindungsnummer und Bindungstyp. Die Bindungsnummer beginnt bei Null und kann bis zu Eins kleiner als die Anzahl der Bindungen im Ring ¨uberhaupt sein. Die Zuordnung der Bindungsnummer zu den einzelnen Bindungen ist folgende:

0 1 2 3 4 5 0 1 2 3 4 0 1 2 3 4 5 0 1 2 3 4 5

Die Angabe des Bindungstyps kennen Sie bereits vom Befehl bond her. Da die Angabe eines oder mehrerer Bindungstypen nicht automatisch das Zeichnen eines Aromatenringes unterbindet, m¨ussen Sie dies gegebenenfalls zus¨atzlich anweisen. F¨ur diesen Zweck stehen die zus¨atzlichen Typen H f¨ur ges¨attigte (Ring wird nicht gezeichnet) und O f¨ur aromatische Kerne (Ring wird gezeichnet) zur Verf¨ugung. Diese Angaben m¨ussen vor einer eventuellen Bindungsliste stehen! Einige Beispiele m¨ogen die Bindungsliste verdeutlichen:

formula(C,C){ ring(,,H){} }

formula(C,C){ ring(,,H0=2=4=){} }

formula(C,C){ ring("cpentane",,0=){} }

formula(C,C){ ring(,,H1=4s){} }

Der Parameter <len> kodiert die Gr¨oß e des Ringes. Die Defaulteinstellung N entspricht bei Polygonen dem Radius des zugrundeliegenden Kreises, bei anderen Ger¨usten der L¨ange der Hauptbindungen. Ist es geometrisch m¨oglich, erhalten alle Bindungen diese L¨ange, ansonst die gr¨oß tm¨ogliche Anzahl. Bei Polygonen folgt aus der Festlegung des Radius als Grundlage, daß die Kantenl¨ange bei Dreiecken erheblich gr¨oß er ist als bei Neunecken. W¨unschen Sie die Kantenlänge festzulegen, m¨ussen Sie die

Gr¨oß envariante mit # verwenden, siehe unten.

Eine Abweichung vom Normalwert der Ringgr¨oß e ist besonders bei komplizierten Ger¨usten wie dem Zyklohexan sinnvoll:

formula(C,C) { ring("chair",,,S){} }

formula(C,C) { ring("chair",,,N){} }

formula(C,C) { ring("chair",,,L){} }

formula(C,C) { ring("chair",,,NN){} }

Die Variante mit einem # vor der Bindungsl¨ange ist nur bei Ger¨usten des Typs ring erlaubt und legt fest, daß <len> nicht dem Radius des Kreises, sondern der Kantenl¨ange des Polygons entspricht. Polygone unterschiedlicher Eckenzahl weisen dann die festgelegte, gleiche Kantenl¨ange auf. Sinnvoll ist dies, wenn kondensierte Ringe unterschiedliche Gliederzahl aufweisen, wie der Vergleich zeigt:

gotoXY(0,0)
formula(C,C) { ring(,5){}
               ring(,2,,,5,0){}
             }

gotoXY(60,0)
formula(C,C) { ring(,5,,#N){}
               ring(,2,,#N,5,0){}
             }

In der Grundform sind die Kantenl¨angen eines Sechsringes etwas k¨urzer als die eines F¨unfringes, was beim Kondensieren der Systeme deutlich auff¨allt (noch extremer bei Dreiringen). In einem solchen Falle muß deswegen jeder der beteiligten Ringe eine konstante Kantenl¨ange aufweisen, mit #N angezeigt.

Variable Parameter

Die Angaben <p1> und <p2> haben je nach dem Ger¨usttyp unterschiedliche Bedeutung.

ring : Bei diesem Grundger¨ust haben sie die Bedeutung der Anzahl der Ecken des Ringes (Default ist 6 f¨ur einen Benzenring) und eines Winkels in Altgraden, um den der Ring gedreht ist. Bezugsachse ist hierbei die positive x-Achse, der Drehsinn mathematisch negativ (Uhrzeigersinn). Der Defaultwert von -90^o sorgt daf¨ur, daß der Benzenring auf der Ecke mit der Nummer 0 steht. Die wichtigsten alizyklischen Grundk¨orper k¨onnen wie folgt gesetzt werden:

0 formula(C,C) { ring(,,H,,3,90)

{ 0: atom("0"); } }

0 formula(C,C) { ring(,,H,,4,-45)
               { 0: atom("0"); } }

0 formula(C,C) { ring(,,H,,5,-90)
               { 0: atom("0"); } }

0 formula(C,C) { ring(,,H,,6,-90)
               { 0: atom("0"); } }

0 formula(C,C) { ring(,,H,,7,90)

{ 0: atom("0"); } }

Einige wichtige Winkel im Zusammenhang mit diesen grundlegenden Polygonen sind in den folgenden Formeln angegeben. Die Winkel gelten f¨ur die Bindung in Pfeilrichtung vom n¨achsten Ringatom aus. Sind die Ringe gedreht, so ist der Drehwinkel zu den angegebenen Werten zu addieren.

0^o

-72^o

-144^o

144^o

72^o 54^o

-18^o -90^o

-162^o

126^o

-54^o -126^o

162^o

90^o 18o

0^o

-60^o

-120^o

180^o

120^o 60^o 60^o

0^o -60^o

-120^o

180^o

120^o

-60^o -120^o

180^o

120^o

60^o 0^o 0^o

-51^o -103^o

-154^o

154^o

103^o 51o

cpentane : Der Parameter <p1> ist ohne Bedeutung, <p2> stellt wie beim Grundger¨ust ring den Drehwinkel des gesamten Ringes dar.

chair : <p2> stellt wie beim Grundger¨ust ring den Drehwinkel des gesamten Ringes dar.

<p1> kodiert mit den drei m¨oglichen Werten 1, 0 und −1 die beiden Sessel- und

twist-Konformeren:

p_1 = 1 p_1 = 0 p_1 = −1

newman : Dieses Ger¨ust entspricht dem axialen Blick auf eine Einfachbindung (Darstellung in der Newman-Projektion). Vom vorderen Atom ausgehende Bindungen zweigen unter den festen Winkeln 90^o, 210^o und 330^o ab. Die Bindungen des hinteren Atoms werden um den Winkel p1 verdreht dargestellt:

BrBr

syn

Br

Br anti formula(L,R,"syn",HA,36)
{ ring("newman",,,L,-20){ 0: atom("Br");
                          3: atom("Br");} }

formula(L,R, "anti",HA,36)
{ ring("newman",,,L,180){ 0: atom("Br");
                          3: atom("Br");} }

Substituenten

Mit den einzelnen <aliste_i>-Komponenten im Rumpf des Befehls spezifi zieren Sie Substituenten, die ihren Ursprung im i-ten Ringatom haben. Die Numerierung der Ringatome wurde bereits im Zusammenhang mit <start> vorgestellt. Die Substituenten k¨onnen einzelne Bindungen, weitere Ringe oder Kombinationen aus allen Formelelementen sein. Beispiele:

H_2N N=N OH formula(L,R)
{ ring()
  { 1: bond(r) atom("H$_2$N",R);
    4: bond(r) atom("N=N",L,R) bond(r-) ring(,2){};
    3: bond(r) atom("O",C,R) atom("H",L);
  }
}

108 Kapitel 3 — Alphabetische Befehlsreferenz

N N H formula(L,R)
{ ring("cpentane",,2=4=)
  { 0: atom("N");
    1: atom("N") bond(r) atom("H");
  }
}

Br Br Br Br formula(L,R)
{ ring(,,H,,5,0)
  { 2: bond(r,<<) atom("Br");
    3: bond(r,<.) atom("Br");
  }
}
formula(L,R)
{ bond(150)
  ring("newman",3)
  { 0: atom("Br");
    1: bond(30,t);
    4: atom("Br");
  }
}

Der vertex-Befehl

Bei den Ringtypen, die einen Drehwinkel unterst¨utzen, ist eine erweiterte Syntax mit vertex-Befehlen m¨oglich. Diese dienen dazu, weitere Ringe ¨uber eine Kante mit dem aktuellen Ring zu verkn¨upfen. <kante_1> und <kante_2> kodieren die betroffenen Kanten des Basisringes (des aktuellen Ringes) und des zweiten Ringes in Form der Zahlen, die auch zum Aufbau der Bindungsliste benutzt werden. <bliste_1> ist die Bindungsliste f¨ur den neuen Ring, der ebenfalls nur vom Typ ring sein darf.

<p1> schließ lich ist die Ringgr¨oß e des neuen Ringes.

Der Rumpf des vertex-Befehls ist identisch zum Rumpf von ring, das heiß t es k¨onnen, durch Positionszahlen spezifi ziert, Substituenten oder auch weitere vertex- Befehle eingesetzt werden. Einige Beispiele sollen den Einsatz verdeutlichen.

Triphenylen formula(L,R,"Triphenylen",HR,24)
{ ring(){ vertex(,0){};
          vertex(,2){};
          vertex(,4){}; }
}

formula(L,R)
{ ring(,,H,,5,90)
  {
    vertex(,1,3,H0=,5){ 0: bond(r); };
    vertex(,3,1,H0=,6){};
  }
}
formula(L,R)
{ ring("cpentane",,,L)
  { vertex("cpentane",3,1,H3=)
    {3: bond(r); vertex(,4,0,H){};};
  }
}

formula(L,R)
{ ring("cpentane",,,,,0)
  { vertex("cpentane",2,2,H0=)
    {3: bond(b); vertex(,4,0,H,5){};};
  }
}

Symbolische Drehwinkel

Bei den Ringtypen ring und cpentane sind anstelle ganzer Zahlen auch die symbolischen Angaben r, t und b f¨ur den Drehwinkel <p2> m¨oglich. ¨Ahnlich den analogen Angaben des bond-Befehls, werden auch an dieser Stelle die Winkel der radialen beziehungsweise tangentialen Bindungen kodiert: der Drehwinkel des Ringes wird so bestimmt, daß die letzte gezeichnete Bindung, die zum spezifi zierten Startatom f¨uhrt, in der symbolischen Relation zum Ring steht:

As

Tetraphenylarsol formula(L,R,"Tetraphenylarsol",HR,24)
{ ring(,,H1=3=,,5,90)
  { 0: atom("As");
    1: bond(r) ring(,,,,,r){};
    2: bond(r) ring(,,,,,r){};
    3: bond(r) ring(,,,,,r){};
    4: bond(r) ring(,,,,,r){}; }
}

Weitere Beispiele f¨ur symbolische Drehwinkel sind im Tutorial zu fi nden.

3.1.17 save

save(#<n>)

Dieser Befehl sichert die beim Formelsatz benutzten graphischen Parameter unter der ganzen Zahl <n>. Sie verwenden den Befehl, wenn Sie mit →set lokale ¨Anderungen am Parametersatz vornehmen wollen und die Originalwerte wieder ben¨otigen (→restore). Beispiele zu diesem Befehl fi nden Sie im Tutorial, Abschnitt 2.5.

3.1.18 savecontext

savecontext(#<n>)

Mit diesem Befehl wird der Kontext, der durch die zuletzt dargestellte Formel festgelegt wird, unter einer Nummer <n> gespeichert. Sie k¨onnen so sp¨ater auf diesen Formelkontext zur¨uckgreifen (→setcontext), um zum Beispiel Abzweigungen im Schema zu erzeugen, auch wenn bereits andere Formeln gezeichnet wurden.

3.1.19 saveXY

saveXY(#<n>)
saveXY(#<n>, <iAngle>, <Len>)

Der Befehl speichert in der oberen Syntaxvariante die aktuellen Koordinaten unter der Nummer <n>. Die erweiterte, untere Syntax speichert die Koordinaten eines Punktes, der vom aktuellen Punkt um die L¨ange <Len> in Richtung des Winkels

<iAngle> verschoben ist. Beide Angaben entsprechen denen des bond-Befehls. Der

aktuelle Punkt wird dabei nicht verschoben.

Einfache Syntax

Sie benutzen den gespeicherten Punkt, um sp¨ater mit bond(#<n>) eine Bindung von einem anderen Atom zu diesem Punkt zu zeichnen, ohne die Richtung und L¨ange der Bindung angeben zu m¨ussen. Sie k¨onnen so auf einfache Weise polyzyklische und irregul¨are Ger¨uste, wie sie f¨ur Naturstoffe typisch sind, generieren. Das Tropin und Nor-Pinan sind als Grundger¨uste in bicyclib.pm vorhanden, hier soll jedoch gezeigt werden, wie Sie sich behelfen k¨onnen, falls ein Ger¨ust nicht in einer Bibliothek zu fi nden ist: N H

Desoxy-Tropin formula(C,C,"Desoxy-Tropin",HA,36)
{ ring("chair",,,L)
  { 0: atom("N") bond(150) atom("H");
    1: bond(-100) saveXY(#1);
    5: bond(-100,t; #1,t); }
}

Nor-Pinan formula(L,R,"Nor-Pinan",HA,36)
{ ring(,,H)
  { 2: saveXY(#1);
    0: bond(90; #1); }
}
space(R)
formula(L,R)
{ ring("chair",,,L,-1)
  { 2: saveXY(#1);
    4: bond(-60,t; #1); }
}

112 Kapitel 3 — Alphabetische Befehlsreferenz

7,8,9-Trinor-thujan Thujan formula(L,R,"7,8,9-Trinor-thujan",HR,24)
{ ring(,,H)
  { 3: bond(r);
    4: saveXY(#1);
    0: bond(#1); }
}
space(R)
formula(L,R,"Thujan",HR,24)
{ ring(,,H0t1t,,5,0)
  { 4: bond(90);
    1: saveXY(#1);
    0: bond(90; #1,p; -60)
       branch { bond(30); bond(-90); };
  }
}

Nor-Caran formula(C,C,"Nor-Caran",HA,36)
{ ring(,,H)
  { 1: saveXY(#1);
    0: bond(-150; #1); }
}

Da die Thujan- und Caranger¨uste in planarer Darstellung regul¨are Bindungswinkel und -l¨angen aufweisen, k¨onnen sie einfach wie folgt formuliert werden:

\begin{chemistry}[savexy2]
  formula(L,R,"Nor-Pinan",HA,36)
  { ring(,,H){ 0: bond(90; 150); } }

  formula(L,R,"Nor-Caran",HA,36)
  { ring(,,H){ 0: bond(-150; 90); } }
\end{chemistry}

O O

(-)-(1S,5R)-Frontalin formula(C,C,"(-)-(1S,5R)-Frontalin",HA,36)
{ ring("chair",,,L)
  { 3: bond(t);
    3: bond(b,t) atom("O") saveXY(#1);
    4: atom("O");
    5: bond(t);
    5: bond(b,t; #1,t); }
}

Die Zusammenarbeit mit dem Befehl →restoreXY erweitert die M¨oglichkeiten, irregul¨are Naturstoffger¨uste aufzubauen, erheblich; siehe Befehlsbeschreibung zu

restoreXY. Beachten Sie, daß der G¨ultigkeitsbereich der gespeicherten Punkte mit

der aktuellen Formel nicht automatisch endet, sodaß eine sp¨atere irrt¨umliche Verwendung dieses Punktes ohne erneute Zuweisung auf eine dann falsche Formel verweist! Sie k¨onnen sich dies nat¨urlich zunutze machen, wenn Sie zwei Formeln miteinander verkn¨upfen wollen, zum Beispiel den Elektronen¨ubergang darstellen oder etwas besonders Trickreiches planen.

Erweitere Syntax

Diese Befehlsvariante ist vor allem bei der Darstellung von Elektronenverschiebungen von Interesse, entsprechende Beispiele sind bei emove sowie im Tutorial, Abschnitt 2.6 zu fi nden. In einem anderem Zusammenhang ist diese Variante brauchbar, wenn Sie unkompliziert Punkte neben der Hauptlinie der Formel markieren m¨ussen, ein Beispiel w¨aren Ladungssymbole neben einer Alkankette:

O ⊕

⊖ formula(L,R)
{ bond(-30) saveXY(#1,-90,n)
  bond(30; -30,=; 30)
  bond(90) saveXY(#2,45,N)
  atom("O")
  restoreXY(#1) atom("$\oplus$")
  restoreXY(#2) atom("$\ominus$")
}

114 Kapitel 3 — Alphabetische Befehlsreferenz

3.1.20 scale

scale(<p>)

Der Befehl skaliert alle internen graphischen Parameter mit dem Faktor <p>. <p> ist eine Angabe des Bruchteils; um die Formeln auf 70% zu verkleinern, schreiben Sie daher

scale(0.7)

Zahlen gr¨oß er als 1 f¨uhren zu einer Vergr¨oß erung, Zahlen unter 1 zu einer Verkleinerung der folgenden Zeichnung.

Der Befehl wird im allgemeinen innerhalb eines →save/restore-Paares angewandt, um die Originalwerte wieder reaktivieren zu k¨onnen. Als Beispiel soll eine Hilfsreaktion auf einem Reaktionspfeil verkleinert dargestellt werden:

COOH

SOCl_2

COCl O

\begin{chemistry}[scale]
  formula(L,R)
  { ring(){} }

  arrow()
  { text(T,L)
    { save(#1)
      scale(0.8)
      formula(T,B){ bond(30; -30)atom("COOH",L) }
      arrow(-90)
      { text(T,L) { formula(C,C){ atom("SOCl$_2$") } } }
      formula(T,B){ bond(30; -30)atom("COCl",L) }
      restore(#1)
    }
  }

  formula(L,R)
  { ring()
    { 4: bond(r) branch { bond(r+,=C) atom("O"); }
         bond(r-; r); }
  }
\end{chemistry}

3.1.21 shiftXY

shiftXY([<dx>] [, [<dy>]])

Default: <dx> = 0
         <dy> = 0

Der Befehl erlaubt es, auß erhalb von Formeln den aktuellen Punkt um die Gr¨oß e

<dx> auf der x-Achse, um <dy> auf der y-Achse zu verschieben. Lassen Sie Angaben

fort, fi ndet auf der entsprechenden Achse keine Verschiebung statt.

Hauptnutzen des Befehls ist es, mehrere Reaktanden, die nicht streng nacheinander aufgef¨adelt erscheinen sollen, sondern eher eine Art ” Wolke“ bilden, richtig gegeneinander anzuordnen. Im Gegensatz zu →gotoXY werden jedoch keine festen Punkte angegeben, sondern nur Relationen. Verschiebungen des Bezugspunktes (aktueller Punkt) betreffen gleichartig alle relativen Positionierungen. Es ist zweckm¨aß ig, aber nicht notwendig, die Formeln mit den Positionierungsparametern C,C zu setzen. Ein Beispiel ist im Tutorial, Abschnitt 2.6, gegeben. Dort sind drei Reaktanden so zu setzen, daß zwischen ihnen die Elektronen¨uberg¨ange darstellbar sind. Die Zusammenfassung aller ” wild verteilten“ Reaktanden zu einer Einheit gelingt wieder einmal mit einem einzeiligen multiline-Kommando.

Tabelle 3.1 Interne Satzparameter, die mit set ge¨andert werden k¨onnen.

Parameter Bedeutung

rLW Liniendicke f¨ur alle Linien wie Bindungen, Kreise, Pfeile und

sonstige Linien.

rBW Maximale Dicke bei projektiven Bindungsformen.
rAW Dicke der Spitze von Reaktionspfeilen.
rBD Strichelungsmuster f¨ur projektive Bindungen.
rXS Horizontaler Abstand von Formeln und Pfeilen zueinander.
rXY Vertikaler Abstand von Formeln und Pfeilen zueinander.
rArrowExtend Zus¨atzlicher Abstand, um den ein Reaktionspfeil auf beiden

Seiten ¨uber die Beschriftung hinausreicht.

rArrowSkip Abstand der Pfeilbeschriftung vom Pfeil selbst.
rChainSep Abstand von Reaktionsketten zueinander bei den Befehlen
               joinh und joinv.
rMultilineSep Abstand von Reaktionsketten zueinander beim Befehl
               multiline.
fboxsep Abstand des Rahmens/Klammer von der Formel.
rTextSep Vertikaler Abstand der Formellegenden von der Unterkante

der tiefsten Formel.

iAngleR Winkel f¨ur radiale Bindungen, Symbol r.
iAngleT Winkel f¨ur radiale Bindungen, Symbol t.
iAngleB Winkel f¨ur radiale Bindungen, Symbol b.

3.1.22 set

set(<cPar>, <iVal>)

Mit diesem Befehl k¨onnen Sie jeweils einen der in Tabelle 3.1 aufgelisteten Parameter <cPar>, die das Layout des Formelsatzes steuern, ¨andern und ihn auf den Wert

<iVal> setzen. Die Wirkung bezieht sich teilweise auf Satzparameter wie Abst¨ande

von Formelzeilen zueinander, teilweise auf Druckparameter wie Linienbreite und Pfeilst¨arke. ¨Anderungen an Parametern der letztgenannten Gruppe werden in die Ausgabedatei geschrieben.

Mit diesem Befehl vorgenommene ¨Anderungen sind permament, es ist gegebenenfalls erforderlich, sie lokal durchzuf¨uhren, indem der Parametersatz vor dem Abschnitt des Programmes, f¨ur den sie G¨ultigkeit besitzen sollen, unter einer bestimmten Nummer gespeichert und nach Ende des G¨ultigkeitsbereichs unter dieser Nummer wieder aktiviert wird (Befehle save und restore):

...
save(#1)
set("rLW", 2) % erhoehe Linienstaerke
formula...
...
restore(#1) % hole alte Werte
... % wieder normale Linienbreite

Ein Beispiel f¨ur eine Anwendung ist im Tutorial, Abschnitt 2.5 gegeben.

3.1.23 setcontext

setcontext(#<n>, <pos>)
[nospace]

<pos> := C | L | R | T | B | TL | TR | BL | BR

Der Befehl legt den unter der Nummer <n> zuvor gespeicherten Kontext als aktuellen Kontext ab und ermittelt einen Fortsetzungspunkt, der sich aus der Kontext- Bounding Box und dem Positionierungsparameter <pos> ergibt. Haupteinsatzgebiet des Befehls ist das Bilden von Verzweigungen in Reaktionsketten: in Schema 3– 4 wird, vom Azlakton ausgehend, eine Reaktionskette ” normal“ notiert in der Folge (L,R) – (L,R) – (L,TR). Damit vom Azlakton noch ein zweiter Kettenstrang abzweigen kann, muß zun¨achst nach Satz der Laktonformel deren Kontext gespeichert (→savecontext) und nach Ende der ersten Kette wieder aktiviert werden. Da die zweite Kette nach rechts unten verlaufen soll, wird als Positionierung dabei BR angegeben:

\begin{chemistry}[cont1]
multiline(1)
{ formula(L,R, "Hippurs""aure")
  { ring(,,H3s4=C,,5)
    { 0: bond(r) ring(,3){};
      1: atom("N") bond(r) atom("H");
      3: atom("COOH",L,C);
      4: atom("O"); }
  }

  arrow()
  { text(T,L) { formula(C,C){ atom("- H$_2$O") } } }

  formula(L,R, "Azlacton")
  { ring(,,H0=,,5)
    { 0: bond(r) ring(,3){};
      1: atom("N");
      3: bond(r,=C) atom("O");
      4: atom("O"); }
  }
  savecontext(#1);
}

setcontext(#1,R)
arrow(){ text(T,L) { formula(C,C){ atom("R-CHO") } } }

formula(L,TR)
{ ring("cpentane",,0=)
  { 0: bond(r) ring(,3){};
    1: atom("N");
    2: bond(r,=U) bond(-150) atom("R");
    3: bond(r,=C) atom("O");
    4: atom("O");
  }
}

118 Kapitel 3 — Alphabetische Befehlsreferenz

savecontext(#2)

arrow(45)
{ text(T,L){ formula(C,C){ atom("3 H$_2$O") } }
}

formula(L,R,"$\alpha$-Aminos""aure", HR, 24)
{ atom("H$_2$",C,R) atom("N",L,C) bond(90)
  branch
  { bond(150; -150) atom("R");
    bond(30) atom("COOH",L,R);
  }
}

setcontext(#2,BR)
arrow(-45)
{ text(T,L){ formula(C,C){ atom("1. H$_2$") } }
  text(B,L){ formula(C,C){ atom("2. H$_2$O") } }
}

formula(L,R,"2-Oxos""aure", HR, 24)
{ atom("O") bond(90,=C)
  branch
  { bond(150; -150) atom("R");
    bond(30) atom("COOH",L,R);
  }
}
\end{chemistry}

Sie k¨onnen durch Speichern mehrerer Kontexte und durch Verwendung mehrerer setcontext-Befehle verschiedene Kettenverzweigungen formulieren sowie von einer Formel mehrere Ketten abzweigen lassen, was im Tutorial, Abschnitt 2.3.4, ausf¨uhrlich erl¨autert wird.

N H COOH O

Hippursäure

- H_2O N O O

Azlacton

R-CHO N R O O

3 H_2O

H_2N

R COOH

α-Aminosäure

1. H_2 2. H_2O

O R COOH

2-Oxosäure

Schema 3–4 Mehrere Reaktionsketten zweigen von derselben Formel ab. Der Kontext dieser Formel muß daf¨ur gespeichert werden.

3.1.24 space

space(<pos>)

<pos> := C | L | R | T | B | TL | TR | BL | BR

Mit diesem Befehl k¨onnen Sie zus¨atzlichen Zwischenraum in der durch <pos> spezifi zierten Richtung einf¨ugen – oder in Form eines gezielten R¨uckschrittes wieder entfernen. (Nach einem Pfeil oder einer Formel ist dies in der Fortsetzungsrichtung mit nospace m¨oglich.)

3.2 Sonderbefehle

Die in diesem Abschnitt enthaltenen Kommandos d¨urfen keinesfalls innerhalb einer chemistry-Umgebung in L^ATEX-Dokumenten auftreten. Einige von ihnen sind Bestandteil des eigentlichen Sourcefi les, das w¨ahrend der Bearbeitung von L^ATEX erzeugt oder von Ihnen per Hand geschrieben werden kann. Ihnen entsprechen gem¨aß dem folgenden Schema L^ATEX-Befehle, sodaß keine Notwendigkeit besteht, die grundlegenden Befehle einzusetzen:

\begin{chemistry}{...}↔schema("...")
\end{chemistry} ↔endschema
\chemfontname ↔font

Es gibt jedoch auch einige Sonderbefehle, die w¨ahrend der Initialisierung von OCHEM ben¨otigt werden, nicht jedoch innerhalb jeder einzelnen chemistry-Umgebung. Diese k¨onnen innerhalb der chemspecial-Umgebung eingef¨ugt werden. Diese Umgebung ¨ubernimmt den eingeschlossenen Text ohne ¨Anderung in das eigentliche Sourcefi le und steht in der Regel zu Beginn des Dokumentes. Zu den Befehlen, die sinnvoll in einem L^ATEX-Dokument chemischen Inhalts eingesetzt werden k¨onnen, z¨ahlen require und package.

Wenn Sie sich das nach einem L^ATEX-Lauf erstellte .chm-File anschauen, k¨onnen Sie sehen, wie Ihr Dokument in das Sourcefi le ¨ubertragen wurde.

3.2.1 font

font("<name>")

Dieses Kommando enth¨alt die L^ATEX-Befehle, die den f¨ur chemische Textsymbole des

atom-Befehls zu verwendenden Font einstellen. Diese Form des Befehls dient nur

internen Zwecken, da f¨ur den L^ATEX-Anwender der Befehl \chemfontname bereitsteht.

3.2.2 package

package("<Paket>")
package("[<Optionen>]<Paket>")

L¨adt das angegebene Paket, ggf. mit Optionen, w¨ahrend der L^ATEX-Bearbeitung der Textst¨ucke der Formeln. Der Befehl kann mehrfach auftreten, muß jedoch vor der ersten Formel stehen.

Die Berechnung der Gr¨oß e, die ein Text, z. B. ein Atomsymbol, einnimmt, wird f¨ur ein L^ATEX-Ausgabeformat dem L^ATEX-Compiler ¨ubertragen. Dieser wird mit einer tempor¨aren Datei aufgerufen, die alle im Reaktionsschema benutzten Textst¨ucke enth¨alt. Falls in diesen Texten Makros verwendet werden, m¨ussen diese auch f¨ur die L^ATEX- Bearbeitung der tempor¨aren Datei bekannt sein. OCHEM verwendet daher folgendes Verfahren: alle Makros, die Sie innerhalb von Textpositionen in Formeln verwenden, m¨ussen Sie in einem (oder mehreren) separaten Paketen, z. B. biochem.sty, sammeln. Die Namen dieser Pakete werden ¨uber den package-Befehl von der tempor¨aren Datei geladen und somit verf¨ugbar. Auch Paketoptionen k¨onnen ¨ubergeben werden:

\begin{chemspecial}
package("biochem")
package("[biochem]chemmacros")
\end{chemspecial}

Ein Beispiel w¨are ein P in einem Kreis, das in der Biochemie h¨aufi g als Symbol f¨ur aktiviertes Phosphat benutzt wird. Sie k¨onnen das Symbol mit Hilfe eines L^ATEX-Makros erzeugen. Die Makrodefi nition schreiben Sie in eine Datei biochem.sty:

% biochem.sty

\newcommand{\Phosphat}
{{\setlength{unitlength}{1pt}
 \begin{picture}(20,20)
   \put(10,10){\makebox(0,0){P}\circle{20}}
 \end{picture}
}}

Das Makro k¨onnen Sie dann in Formeln wie folgt verwenden:

O OH

OH OH O P P H

Xylose-4-diphosphat

\usepackage{biochem}

\begin{chemspecial}
package("biochem")
\end{chemspecial}

...

\begin{chemistry}
formula(C,C,"Xylose-4-diphosphat",HR,24)
{ ring("furanose",,,L)
  { 0: atom("O");
    1: bond(90) atom("O",C,R) atom("H",L);
    2: bond(-90) atom("O",C,R) atom("H",L);
    3: bond(90) atom("O",C,R) atom("H",L);
    4: bond(90;180) atom("O") bond(180)
         atom("\Phosphat",R,L) bond(180)
         atom("\Phosphat",R,L) bond(180)
         atom("H",R);
  }
}
\end{chemistry}

Wie das Beispiel zeigt, muß das Paket mit den eigenen Makros auch im L^ATEX-Doku-ment geladen werden, da dieses den Text in der endg¨ultigen, vom Chemiecompiler errechneten Positionierung setzen muß und daher ebenfalls auf die Makrodefi nitionen zugreift.

3.2.3 require

require(<File>)

Dieser Befehl l¨adt das Perl-Modul <File>.pm. Er muß vor dem ersten schema-Befehl stehen, kann jedoch mehrfach auftreten. Sein prim¨arer Zweck ist nicht, ” echte“ Perl- Module zu laden (wiewohl dies m¨oglich w¨are), sondern eigene Ringstrukturen, die als Perl-Code vorliegen m¨ussen, verf¨ugbar zu machen. Sie k¨onnen so eine Vielzahl an seltener gebrauchten Strukturen bereitstellen, ohne jede einzelne im Compilercode verankern zu m¨ussen und ihn so zu ¨uberfrachten. Auch ist es so m¨oglich, Strukturen zu entwickeln, ohne automatisch eine eigene Version des Compilers zu erzeugen, Sie bleiben somit kompatibel und k¨onnen den Compiler jederzeit gegen die neuste Version austauschen, ohne Ihrer Defi nitionen verlustig zu gehen.

Der Aufbau von Ger¨uststrukturen wird im Tutorial, Abschnitt 2.9, geschildert, einige wie bicyclib.pm, sind in der OCHEM-Distribution standardm¨aß ig enthalten. M¨ochten Sie dieses Modul laden, um das Ger¨ust tropin (Abschnitt 3.4) zu laden, schreiben Sie

\begin{chemspecial}
  require("bicyclib")
\end{chemspecial}

\begin{chemistry}
  formula(C,C)
  { ring("tropin",,,L)
    { ... }
  }
\end{chemistry}

3.2.4 schema

schema("<name")
<sliste>
endschema

Dieses Kommando kennzeichnet Anfang und Ende einer Formel oder eines ganzen Reaktionsschemas. <sliste> wird als Einheit ¨ubersetzt und unter dem Dateinamen <name> im Ausgabeformat gespeichert. Der Befehl stellt die logische Grundeinheit des Chemiepaketes dar und korrespondiert zur chemistry-Umgebung auf L^ATEX-Benutzerebene.

3.3 Die Bibliothek mncyclib.pm

In diesem Abschnitt werden die Ringstrukturen erl¨autert, die durch das Bibliotheksmodul mncyclib.pm verf¨ugbar gemacht werden. Es wird durch die Zeilen

\begin{chemspecial}
  require("mncyclib")
\end{chemspecial}

im Vorspann eines L^ATEX-Dokumentes verf¨ugbar gemacht. Die folgenden Erkl¨arungen folgen dem Schema, das bereits bei →ring angewandt wurde, f¨ur n¨ahere Erkl¨arungen siehe die Beschreibung des Befehls.

Die Bibliothek enth¨alt die folgenden monozyklischen Strukturen und ihre Kodierung <type>:

cb cp furanose

Sie k¨onnen mit den Ger¨usten dieser Bibliothek monozyklische Strukturen in schematisierter Stereodarstellung erzeugen.

Die Numerierung der Atome ist durch die folgenden Formeln beschrieben: 0 1 2 3 0 1 2 3 4 0 1 2 3 4

Die Numerierung der Bindungen erfolgt nach den folgenden Schemata

0 12 3 0 1 2 3 4 0 1 2 3 4

Die Bedeutung der beiden variablen Parameter <p1> und <p2> ist folgende:

cb : (Zyklobutan-Ger¨ust) Der Parameter <p1> entf¨allt, <p2> stellt den Drehwinkel des gesamten Ger¨ustes dar.

Die Anordnung der Bindungen bei den Richtungstypen r, t und b wird in den nachfolgenden Formeln gezeigt. t-Bindungen werden durch dicke, b-Bindungen durch gestrichelte Linien symbolisiert.

radial top/bottom

3.3 — Die Bibliothek mncyclib.pm 127

cp : (Zyklopentan-Ger¨ust) Der Parameter <p2> stellt wiederum den Ger¨ustdrehwinkel dar. <p1> erm¨oglicht die Auswahl zwischen den folgenden Konformeren:

p1=0 p1=0 p1=0 p1=0 Die Anordnung der Bindungen bei den Richtungstypen r, t und b wird in den nachfolgenden Formeln gezeigt. t-Bindungen werden durch dicke, b-Bindungen durch gestrichelte Linien symbolisiert.

radial top/bottom radial top/bottom

radial top/bottom radial top/bottom

furanose : (Furanose-Ger¨ust) Die Parameter sind hier bedeutungslos. Die Anordnung der Bindungen bei den Richtungstypen r, t und b wird in den nachfolgenden Formeln gezeigt. t-Bindungen werden durch dicke, b-Bindungen durch gestrichelte Linien symbolisiert.

radial top/bottom

3.4 Die Bibliothek bicyclib.pm

In diesem Abschnitt werden die Ringstrukturen erl¨autert, die durch das Bibliotheksmodul bicyclib.pm verf¨ugbar gemacht werden. In einem L^ATEX-Dokument laden Sie es mit den Zeilen

\begin{chemspecial}
  require("bicyclib")
\end{chemspecial}

im Vorspann. Die folgenden Erkl¨arungen folgen dem Schema, das bereits bei →ring angewandt wurde, f¨ur n¨ahere Erkl¨arungen siehe die Beschreibung des Befehls.

Die Bibliothek enth¨alt die folgenden bizyklischen Strukturen und ihre Kodierung

<type>:

bc221h bc311h bc222o bc321o

Die Kodierung stellt dabei eine Abk¨urzung f¨ur Bicyclo[n.m.o]heptan und Bicyclo[n.m.o]oktan dar. Sie erfassen mit den Strukturen die Ger¨uste von Tropin, Norpinan und Norbornan.

Die Numerierung der Atome ist durch die folgenden Formeln beschrieben:

0 1 2 3 4 5 6 0 1 2 3 4 5 6 0 1 2 3 4 5 6 7 0 1 2 3 4 5 6 7 Die Numerierung der Bindungen erfolgt nach den folgenden Schemata.

0 1 2 3 4 5 6 7 0 1 2 3 4 56 7 0 1 2 3 45 67 8 0 1 2 3 4 5 67 8

Die Anordnung der Bindungen bei den Richtungstypen r, t und b wird in den nachfolgenden Formeln gezeigt. t-Bindungen werden durch dicke, b-Bindungen durch gestrichelte Linien symbolisiert.

3.4 — Die Bibliothek bicyclib.pm 129

radial top/bottom radial top/bottom

radial top/bottom radial top/bottom

Die Bedeutung der beiden variablen Parameter <p1> und <p2> ist folgende:

bc221h : (Norbornan-Ger¨ust) Beide Parameter entfallen ersatzlos. (Nach neuen IUPAC-Regeln muß dieses Ger¨ust als 8,9,10-Trinorbornan bezeichnet werden.)

bc311h : (Norpinan-Ger¨ust) Beide Parameter entfallen ersatzlos.

bc321o : (Tropin-Ger¨ust) Beide Parameter entfallen ersatzlos.

Nachfolgend sind einige bizyklische Naturstoffe aus den Ger¨usten der Bibliothek aufgebaut worden.

O Cineol formula(C,C,"Cineol",HA,48)
{ ring("bc222o",,,L)
  { 2: bond(t); 2: bond(b);
    3: atom("O");
    4: bond(r);
  }
}

O

Campher formula(C,C,"Campher",HA,48)
{ ring("bc221h",,,L)
  { 3: bond(r,=C) atom("O");
    4: bond(r);
    6: bond(t); 6: bond(b);
  }
}

COOCH_3

OCOC_6H_5 N H

Kokain formula(C,C,"Kokain",HA,48)
{ ring("bc321o",,,L)
  { 2: bond(t) atom("C",C,R) atom("OOCH$_3$",L);
    3: bond(b) atom("O",C,R) atom("COC$_6$H$_5$",L);
    0: atom("N") bond(r) atom("H");
  }
}

3.5 Die Bibliothek polycyclib.pm

In diesem Abschnitt werden die polyzyklischen Ringstrukturen erl¨autert, die durch das Bibliotheksmodul polycyclib.pm verf¨ugbar gemacht werden. In einem L^ATEX- Dokument laden Sie es mit den Zeilen

\begin{chemspecial}
  require("polycyclib")
\end{chemspecial}

im Vorspann. Die folgenden Erkl¨arungen folgen dem Schema, das bereits bei →ring angewandt wurde, f¨ur n¨ahere Erkl¨arungen siehe die Beschreibung des Befehls.

Die Bibliothek enth¨alt die folgenden polyzyklischen Strukturen und ihre Kodierung

<type>:

tetrahedrane cubane

Sie erfassen mit den Strukturen die Ger¨uste von Tetrahedran und Kuban.

Die Numerierung der Atome ist durch die folgenden Formeln beschrieben: 0 1 2 3 0 1 2 3 4 5 6 7 Die Numerierung der Bindungen erfolgt nach den folgenden Schemata.

0 1 2 3 4 5

0 1 2 3 4 5 6 7 8 9 10 11 Die Anordnung der Bindungen bei den Richtungstypen r, t und b wird in den nachfolgenden Formeln gezeigt. t-Bindungen werden durch dicke, b-Bindungen durch gestrichelte Linien symbolisiert.

radial top/bottom radial top/bottom

Die Bedeutung der beiden variablen Parameter <p1> und <p2> ist folgende:

3.5 — Die Bibliothek polycyclib.pm 131

tetrahedrane : Der Parameter <p1> ist bedeutungslos, wohingegen <p2> den Drehwinkel des gesamten Ger¨ustes darstellt.

cubane : Der Parameter <p1> ist bedeutungslos, wohingegen <p2> den Drehwinkel des gesamten Ger¨ustes darstellt.

Als Beispiel folgt die Formel des 2-tert-Butylcubylcubans.

(H_3C)_3C \begin{chemistry}

ring("cubane",,,L){
  1: bond(r,,L)
     ring("cubane",7,,L,,0){
         0: bond(r) atom("(H$_3$C)$_3$C",R);
     };
}

Als umfangreicheres Beispiel folgt die Synthese des Kubans:

O Zyklopenten-5-on

1. NBS, 2. Br_2

3. N(C_2H_5)_3 O Br

O Br

Br O

hν O Br O Br

1. KOH/H_2O

2. H_3O^⊕

HOOC

COOH

1. SOCl_2

2. HO-C(CH_3)_3, 3. Δ

Kuban

multiline(3,L)
{ % upper line
  formula(L,R,"Zyklopenten-5-on"){
    ring(,,H3=,,5,-90){
      0: bond(r,=C) atom("O");
    }
  }
  arrow(){
    text(T,C){ formula(C,C){ atom("1. NBS, 2. Br$_2$") }}
    text(B,C){ formula(C,C){ atom("3. N(C$_2$H$_5$)$_3$") }}
  }
  space(R)

132 Kapitel 3 — Alphabetische Befehlsreferenz

  formula(L,R){
    ring(,,H1=3=,,5,-90){
      0: bond(r,=C) atom("O");
      4: bond(r) atom("Br",L);
    }
  }
  bracket()
  space(R)
  arrow(){}
  formula(L,R){
    ring("bc221h",,0=,L){
      6: bond(r,=C) atom("O");
      4: bond(r) atom("Br",L);
      3: bond(-30;-90,=) branch { bond(-45) atom("Br",L); } bond(-170)
         saveXY(#1) bond(-120,=) atom("O");
      2: bond(#1);
    }
  }
  ;

  % middle line
  arrow(){
    text(T,C){ formula(C,C){ atom("h$\nu$") }}
  }
  formula(L,R){
    ring("cubane",,1-2-3s5s6-7-9-10-11-,NN){
      3: saveXY(#1); 0: bond(150) branch { bond(135,=C) atom("O"); } bond(#1);
      3: bond(r) atom("Br", R);
      5: saveXY(#2); 1: bond(-60) branch { bond(0,=C) atom("O"); } bond(#2);
      5: bond(r) atom("Br", L);
    }
  }
  arrow(,5){
    text(T,C){ formula(C,C){ atom("1. KOH/H$_2$O") }}
    text(B,C){ formula(C,C){ atom("2. H$_3$O$^\oplus$") }}
  }
  formula(L,R){
    ring("cubane",,1-2-3-6-7-9-10-11-,NN){
      3: bond(r) atom("HOOC", R);
      5: bond(r) atom("COOH", L);
    }
  }
  ;

  % lower line
  arrow(,5){
    text(T,C){ formula(C,C){ atom("1. SOCl$_2$") }}
    text(B,C){ formula(C,C){ atom("2. HO-C(CH$_3$)$_3$, 3. $\Delta$") }}
  }
  formula(L,R,"Kuban"){
    ring("cubane",,,NN){}
  }
  ;
}

3.6 Das Includefile natur.inc

Dem Chemiepaket liegt eine kleine Includedatei natur.inc bei (Listing in Abschnitt ??), in der eine gewiß e (gr¨oß er werdende) Anzahl an Makros zusammengefaß t ist, die die Arbeit mit bestimmten Verbindungsklassen, vor allem Naturstoffen, erleichtern sollen. Sie wird wie folgt in ein L^ATEX-Dokument eingebunden:

\begin{chemspecial}
include(‘natur.inc’)
\end{chemspecial}

Es ist wahrscheinlich, daß Sie die Makros in der vorliegenden Form noch nicht besonders n¨utzlich fi nden; da es aber schwer ist, Makros f¨ur alle denkbaren Anwendungsf¨alle zu schreiben, ohne die Makros mit 50 Parametern zu versehen, schlage ich vor, sie als Fundus zu betrachten und f¨ur die eigenen Bed¨urfnisse Anpassungen vorzunehmen. Wie immer, sind auch in dieser Beziehung Anregungen sehr willkommen, um f¨ur nachfolgende Distributionen eine eventuell sinnvollere Datei zu liefern:-)

Die einzelnen Makros werden im folgenden besprochen. Sie erzeugen jeweils ein typisches Ger¨ust f¨ur eine bestimmte Verbindungsklasse. Wichtige Schl¨usselpositionen werden dabei abgespeichert und k¨onnen f¨ur Modifi kationen mittels restoreXY angesprungen werden; die Nummern der betreffenden Punkte wird jeweils im folgenden Text mitgeteilt. Einige Makros erwarten Parameter, die das Grundger¨ust ebenfalls modifi zieren, zum Beispiel androstane oder cholane f¨ur das Steroidmakro, um Androstan- oder Cholanger¨uste zu erzeugen. Weitere Modifi kationen sind m¨oglich, indem Bindungslisten f¨ur Ringe als Parameter ¨ubergeben werden. Schließ lich k¨onnen kleine Formelbeschreibungen als Parameter auftreten. Anhand der Beispiele sollten Sie jedoch die Anwendung der Makros verstehen k¨onnen.

Beachten Sie, daß diese Makros nicht beliebig verallgemeinert sind, es werden daher immer Strukturen auftreten, die nicht unmittelbar mit diesen Makros zu setzen sind; in diesen F¨allen k¨onnen Sie das Makro kopieren und modifi zieren oder – in komplizierten F¨allen – die ganze Struktur ohne Makro beschreiben, was unter Umst¨anden einfacher sein kann.

TERPEN

Dieses Makro erzeugt ein Menthanger¨ust, das als Grundlage f¨ur einige monozyklische Monoterpene dienen kann. Die allgemeine Form lautet

TERPEN_(<bliste>, <C2-Subs>)

<bliste> ist dabei die Bindungsliste f¨ur den Ring, <C2-Subs> ist die Beschreibung

eines Substituenten an C^2. Ist kein Substituent vorhanden, schreiben Sie einfach ‘’ als leere Liste. Die Positionen von C^1, C^8 und C^3 werden in den Koordinaten Eins bis Drei gespeichert. Die folgende Formel zeigt Grundger¨ust und Lage der gelieferten Punkte:

1

2 3

TERPEN

Einige Beispiele zeigen die Anwendung der Bindungsliste, des Substituenten sowie der gespeicherten Punkte:

1-p-Menthen formula(L,R,"1-p-Menthen",HR,24)
{ TERPEN_(H3=) restoreXY(#1) bond(90) }

OH

Nerol formula(L,R,"Nerol",HR,24)
{ TERPEN_(H3=5s)
  restoreXY(#1) bond(90)
  restoreXY(#3) bond(-30) atom("O",C,R) atom("H",L)
  restoreXY(#2) bond(90,=)
}

O

Carvenon formula(L,R,"Carvenon",HR,24)
{ TERPEN_(H5=,bond(30) atom("O"))
  restoreXY(#1) bond(90)
}

STEROID

Dieses Makro hat den Aufbau

STEROID_(<bliste>, <type>)

und erzeugt Steroidstrukturen gem¨aß <type>. Die m¨oglichen Typen entsprechen den wichtigsten Grundkohlenwasserstoffen und sind in der folgenden Aufstellung enthalten. Die Ziffern an manchen Atomen entsprechen der Nummer des betreffenden gespeicherten Koordinatenpaares:

1 2 3 4 5 6 7

STEROID estrane androstane cholane cholestane ergostane

<bliste> ist eine Bindungsliste, die zum Ring A geh¨ort. Mit ihrer Hilfe kann der Ring

aromatisiert oder mit Doppelbindungen versehen werden. Zur weiteren Modifi kation k¨onnen die folgenden Koordinaten benutzt werden, die auch in der ¨Ubersicht oben durch Ziffern wiedergegeben wurden:

Nummer Atom Nummer Atom

1 C^3 2 C^{10} 3 C^{13} 4 C^5 5 C^{14} 6 C^{17} 7 C^{11}

Einige Beispiele sollen die Anwendung zeigen.

OH

OH

Östradiol formula(L,R,"{}""Ostradiol",HR,24)
{ STEROID_(O)
  restoreXY(#2) bond(90,<<)
  restoreXY(#1) bond(-150) atom("OH",R)
  restoreXY(#6) bond(30) atom("OH",L)
}

O

CHO

Testosteron formula(L,R,"Testosteron",HR,24)
{ STEROID_(H5=, androstane)
  restoreXY(#1) bond(-150,=C) atom("O")
  restoreXY(#6) bond(30) atom("CHO",L)
}

136 Kapitel 3 — Alphabetische Befehlsreferenz

O

O O OH OH

Cortison formula(L,R,"Cortison",HR,24)
{ STEROID_(H5=, androstane)
  restoreXY(#1) bond(-150,=C) atom("O")
  restoreXY(#7) bond(150,=C) atom("O")
  restoreXY(#6) bond(30) branch{ bond(90,=C) atom("O"); }
                bond(-30; 30) atom("OH",L)
  restoreXY(#6) bond(90) atom("O",C,L) atom("H",R)
}

ISOPREN

Dieses Makro erzeugt eine Isopreneinheit in zwei verschiedenen Formen. Sie dient zum Aufbau von Polyprenketten Die allgemeine Form lautet

ISOPREN_(<bond_len>)
ISOPREN1_(<bond_len>, <bond_type>)

<bond_len> ist die Bindungsl¨ange einer Bindung, die zum Startpunkt der eigentli-

chen Einheit f¨uhrt. Sie muß mit der Angabe ‘0’ auf Null gesetzt werden f¨ur eine isolierte Einheit und kann mit ‘’ leer gelassen oder ignoriert werden f¨ur den Fall aneinanderh¨angender Einheit f¨ur Polyprene. Mit <bond_type> k¨onnen Sie die Bindungsform f¨ur eine der Doppelbindungen angeben (Verwendung bei der terminalen ψ-Einheit des Carotens). Die folgenden Formeln zeigen die Grundger¨uste mit

<bond_len> gleich Null:

ISOPREN ISOPREN1 formula(L,R,"I{}SOPREN\_",HR,24)
{ ISOPREN_(0) }
formula(L,R,"I{}SOPREN1\_",HR,24)
{ ISOPREN1_(0) }

Einige Beispiele zeigen die Anwendung der Einheiten zum Aufbau von Carotenen:

β, β-Caroten formula(T,B,"$\beta,\beta$-Caroten",HR,24)
{ ring(,,H4=)
  { 3: bond(t); 3: bond(b);
    5: bond(r);
    4: ISOPREN_
       ISOPREN_ bond(r; r/,=; r)
       ISOPREN1_
       ISOPREN1_ ring(,0,H0=,,,r)
                 { 1: bond(r);
                   5: bond(t); 5: bond(b);
                 };
  }
} β, ψ-Caroten formula(T,B,"$\beta,\psi$-Caroten",HR,24)
{ ring(,,H4=)
  { 3: bond(t); 3: bond(b);
                5: bond(r);
                4: ISOPREN_
                ISOPREN_ bond(r; r/,=; r)
                ISOPREN1_
       ISOPREN1_
       ISOPREN1_(,‘-’)
       bond(r/,=) branch { bond(r/t); } bond(r);
  }
}

AA

Dieser Baustein erleichtert die Darstellung von Aminos¨auren in einer textuellen Art und Weise ohne Ber¨ucksichtigung der Stereostruktur. Die allgemeine Syntax

AA_(<N-Terminus>, <C-Terminus>, <Seitenkette>)

sieht zwei Parameter vor, die den N- beziehungsweise C-Terminus beschreiben. Die Angaben H beziehungsweise OH erzeugen die terminalen Sequenzen H_2N- und - COOH. Diese Positionen k¨onnen frei bleiben und erlauben auf diese Weise die Darstellung von Oligopeptiden. Sie k¨onnen aber auch beliebige Formelsequenzen enthalten, um ungew¨ohnliche Endgruppen zu zeichnen. Der dritte Parameter erzeugt die Seitenkette der Aminos¨aure.

<N-Terminus>NH
                 <Seitenkette>

CH CO<C-Terminus>

AA Die folgenden Beispiele zeigen den Einsatz des Makros. Zur besseren ¨Ubersicht wird dabei die Includedatei utils.inc, die im Abschnitt 3.7 beschrieben wird, eingesetzt.

Die Grundform einer Aminos¨aure wird durch die Termini H und OH erreicht:

H_2N H CH COOH

Glycin formula(L,R,"Glycin",HR,24)
{ AA_(H,OH,atom("H")) }

F¨ur ein einfaches Dipeptid darf die erste Aminos¨aure keinen C-Terminus, die zweite keinen N-Terminus erhalten:

H_2N H CH CO NH CH_2

CH COOH

H-Gly-Val-OH formula(L,R,"H-Gly-Val-OH",HR,24)
{ AA_(H,,atom("H"))
  bond(0)
  AA_(,OH,branch{atom("C",C,R) atom("H$_2$",L);} iPr_(-90))
}

Um eine ungew¨ohnliche Abs¨attigung der terminalen Atome anzuzeigen, k¨onnen auch vollkommen andere Formelelemente spezifi ziert werden:

NH H CH CO NH CH_2

CH CO OCH_2CH_3

Ph-Gly-Val-OiPr formula(L,R,"Ph-Gly-Val-OiPr",HR,24)
{ AA_(ring(,0,,,,0){} bond(0),,atom("H"))
  bond(0)
  AA_(,bond(0) OEt_,
    branch{atom("C",C,R) atom("H$_2$",L);}
    iPr_(-90))
}

3.7 — Das Includefile utils.inc 139

3.7 Das Includefile utils.inc

Eine weitere Includedatei ist utils.inc bei (Listing in Abschnitt ??), in der sehr kleine Makros zu fi nden sind, die immer wiederkehrende Schreibarbeiten abk¨urzen helfen. Sie wird wie folgt in ein L^ATEX-Dokument eingebunden:

\begin{chemspecial}
include(‘utils.inc’)
\end{chemspecial}

Sie enth¨alt die folgenden Makros.

Me_ % erzeugt -CH3
eM_ % dto H3C-
Et_ % erzeugt -CH2CH3
tE_ % dto H3CH2C-
iPr_(<dir>) % dto -CH(CH3)2

OH_ % erzeugt -OH
HO_ % dto HO-
NH2_ % erzeugt -NH2
H2N_ % dto H2N-
NH2p_ % erzeugt -N(+)H2
pH2N_ % dto H2N(+)-
NMe2_ % erzeugt -N(CH3)2
Me2N_ % dto (H3C)2N-
NMe2p_ % erzeugt -N(+)(CH3)2
pMe2N_ % dto (H3C)2N(+)-
OMe_ % dto -OCH3
MeO_ % dto H3CO-
OEt_ % dto -OCH2CH3
EtO_ % dto H3CH2CO-

COOH_ % dto -COOH
HOOC_ % dto HOOC-
COOMe_ % dto -COOCH3
MeOOC_ % dto H3COOC-
COOEt_ % dto -COOCH2CH3
EtOOC_ % dto H3CH2COOC-

SO3H_ % dto -SO3H
HO3S_ % dto HO3S-
SO3m_ % dto -SO3(-)
mO3S_ % dto (-)O3S-

Ph_{...} % Phenyl
Ph_(<angle>){...} % Phenyl

Der Rumpf von Ph_ entspricht dem Rumpf eines normalen ring-Befehls und kann alle dort erlaubten Operationen enthalten oder leer bleiben. iPr_ erzeugt einen abzweigenden Isopropylrest, dessen Richtung mit dem Argument festgelegt werden muß . Dieses besteht aus einer Winkelangabe.

140 Kapitel 3 — Alphabetische Befehlsreferenz

Einige Beispiele zeigen die Anwendungsm¨oglichkeiten. Das Makro zur Erzeugung des Phenylringes weist einen optionalen Parameter zur Angabe des Drehwinkels des Ringes auf, der zur Darstellung von Benzen als Grundk¨orper der Struktur verwendet wird, in den Beispielen 90^o. Der Defaultwinkel r f¨ur diesen Parameter f¨uhrt zur Darstellung von Phenylsubstituenten:

COOH C HOO OH

O H_3C formula(L,R)
{ Ph_(90)
  { 0: iPr_(r);
    1: bond(r) COOH_;
    4: bond(r) iPr_(rt) bond(r/) HOOC_; }
}
formula(L,R)
{ Ph_(90)
  { 1: bond(r) OH_;
    4: bond(r) Ph_{};
    2: bond(r) Ph_{ 5: bond(r) MeO_; };
  }
}

3.8 — Includefiles für Strukturen nach Kekule: condensed.inc und cyclohexanes.inc141

3.8 Includefiles für Strukturen nach Kekule: condensed.inc und cyclohexanes.inc

Diese Includedateien wurden freundlicherweise von Herrn Dr. Kenvard Vaughan bereitgestellt und sind im Unterverzeichnis kv zu fi nden. Sie werden wie die anderen Includedateien auch eingebunden und enthalten Makros, die den Satz von Strukturformeln nach KEKULE bedeutend vereinfachen. Beispiele aus der Dokumentation:

\begin{chemspecial}
define(‘bndlen_’, ‘N’)
include(‘condensed.inc’)
include(‘cyclohexanes.inc’)
\end{chemspecial}

\begin{chemistry}
  formula(L,R){_h(0) hk_(_h) co_ hk_(_h, bond(l,,L) vk_(_h) _h) hk_(_h) _h}
  space(R) space(R)
  formula(L,R){_ch3(,0) ch2_ ch_(,90, _ch2ch3(-), -45) _ch2ch3 }
  space(R) space(R)
  formula(L,R){_cho(-,-30) _kPh(_cho,_br,_och3,,_cooch2ch3(-)) }
\end{chemistry}

H C H H C O C H

C H H H

C H H H CH_3CH_2CH

CH_2C

CH_3

CH_2CH_3 C OH C CHO C Br C OCH_3 C H C C CH_3CH_2OO C

Beispiele f¨ur Zyklohexane:

\begin{chemistry}
  defaulz
  formula(L,R){ chex_(C)
     c1_(_h,_h)
     c2_(_h,_h)
     c3_(_h,_h)
     c4_(_h,_h)
     c5_(_h,_h)
     c6_(_h,_h)
  }
  space(R) space(R) space(R) space(R)
  formula(L,R){ chex_(C)
     c1_(_o)
     c2_(,_oh)
     c3_(,_br)
     c4_(o_(r) _ch3)
  }
\end{chemistry}

C C C C C C

H H H

H H H

H H H

H H H C C C C C C

O OH Br

O CH_3

Zu den beiden Includedateien existieren die ausf¨uhrlichen Dokumentationen condensed_doc.ps und cyclohexanes_doc.ps.

A. Das Programm

A.1 History

20-OCT-1999 : (chemie.pl version 1.0a 1999-10-20, be.pm) Setting of \unitlength enclosed in a group.

09-JAN-2000 : (ochem.sty version 3-0b 2000-01-09, streambuf.pm) Additional bond types >>, >. and b.

17-JAN-2000 : (mncyclib.pm version 1.1)

furanose ring type for saccarid chemistry.

19-JAN-2000 : (ochem.sty version 3-0c 2000-01-19, streambuf.pm, chemie.pl) Additional bond type ~. The length of a bond can now be marked by a various number of markers N,n,L,l,S and s, representing additive terms. Command package allows to pass packages to the intermediate L^ATEX fi le, supporting macros in text position of atom.

19-APR-2000 : Some undesired spaces are eliminated thanks to S. Seckinge, Freiburg (ochem.sty version 3-0d 2000-04-19, be.pm).

10-APR-2001 : Creation of EPS, JPG and PNG supplied. Minor changes in the perl code.

A.2 Danke schön!

Einigen Benutzer (Herr Dr. Kenward Vaughan sowie Herrn Oliver B¨urschaper) verdanke ich wichtige Anregungen (fehlende Bindungstypen, Furanoseringe, die M¨oglichkeit, Makros in atom-Befehlen einzusetzen) und Meldungen von Fehlern.

Herrn Dr. Kenward Vaughan danke ich fernerhin f¨ur seinen Einsatz bei der Entwicklung der Makros zur Darstellung von Formeln in der Kekule-Darstellung, die im Unterverzeichnis kv enthalten sind.

144

B. Mehr Stoff für Chemiesüchtige

Wer Formeln zeichnet, muß sie m¨oglicherweise auch benennen, was unter Umst¨anden beliebig diffi zil werden kann. Auf den Internetseiten der Firma acdlabs ist jedoch eine online-Version des IUPAC-Reglements zu fi nden:

http://www.acdlabs.com/iupac/nomenclature/

Eine sch¨one Einf¨uhrung in die korrekte Nomenklatur organischer Verbindungen ist D. Hellwinkel, Die systematische Nomenklatur organischer Verbindungen, Springer- Verlag Heidelberg.

Als WYSIWYG-Konkurrenzprodukt zum vorliegenden L^ATEX-Paket ist das chemische Zeichenprogramm ChemSketch der oben genannten Firma als Freeware-Version f¨ur Dosenrechner verf¨ugbar:

http://www.acdlabs.com/download

Es kann allerdings nicht nur Formeln zeichnen (sehr sch¨on sogar), sondern auch verschiedene Berechnungen durchf¨uhren und dreidimensionale Stab- oder Kugerl- (Kalotten)modelle zeichnen. Auß erdem besitzt es einen umfangreichen Katalog mit Substanzen, Eigenschaften ....

146

Index

A AA, 137 Abstand, 119 Aktives Phosphat, 122 Aminosäure, 137 Anomer, 57 Anomerenbindung, 85 atom, 81 Atropin, 111, 129 B Befehle

atom, 81 bond, 83 bracket, 90 branch, 91 emove, 94 fbox, 96 font, 121 formula, 97 gotoXY, 98 joinh, 92 joinv, 93 multiline, 99 nospace, 80, 97 orbital, 100 package, 122 require, 124 restore, 101 restoreXY, 102 ring, 103 save, 110 savecontext, 110 saveXY, 111 scale, 114 schema, 125 set, 116 setcontext, 117 shiftXY, 115 space, 119 Bibliothek

bicyclib.pm, 128 entwickeln, 72 mncyclib.pm, 126 polycyclib.pm, 130 Bibliotheken

laden, 7 bicyclib.pm, 128 Bindung

anomere, 57 Anomeren-, 85 Bizyklen, 129 bond, 83

r+, 21 r-, 21 r/, 22, 24 r/t, 23 rt, 23 Winkel, 86 Bornan, 129 bracket, 90 branch, 91 C Cadinen, 25 Campher, 129 Carminsüre, 9 Caroten, 136 chemistry (Umgebung), 3 Chlorin, 24 Cineol, 129 Crassanin, 29 E Elektronen

Konfiguration, 62 Verschiebung, 63 emove, 94 EPS-Format, 9 F Fünfringe, 56 fbox, 96 font, 121 Formeln

als Einheit, 41 formula, 97 Furanosen, 56

G gotoXY, 98 Graphikformate (EPS, JPG, PNG), 9 Guajan, 25 H Hayworth-Wiedergabe, 56, 85 Hirsutan, 25 Humulen, 27 I Indigo, 9 Internet

Publikation im, 9 Isoprene, 136 J joinh, 92 joinv, 93 JPG-Format, 9 K Kampher, 129 Kauran, 32 Kekule-Strukturformel, 141 Kette

Reaktions-, 33 verzweigt, 40 Kohlenhydrate, 55 Kokain, 129 Kommentarzeile, 6 Kontext, 40

Super-, 43 Kuban, 131

Synthese, 131

L Lycopodin, 31 M Makro, 12

AA, 137 Entscheidungen, 13 Makros

Isoprene, 136 Steroide, 134 Terpene, 133 mncyclib.pm, 126 Monozyklen, 126 Morphin, 31 multiline, 99 N Naturstoffe, 102, 111, 128, 130 Neue Ringstrukturen, 124 Norbornan, 129 Norpinan, 129 nospace, 80, 97 O orbital, 100 P package, 122 Phenylsubstituent, 109 Phosphat

aktives, 122 Pinan, 129 PNG-Format, 9 polycyclib.pm, 130 Polyzyklen, 102, 111, 130 Porphin, 24

Präprozessor, 12 Publizieren

im Internet (WWW), 9 Purpur, 9 Pyranosen, 57 R r+, 21 r-, 21 r/, 22, 24 r/t, 23 Reaktionskette, 33

verzweigt, 40 require, 7, 124 restore, 101 restoreXY, 102 ring, 103 Ringe, 55

eigene, 124 eigene Strukturen, 72 irregulär, 102, 111 rt, 23 S Saccharide, 55 save, 110 savecontext, 110 saveXY, 111 scale, 114 schema, 125 set, 116 setcontext, 117 shiftXY, 115 space, 119 Stereodarstellung, 102, 126 Zyklobutan, 126

Zyklohexan, 103 Zyklopentan, 59 Zyklpentan, 126 Steroide, 59, 134 Strukturen

entwickeln, 72 Strukturformel

nach Kekule, 141 Substituent

Phenyl, 109 Superkontext, 43 Symbolische Winkel, 20, 86 Synthese

Kuban, 131 T Terpene, 133 Tetrahedran, 131 Thujan, 111 Tropin, 129 Tutorial, 15 U Umgebung

chemistry, 3 V Valeran, 59 Verbindungslinie, 88 Verzweigung

in Formel, 91 in Reaktionskette, 40 W Webformate, 9 Winkel r+, 21 r-, 21 r/, 22, 24 r/t, 23 rt, 23 symbolische, 20, 86 WWW Publikation im, 9 Z Zucker, 55 Zyklobutan, 126

Stereodarstellung, 126 Zyklohexan, 103

Stereodarstellung, 103 Zyklopentan, 59, 126 Stereodarstellung, 59, 126, 127